

Guide Definition Language (GDL)

Revision	0.11
Editors	Rong Chen ^a , Iago Corbal ^a
Date of issue	20 December 2012

^a Cambio Healthcare Systems

© 2012 Cambio Healthcare Systems

Contents

1 Introduction.....	4
1.1 Background.....	4
1.2 Purpose.....	4
1.3 Scope	4
1.4 Related Documents	4
1.5 Nomenclature.....	4
1.6 Requirements	4
1.7 Design Principles.....	5
1.7.1 Archetypes both as Input and Output of Rules	5
1.7.2 Natural Language Neutrality	5
1.7.3 Rule Language Neutrality	5
1.7.4 Grouping and Reuse of Rules	6
1.7.5 Meta-information of the CDS rules	6
1.8 An Example.....	6
2. Guide Object Model	8
2.1 Design Background	8
2.2. Packages Structure	8
3 Guide Package	9
3.1 Overview.....	9
3.2 Class Definitions	9
3.2.1 GUIDE Class	9
3.2.2 GUIDE_DEFINITION.....	10
3.2.3 ARCHETYPE_BINDING.....	10
3.2.4 ELEMENT_BINDING	10
3.2.5 RULE.....	10
4 Expressions Package.....	11
4.1 Overview.....	11
4.2 Class Definitions	11
4.2.1 EXPRESSION_ITEM.....	11
4.2.2 UNARY_EXPRESSION	11
4.2.3 BINARY_EXPRESSION.....	12
4.2.4 ASSIGNMENT_EXPRESSION	12

4.2.5 FUNCTIONAL_EXPRESSION	12
5 Implementation	12
5.1 Drools	13
6 Tools	14
6.1 GDL Editor.....	14

Address for correspondence

Rong Chen, M.D, Ph.D.

Cambio Healthcare Systems

Ringvägen 100

SE-118 60 Stockholm

Sweden

rong.chen@cambio.se

1 Introduction

1.1 Background

Expressing and sharing computerized clinical decision support (CDS) content across languages and technical platforms has been an evasive goal. Lack of commonly shared clinical information models and flexible support for various terminology resources have been identified as two main challenges for sharing detailed clinical rules between sites.

1.2 Purpose

This document contains the design specifications of the guide definition language (GDL). GDL is a formal language designed to represent clinical knowledge for computerized decision support. GDL is designed to be natural language- and reference terminology- agnostic by leveraging the designs of openEHR Reference Model and Archetype Model.

1.3 Scope

The scope of the GDL is to support expressions of clear-cut clinical knowledge for single-decision making. Discrete GDL guides, artifacts written in a self-contained document in GDL format, can be combined together to support complex decision making. It does not cover the process aspects of the clinical guideline.

1.4 Related Documents

- *openEHR* Reference Model Data Types Information Model ([1.0.2](#))
- *openEHR* Reference Model Data Structures Information Model ([1.0.2](#))
- *openEHR* Reference Model EHR Information Model ([1.0.2](#))
- *openEHR* Reference Model Common Information Model ([1.0.2](#))
- *openEHR* Archetype Model Archetype Object Model (AOM) ([1.0.2](#))
- *openEHR* Archetype Model Archetype Definition Language (ADL) ([1.0.2](#))

1.5 Nomenclature

In this document, the term ‘attribute’ denotes any stored property of a type defined in an object model, including primitive attributes and any kind of relationship such as an association or aggregation.

1.6 Requirements

1. It must be possible to express CDS rules using archetypes both as input and output for the rule execution.

2. It must be natural language-agnostic and able to support multiple language translations without changing the rule definitions.
3. It must be reference terminology-agnostic so different terminologies can be used to support reasoning.
4. It should be straight-forward to convert the CDS rules to main-stream general-purpose rule languages for execution.
5. It must be able to express medical knowledge as first-order predicate rules.
6. There must be sufficient meta-information about the CDS rules, e.g. authorship, purpose, versions and relevant references.
7. It must be possible to reuse CDS rules written in the new format in different clinical contexts.
8. It should be possible to group a set of related CDS rules in order to support complex decision making.

1.7 Design Principles

In response to the above mentioned requirements, the following principles are applied in the GDL design.

1.7.1 Archetypes both as Input and Output of Rules

This is achieved by creating bindings between data elements defined by archetypes and variables used by the CDS rules. Each CDS rule variable is uniquely identified in the context of a guide and bond to a specific data element defined by an archetype using Archetype ID and a path. Once defined, the variable can be used inside the **when** and **then** statements as input or output during rule execution.

1.7.2 Natural Language Neutrality

Several design ideas from openEHR archetype formalism are used to achieve natural language neutrality. First of all, all language-dependent meta-information about the purpose, use, misuse and references of the guide are grouped together under **description** and indexed by ISO language codes inside the GDL guide. Secondly, all natural language-dependent labels and descriptions, e.g. the name of a rule variable, are defined in the **term_definitions** section of the guide and indexed by ISO language codes. Thirdly, unique identifiers for variables and rules are used in rule expressions instead of their names, which are language dependent.

1.7.3 Reference Terminology Neutrality

Again, useful openEHR design is reused here to keep the CDS rules agnostic to reference terminologies. When IS_A operator is used in the evaluation statements for subsumption relationship checking, a locally defined term is used instead of an external code. This indirection makes it possible to modify the code or to add new codes from other terminologies without changing the rule definitions. The bindings between locally defined codes and external reference terminologies are maintained in the **term_bindings** section of the guide.

1.7.4 Rule Language Neutrality

GDL only uses a set of common rule language features, like **when** and **then**. The expressions in the when and then statements support common arithmetic calculations, logic operator and functions.

1.7.5 Grouping and Reuse of Rules

A GDL guide may contain several rules that relate to each other. Each guide is self-containing and should be reusable across different clinical contexts. Different guides can be chained together to support complex decision support. This is achieved by selecting the output of a rule, as a specific element of an archetype, as the input of another rule.

1.7.6 Meta-information of the CDS rules

Authoring information, lifecycle state and various meta-information are supported by reuse of RESOURCE_DESCRIPTION class from the openEHR design.

1.8 An Example

The following is a simple GDL example that allows us to calculate [CHA₂DS₂VASc Score](#). The definition for each one of the keywords used here can be found on the next section. The GDL header follows the same specifications as the [openEHR's ADL Description Section](#). The GDL source illustrates the current version of the guide, the natural languages, to which it has been translated, authors, lifecycle state, keywords and information about the purpose, use and misuse of the guide.

```
(GUIDE) <
  gdl_version = <"0.1">
  id = <"CHA2DS2VASc_Score.v1">
  concept = <"gt0036">
  language = (LANGUAGE) <
    original_language = <[ISO_639-1::en]>
  >
  description = (RESOURCE_DESCRIPTION) <
    details = <
      ["en"] = (RESOURCE_DESCRIPTION_ITEM) <
        copyright = <" ">
        purpose = <" ">
      >
    >
  >
  lifecycle_state = <"Author draft">
  original_author = <
    ["date"] = <"2012/12/03">
    ["email"] = <"rong.chen@cambio.se">
    ["name"] = <"Rong Chen">
    ["organisation"] = <"Cambio Healthcare Systems">
  >

  other_contributors = <"Dr. Carlos Valladares",...>

>
```

The following block contains the definition section, where the bindings to the archetypes and elements used inside the guide are defined.

```
definition = (GUIDE_DEFINITION) <
  archetype_bindings = <
    [1] = (ARCHETYPE_BINDING) <
      archetype_id = <"openEHR-EHR-EVALUATION.problem-diagnosis.v1">
      domain = <"EHR">
      elements = <
        ["gt0003"] = (ELEMENT_BINDING) <
          path = <"/data[at0001]/items[at0002.1]">
        >
      >
    >
  >
```

GDL provides also a section to define a set of conditions that have to be met before the rules inside the guide can be executed. In the case of [CHA₂DS₂VASc Score](#), the guideline will not be executed unless the patient has been diagnosed with atrial fibrillation. In the example below, a pre-condition checks all the diagnosis of the patient (*gt0003*) against a local code (*gt0017*) which represents the meaning of atrial fibrillation. This code can be bound with external terminologies, e.g. SNOMED CT, later on in the ***term_bindings*** section.

```
definition = (GUIDE_DEFINITION) <
...
  pre_condition = <"gt0003 is_a local::gt0017|Atrial fibrillation|",...>
...
>
```

The ***rule*** section makes exclusive use of the locally defined elements to express the core logic of the guides. Each rule has a local gt code as an identifier, with which its language-dependent name and description are indexed in the ***term_definitions*** section. Also a ***priority*** can be assigned to ensure execution of the rules can be prioritized. This example illustrates rules that inspect different diagnoses relevant to [CHA₂DS₂VASc Score](#) and set the values of the DV_ORDINALS inside a [CHA₂DS₂VASc Score](#) Archetype, the rule gt0026 (Calculate total score) sums up all the values and sets the total score.

```
rules = <
  ["gt0018"] = (RULE) <
    when = <"$gt0003 is_a local::gt0101|Hypertension|",...>
    then = <"$gt0014=1|local::at0031|Present|",...>
    priority = <11>
  >
  ["gt0019"] = (RULE) <
    when = <"$gt0003 is_a local::gt0102|Diabetes|",...>
    then = <"$gt0010=1|local::at0034|Present|",...>
    priority = <10>
  >
  ...
  ["gt0026"] = (RULE) <
    then = <"$gt0016.magnitude((((($gt0009.value + $gt0010.value) + $gt0011.value) +
$gt0015.value) + $gt0012.value) + $gt0013.value) + $gt0014.value)",...>
    priority = <1>
  >
```

Finally we have the ontology block (see *openEHR's* [ADL Ontology Section](#)), where all the terms are translated into the natural languages supported by the guideline ***term_definitions***.

```
term_definitions = <
  ["en"] = (TERM_DEFINITION) <
    terms = <
      ["gt0003"] = (TERM) <
        text = <"Diagnosis">
      >
      ["gt0101"] = (TERM) <
        text = <"Hypertension">
      >
      ["gt0102"] = (TERM) <
        text = <"Diabetes">
      >
      ["gt0017"] = (TERM) <
        text = <"Atrial fibrillation">
      >
      ["gt0018"] = (TERM) <
        text = <"Set hypertension">
      >
      ["gt0019"] = (TERM) <
        text = <"Set diabetes">
      >
      ["gt0026"] = (TERM) <
        text = <"Calculate total score">
      >
    >
  >

```

Inside the **term_bindings** local defined terms are bond to external terminologies. In this sample, the diagnosis of atrial fibrillation is bond to two external terminologies, ICD10 and SNOMED CT respectively. Note that the binding of SNOMED CT is via a URI, which points to a reference set maintained by an external terminology service.

```
term_bindings = <
  ["ICD10"] = <
    items = <
      ["gt0017"] = (TERM) <
        codes = <"I21, I22, I73.9, I70">
      >
    >
  >
  ["SNOMED-CT"] = <
    items = <
      ["gt0017"] = (TERM) <
        refset = <"http://termserver.imt.liu.se?UID=2342342">
      >
    >
  >
  >

```

2 Guide Object Model

2.1 Design Background

The underpinning of GDL design is the *openEHR* archetypes, both as input and output of CDS rules. This is the key to achieve natural language-independence and reference terminology-independence. Because of this design choice, *openEHR* specification plays a major role in the GDL design. In other words, the GDL design is aimed to make substantial reuse of existing *openEHR* specifications. In areas where existing *openEHR* design is not sufficient, additional designs are introduced.

2.2. Packages Structure

The Guide Object Model, the object model of the GDL, consists of two packages, the guide package and the expressions package described in detail through the next two sections.

3 Guide Package

3.1 Overview

The overview of the guide package is illustrated in Figure 1. Note that classes in blue color are loosely based on the original design from the openEHR specifications.




Figure 1 – The Guide Package

3.2 Class Definitions

3.2.1 GUIDE Class

CLASS	GUIDE	
Purpose	Main class of a discrete guide, which defines archetype bindings, rules and meta-information.	
Attributes	Signature	Meaning
0..1	gdl_version: String	the version of the GDL the guide is written in.
1..1	id: String	Identification of this guide
1..1	concept: String	The normative meaning of the guide as whole. Expressed as a local guide code.
1..1	language: Language	Natural language resources of this guide. It includes an original language and optional list of translations.
1..1	description: RESOURCE_DESCRIPTION	Resources description of this guide including authorship, use/misuse, life-cycle and references.
1..1	definition: GUIDE_DEFINITION	The main definition part of the guide. It consists of archetype bindings and rule definitions.
1..1	ontology: GUIDE_ONTOLOGY	The ontology of the guide.

3.2.2 GUIDE_DEFINITION

CLASS		GUIDE_DEFINITION
Purpose	The definition of the guide includes a list of archetype bindings and a list of rule definitions.	
Attributes	Signature	Meaning
1..1	archetype_bindings: List<ARCHETYPE_BINDING>	List of archetype bindings, which define specific elements to be used by rules.
1..1	rules: Map<String, Rule>	Map of rules indexed by local gt codes.
0..1	pre_conditions: List<EXPRESSION_ITEM>	List of pre-conditions to be met before the guide should be executed.

3.2.3 ARCHETYPE_BINDING

CLASS		ARCHETYPE_BINDING
Purpose	The binding of list of elements from a selected archetype or template to local gt codes	
Attributes	Signature	Meaning
1..1	archetype_id: String	The ID of the archetype, from where the list of elements is selected.
0..1	template_id: String	The ID of an optional template to be used for selecting elements.
0..1	domain: String	The space in which the rule variables reside. The value can either be "EHR" meaning the value is retrieved from the EHR, or "CDS" meaning the value is derived in the CDS engine. When missing, the assumption is either "EHR" or "CDS".
0..1	function: String	The function to perform when querying the EHR information. Typical function can be "count" or "average".
0..1	time_range: String	The time range, with which the archetype instances are retrieved from EHR.
1..1	Elements: Map<String, ELEMENT_BINDING>	Map of element binding indexed by local gt codes.
0..1	predicate_statements: List<EXPRESSION_ITEM>	List of predicates to be fulfilled before the EHR queries can be performed

3.2.4 ELEMENT_BINDING

CLASS		ELEMENT_BINDING
Purpose	The binding between a specific element in an archetype and a local variable in the guide.	
Attributes	Signature	Meaning
1..1	id: String	The local gt code of this element
1..1	path: String	The path to reach this element in the archetype.

3.2.5 RULE

CLASS	RULE
Purpose	A single rule defined in a guide

Attributes	Signature	Meaning
1..1	id: String	The local gt code of this element
1..1	when_statements: List<EXPRESSION_ITEM>	List of expressions to be evaluated before the rule can be fired.
1..1	then_statements: List<ASSIGNMENT_EXPRESSION>	List of expressions to generate output of the rule.

4 Expressions Package

4.1 Overview

The overview of the expressions package is illustrated by figure 2.




Figure 2 – The Expression Package

4.2 Class Definitions

4.2.1 EXPRESSION_ITEM

CLASS	EXPRESSION_ITEM (abstract)
Purpose	Abstract model of an expression item in the rule.

4.2.2 UNARY_EXPRESSION

CLASS	UNARY_EXPRESSION
Purpose	Abstract model of an expression item in the rule.

Inherit	EXPRESSION_ITEM	
Attributes	Signature	Meaning
1..1	operand: EXPRESSION_ITEM	The operand of this unary expression.
1..1	operator: OPERATOR_KIND	The operator of this unary expression.

4.2.3 BINARY_EXPRESSION

CLASS BINARY_EXPRESSION		
Purpose	Concrete model of a binary expression item.	
Inherit	EXPRESSION_ITEM	
Attributes	Signature	Meaning
1..1	left: EXPRESSION_ITEM	The left operand of this binary expression.
1..1	right: EXPRESSION_ITEM	The right operand of this binary expression.
1..1	operator: OPERATOR_KIND	The operator of this binary expression.

4.2.4 ASSIGNMENT_EXPRESSION

CLASS ASSIGNMENT_EXPRESSION		
Purpose	Concrete model of an assignment expression.	
Inherit	EXPRESSION_ITEM	
Attributes	Signature	Meaning
1..1	variable: String	The gt code of the variable to assign the value to.
1..1	assignment: EXPRESSION_ITEM	The expression item, from which the value is derived from.

4.2.5 FUNCTIONAL_EXPRESSION

CLASS FUNCTIONAL_EXPRESSION		
Purpose	Concrete expression models a function.	
Inherit	EXPRESSION_ITEM	
Attributes	Signature	Meaning
1..1	function: String	The name of the function.
1..1	items: List<EXPRESSION_ITEM>	A list of parameters to the function.

4.2.6 OPERATOR_KIND

CLASS OPERATOR_KIND		
Purpose	Enumeration containing all the operators used.	
Type	Name	Symbol
Arithmetic	Addition	+
Arithmetic	Subtraction	-
Arithmetic	Multiplication	*

Arithmetic	Division	/
Arithmetic	Exponent	^
Logical	And	&&
Logical	Or	
Logical	Not	!
Relational	Equal	==
Relational	Unequal	!=
Relational	Less than	<
Relational	Less than or equal	<=
Relational	Greater than	>
Relational	Greater than or equal	>=
Assignment	Assignment	=
Terminological reasoning	Is a	is_a
Terminological reasoning	Is not a	!is_a
Conditional	For all	for_all

5 Implementation

5.1 Drools

GDL is technology independent, so it can be implemented using different rule engines. We've chosen [JBoss Drools](#), an open source technology, to develop our first implementation of GDL execution engine. Drools allows us to refer directly to our Java objects inside the rules. It provides support for a powerful expression language (MVEL) and has proven robust and efficient enough for our purposes. Our drools module will be able to translate our GDL language into Drools Rule Language (DRL).

The following list summarizes the considerations for the drools-based implementation:

- Each rule in GDL will generate a new rule in DRL
- Each rule will have as a title the guide id and the GT code (*rule "CHADVAS_Score.v1/gt0028"*)
- Priority will be mapped to *salience* in DRL language
- All rules will have a *no-loop* attribute, to disable the execution loops inside one rule
- Current time variable will have to be defined using the *DvDateTime* class
- All elements used inside the expressions will be checked for definition prior to execution

- Modification of the elements will use the *modify* method to propagate the changes inside the knowledge base
- General definition section: Drools does not support declaration of elements outside the rule's scope. Each one of the elements had to be declared inside each rule.
- Preconditions: there is no support for this type of conditions. All preconditions will be copied inside each one of the rules.

6 Tools

6.1 GDL Editor

An authoring tool for GDL will be released as open-source software for the community. The GDL editor (Figure 3) is multiplatform application and will allow users to create, edit and run GDL files. We provide a testing environment capable of generating forms (Figure 4) based on the archetype elements defined in the GDL. These forms can be used to capture data from the user and trigger the rules.

The support for multiple natural languages in a single guide is illustrated in Figure 5, 6. The support for reference terminology binding is shown in Figure 7.

Figure 3 - Description section of the GDL Editor




Figure 4 - GDL Editor using the form generator




Figure 5 – Rule editing in Spanish language




Figure 6 – Spanish user interface terms defined in the GFR guide




Figure 7 – ICD10 terminology bindings in CHA2DS2VAsC Score guide