The *open*EHR Archetype Model

Archetype Definition Language ADL 1.5

*Issuer: open*EHR Specification Program		
Revision: 1.5.0	*Pages:* 127	*Date of issue:* 12 Jan 2014
Status: TRIAL		

Keywords: EHR, ADL, health records, modelling, constraints



Amendment Record

Issue	Details	Raiser	Completed
RELEASE 1.1 candidate			
1.5.0	**SPEC-xxx**: new format term codes & mandatory node ids. **SPEC-xxx**: remove dADLsyntax description to new ODIN specification. **SPEC-xxx**: Add archetype id namespacing. **SPEC-316**: Allow ‘cardinality’ keyword on its own to indicate container attributes, even if no other constraints required. **SPEC-270**: Add specialisation semantics to ADL and AOM. Improve explanation of node identifiers. Correct typographical errors. Make *concept* optional. Refine slot include/exclude rules. Review of text & validity rules. Deprecate *terminologies_available* statement in terminology section. **SPEC-xxx**. Add fractional seconds to dADL grammar. **SPEC-303**. Make existence, occurrences and cardinality optional in AOM. **SPEC-300**. Archetype slot regular expressions should cover whole identifier. Added C_STRING.*is_pattern*.	T Beale T Beale T Beale, I McNicoll D Moner S Heard O Pishev T Beale T Beale S Arikan S Garde S Heard A Flinton	12 Jan 2014
RELEASE 1.0.2			
1.4.1	**SPEC-268**: Correct missing parentheses in dADL type identifiers. dADL grammar and cADL scanner rules updated. **SPEC-284**: Correct inconsistencies in naming of *term_definitions*, *constraint_definitions*, *term_bindings*, *constraint_bindings* attributes in XML-schema and specifications. Improved explanatory text for composite identifiers, including statement on case-sensitivity. Warning on .v1draft non-conformance **SPEC-260**: Correct the regex published for the ARCHETYPE_ID type. Update ADL grammar ARCHEYTPE_ID definition.	R Chen A Torrisi P Gummer, J Arnett, E Browne	12 Dec 2008
RELEASE 1.0.1			

Issue	Details	Raiser	Completed
1.4.0	**CR-000203**: Release 1.0 explanatory text improvements. Improve Archetype slot explanation. **CR-000208**: Improve ADL grammar for assertion expressions. **CR-000160**: Duration constraints. Added ISO 8601 patterns for duration in cADL. **CR-000213**: Correct ADL grammar for date/times to be properly ISO8601-compliant. Include 'T' in cADL patterns and dADL and cADL Date/time, Time and Duration values. **CR-000216**: Allow mixture of W, D etc in ISO8601 Duration (deviation from standard). **CR-000200**: Correct Release 1.0 typographical errors. **CR-000225**: Allow generic type names in ADL. **CR-000226**: Rename C_CODED_TEXT to C_CODE_PHRASE **CR-000233**: Define semantics for *occurrences* on ARCHETYPE_INTERNAL_REF. **CR-000241**: Correct cADL grammar for archeype slot match expressions **CR-000223**: Clarify quoting rules in ADL **CR-000242**: Allow non-inclusive two-sided ranges in ADL. **CR-000245**: Allow term bindings to paths in archetypes.	T Beale T Beale S Heard T Beale S Heard A Patterson R Chen S Garde T Beale M Forss T Beale K Atalag S Heard A Patterson S Heard S Heard	13 Mar 2007
RELEASE 1.0			
1.3.1	**CR-000136**. Add validity rules to ADL document. **CR-000171**. Add validity check for cardinality & occurrences	T Beale A Maldondo	18 Jan 2006
1.3.0	**CR-000141**. Allow point intervals in ADL. Updated atomic types part of cADL section and dADL grammar section. **CR-000142**. Update dADL grammar to support assumed values. **CR-000143**. Add partial date/time values to dADL syntax. **CR-000149**. Add URIs to dADL and remove query() syntax. **CR-000153**. Synchronise ADL and AOM for language attributes **CR-000156**. Update documentation of container types. **CR-000138**. Archetype-level assertions.	S Heard T Beale T Beale T Beale T Beale T Beale T Beale	18 Jun 2005
RELEASE 0.95			
1.2.1	**CR-000125**. C_QUANTITY example in ADL manual uses old dADL syntax. **CR-000115**. Correct "/[xxx]" path grammar error in ADL. Create new section describing ADL path syntax. **CR-000127**. Restructure archetype specifications. Remove clinical constraint types section of document.	T Beale T Beale T Beale	11 Feb 2005

Issue	Details	Raiser	Completed
1.2	**CR-000110**. Update ADL document and create AOM document. Added explanatory material; added domain type support; rewrote of most dADL sections. Added section on assumed values, “controlled” flag, nested container structures. Change language handling. Rewrote OWL section based on input from: - University of Manchester, UK, - University Seville, Spain. Various changes to assertions due to input from the DSTC. Detailed review from Clinical Information Project, Australia. **Remove UML models to “Archetype Object Model” document.** Detailed review from UCL. **CR-000103**. Redevelop archetype UML model, add new keywords: allow_archetype, include, exclude. CR-000104. Fix ordering bug when use_node used. Required parser rules for identifiers to make class and attribute identifiers distinct. Added grammars for all parts of ADL, as well as new UML diagrams.	T Beale A Rector R Qamar I Román Martínez A Goodchild Z Z Tun E Browne T Beale T Austin T Beale K Atalag	15 Nov 2004
RELEASE 0.9			
1.1	**CR-000079**. Change interval syntax in ADL.	T Beale	24 Jan 2004
1.0	**CR-000077**. Add cADL date/time pattern constraints. **CR-000078**. Add predefined clinical types. Better explanation of cardinality, occurrences and existence.	S Heard, T Beale	14 Jan 2004
0.9.9	**CR-000073**. Allow lists of Reals and Integers in cADL. **CR-000075**. Add predefined clinical types library to ADL. Added cADL and dADL object models.	T Beale, S Heard	28 Dec 2003
0.9.8	**CR-000070**. Create Archetype System Description. Moved Archetype Identification Section to new Archetype System document. Copyright Assgined by Ocean Informatics P/L Australia to The *open*EHR Foundation.	T Beale, S Heard	29 Nov 2003
0.9.7	Added simple value list continuation (“,...”). Changed path syntax so that trailing ‘/’ required for object paths. Remove ranges with excluded limits. Added terms and term lists to dADL leaf types.	T Beale	01 Nov 2003
0.9.6	Additions during HL7 WGM Memphis Sept 2003	T Beale	09 Sep 2003
0.9.5	Added comparison to other formalisms. Renamed CDL to cADL and dDL to dADL. Changed path syntax to conform (nearly) to Xpath. Numerous small changes.	T Beale	03 Sep 2003
0.9	Rewritten with sections on cADL and dDL.	T Beale	28 July 2003
0.8.1	Added basic type constraints, re-arranged sections.	T Beale	15 July 2003
0.8	Initial Writing	T Beale	10 July 2003

Trademarks

"Microsoft" and ".Net" are registered trademarks of the Microsoft Corporation.

"Java" is a registered trademark of Sun Microsystems.

"Linux" is a registered trademark of Linus Torvalds.

Acknowledgements

The work reported in this document was funded by:

- University College London, Centre for Health Informatics and Multi-professional Education (CHIME);
- Ocean Informatics;
- Thomas Beale.

Thanks to Sebastian Garde, Central Qld University, Australia, for German translations.

Table of Contents

1 Introduction

1.1 Purpose

This document describes the design basis and syntax of the Archetype Definition Language (ADL). It is intended for software developers, technically-oriented domain specialists and subject matter experts (SMEs). ADL is designed as an abstract human-readable and computer-processible syntax. ADL archetypes can be hand-edited using a normal text editor.

The intended audience includes:

- Standards bodies producing health informatics standards;
- Software development organisations using *open*EHR;
- Academic groups using *open*EHR;
- The open source healthcare community;
- Medical informaticians and clinicians interested in health information;
- Health data managers.

1.2 Related Documents

Prerequisite documents for reading this document include:

- The *open*EHR Architecture Overview

Related documents include:

- The *open*EHR Archetype Object Model (AOM) specification
- The *open*EHR Archetype Profile (oAP) specification
- The *open*EHR Template specification

1.3 Nomenclature

In this document, the term 'attribute' denotes any stored property of a type defined in an object model, including primitive attributes and any kind of relationship such as an association or aggregation. Where XML is mentioned, XML 'attributes' are always referred to explicitly as 'XML attributes'.

1.4 Status

This document is under development, and is published as a proposal for input to standards processes and implementation works.

This document is available at `http://www.openehr.org/svn/specification/Releases/1.0.2/publishing/architecture/am/adl.pdf`.

The latest version of this document can be found in PDF format at http://www.openehr.org/svn/specification/TRUNK/publishing/architecture/am/adl.pdf. New versions are announced on `openehr-announce@openehr.org`.

1.5 Peer review

Known omissions or questions are indicated in the text with a "to be determined" paragraph, as follows:

TBD_1: (example To Be Determined paragraph)

Areas where more analysis or explanation is required are indicated with "to be continued" paragraphs like the following:

To Be Continued: more work required

Reviewers are encouraged to comment on and/or advise on these paragraphs as well as the main content. Please send requests for information to `info@openEHR.org`. Feedback should preferably be provided on the mailing list `openehr-technical@openehr.org`, or by private email.

1.6 Conformance

Conformance of a data or software artefact to an *open*EHR Reference Model specification is determined by a formal test of that artefact against the relevant *open*EHR Implementation Technology Specification(s) (ITSs), such as an IDL interface or an XML-schema. Since ITSs are formal, automated derivations from the Reference Model, ITS conformance indicates RM conformance.

1.7 Changes From Previous Versions

For existing users of ADL or archetype development tools, the following provides a guide to the changes in the syntax.

1.7.1 dADL / ODIN

The object syntax used to represent the description, terminology and annotations sections of an ADL archetype has been historically known as 'dADL' (i.e. 'data ADL'). Since this syntax is completely generic and has no specific dependency on either ADL or *open*EHR, it has been separated out into its own specification known as Object Data Instance Notation (ODIN).

1.7.2 ADL 1.5

Changes

The changes in version 1.5 are made to better facilitate the representation of specialised archetypes. The key semantic capability for specialised archetypes is to be able to support a differential representation, i.e. to express a specialised archetype only in terms of the changed or new elements in its definition, rather than including a copy of unchanged elements. Doing the latter is clearly unsustainable in terms of change management. ADL 1.4 already supported differential representation, but somewhat inconveniently.

The changes for ADL 1.5 include:

- .adls files are introduced as the standard file extension for differential ADL files (.adl files are retained for standalone, inheritance-flattened, or 'flat', archetype);
- optional 'generated' marker in the archetype first line;
- the semantics of reference model subtype matching are now described;
- rules are provided for when node identifiers (at-codes) are required in archetypes;

- the ability to state a path rather than just an attribute, allowing redefinition blocks deep in the structure to be stated with respect to a path rather than having to be nested within numerous containing blocks on which no redefinition occurs;
- new keywords for defining the order of specialised object nodes within container attributes;
- an explanation of how to use the negated match operator (~matches, or ∉) to define value set exclusions in specialised archetypes;
- rules for the construction of node identifier at-codes in specialised archetypes;
- a description of the semantics of 'inheritance-flattened' archetypes;
- optional annotations section added to archetypes;
- 'declarations' and 'invariants' sections merged into 'rules' section;
- In the ADL grammar, the `language` section is now mandatory.

Nearly all the changes occur in the section on cADL - Constraint ADL on page 22 or the new section Specialisation on page 82.

Backward Compatibility

ADL 1.5 and the corresponding changes to the AOM do not change ADL 1.4 or invalidate any existing ADL 1.4 archetypes; they just add further semantic validation rules. ADL 1.4 archetypes are therefore *syntactically* valid under ADL 1.5, but some of them may fail second pass validation, which is useful, because the ADL 1.5 validation rules detect errors that went previously undetected.

ADL 1.5 contains two pieces of new syntax: object path navigation to constraints, and ordering markers, both of which only apply in source files, not flat files. In ADL 1.5, flat (.adl) files look just the same as previously. An ADL 1.5 capable tool will generate .adl files consumable by older tools.

1.7.3 ADL 1.4

A number of small changes were made in this version, along with significant tightening up of the explanatory text and examples.

ISO 8601 Date/Time Conformance

All ISO 8601 date, time, date/time and duration values in dADL are now conformant (previously the usage of the 'T' separator was not correct). Constraint patterns in cADL for dates, times and date/times are also corrected, with a new constraint pattern for ISO 8601 durations being added. The latter allows a deviation from the standard to include the 'W' specifier, since durations with a mixture of weeks, days etc is often used in medicine.

Non-inclusive Two-sided Intervals

It is now possible to define an interval of any ordered amount (integer, real, date, time, date/time, duration) where one or both of the limits is not included, for example:

```
|0..<1000|        -- 0 >= x < 1000
|>0.5..4.0|       -- 0.5 > x <= 4.0
|>P2d..<P10d|     -- 2 days > x < 10 days
```

Occurrences for 'use_node' References

Occurrences can now be stated for use_node references, overriding the occurrences of the target node. If no occurrences is stated, the target node occurrences value is used.

Quoting Rules

The old quoting rules based on XML/ISO mnemonic patterns (&ohmgr; etc) are replaced by specifying ADL to be UTF-8 based, and any exceptions to this requiring ASCII encoding should use the “\Uhhhh” style of quoting unicode used in various programming languuages.

1.7.4 ADL 1.3

The specific changes made in version 1.3 of ADL are as follows.

Query syntax replaced by URI data type

In version 1.2 of ADL, it was possible to include an external query, using syntax of the form:

```
attr_name = <query("some_service", "some_query_string")>
```

This is now replaced by the use of URIs, which can express queries, for example:

```
attr_name = <http://some.service.org?some%20query%20etc>
```

No assumption is made about the URI; it need not be in the form of a query - it may be any kind of URI.

Top-level Invariant Section

In this version, invariants can only be defined in a top level block, in a way similar to object-oriented class definitions, rather than on every block in the definition section, as is the case in version 1.2 of ADL. This simplifies ADL and the Archetype Object Model, and makes an archetype more comprehensible as a “type” definition.

1.7.5 ADL 1.2

ADL Version

The ADL version is now optionally (for the moment) included in the first line of the archetype, as follows.

```
archetype (adl_version=1.2)
```

It is strongly recommended that all tool implementors include this information when archetypes are saved, enabling archetypes to gradually become imprinted with their correct version, for more reliable later processing. The `adl_version` indicator is likely to become mandatory in future versions of ADL.

dADL Syntax Changes

The dADL syntax for container attributes has been altered to allow paths and typing to be expressed more clearly, as part of enabling the use of Xpath-style paths. ADL 1.1 dADL had the following appearance:

```
school_schedule = <
    locations(1) = <...>
    locations(2) = <...>
    locations(3) = <...>
    subjects("philosophy:plato") = <...>
    subjects("philosophy:kant") = <...>
    subjects("art") = <...>
>
```

This has been changed to look like the following:

```
school_schedule = <
    locations = <
        [1] = <...>
```

```
            [2] = <...>
            [3] = <...>
        >
        subjects = <
            ["philosophy:plato"] = <...>
            ["philosophy:kant"] = <...>
            ["art"] = <...>
        >
```

The new appearance both corresponds more directly to the actual object structure of container types, and has the property that paths can be constructed by directly reading identifiers down the backbone of any subtree in the structure. It also allows the optional addition of typing information anywhere in the structure, as shown in the following example:

```
    school_schedule = SCHEDULE <
        locations = LOCATION <
            [1] = <...>
            [2] = <...>
            [3] = ARTS_PAVILLION <...>
        >
        subjects = <
            ["philosophy:plato"] = ELECTIVE_SUBJECT <...>
            ["philosophy:kant"] = ELECTIVE_SUBJECT <...>
            ["art"] = MANDATORY_SUBJECT <...>
        >
```

These changes will affect the parsing of container structures and keys in the description and terminology parts of the archetype.

Revision History Section

Revision history is now recorded in a separate section of the archetype, both to logically separate it from the archetype descriptive details, and to facilitate automatic processing by version control systems in which archtypes may be stored. This section is included at the end of the archetype because it is in general a monotonically growing section.

Primary_language and Languages_available Sections

An attribute previously called 'primary_language' was required in the ontology section of an ADL 1.1 archetype. This is renamed to 'original_language' and is now moved to a new top level section in the archetype called 'language'. Its value is still expressed as a dADL String attribute. The 'languages_available' attribute previously required in the ontology section of the archetype is renamed to 'translations', no longer includes the original languages, and is also moved to this new top level section.

1.8 Tools

A validating reference ADL parser is freely available from http://www.openEHR.org. The *open*EHR ADL Workbench GUI application uses this parser and is available for all major platforms at http://www.openehr.org/svn/ref_impl_eiffel/TRUNK/apps/doc/adl_workbench_help.htm. It has been wrapped for use in the Microsoft .Net, and standard C/C++ environments. A Java ADL parser is available at http://www.openehr.org/svn/ref_impl_java/TRUNK/docs/download.htm.

2 Overview

2.1 Context

Archetypes form the second layer of the *open*EHR semantic architecture. They provide a way of creating models of domain content, expressed in terms of constraints on a reference model. Archetypes are composed into templates, enabling them to be used for building screen forms, message schemas and other derived expressions. Archetype paths provide the basis of querying in *open*EHR as well as bindings to terminology.



FIGURE 1 The *open*EHR Semantic Architecture

Archetypes are defined in terms of the following specifications:

- the Archetype Definition Language (ADL) - this specification;
- The openEHR Archetype Object Model (AOM);
- openEHR Archetype Profile (oAP).

The ADL syntax is semantically equivalent to the AOM. ADL documents are parsed into in-memory objects (known as a ‘parse tree’) which are defined by the Archetype Object Model (AOM) class definitions. AOM instances can in turn be serialised other formalisms, including as XML obeying a W3C XML schema, JSON, YAML and other formats. An archetype can thus be serialised as ADL or in its XML form, and parsed from either form into its object form. The AOM XML-schema (.xsd) form is published on the *open*EHR website.

The AOM is the definitive expression of archetype semantics, and is independent of any particular syntax. The Archetype Definition Language is a formal abstract syntax for archetypes, used to provide a default serial expression of archetypes, and as the explanatory framework for most of the semantics.

2.2 Overview of the Formalism

2.2.1 Syntactic Structure

ADL uses three syntaxes, cADL (constraint form of ADL), ODIN (Object Data Instance Notation), and a version of first-order predicate logic (FOPL), to express constraints on data which are instances of an underlying information model, which may be expressed in UML, relational form, or in a programming language. ADL itself is a very simple 'glue' syntax, which uses two other syntaxes for expressing structured constraints and data, respectively. The cADL syntax is used to express the archetype `definition`, while the ODIN syntax is used to express data which appears in the `language`, `description`, `terminology`, and `revision_history` sections of an ADL archetype. The top-level structure of an ADL archetype is shown in FIGURE 2.

This main part of this document describes cADL and ADL path syntax, before going on to describe the combined ADL syntax, archetypes and domain-specific type libraries.



FIGURE 2 ADL Archetype Structure

2.2.2 An Example

The following is an example of a very simple archetype, giving a feel for the syntax. The main point to glean from the following is that the notion of 'guitar' is defined in terms of *constraints* on a *generic* model of the concept `INSTRUMENT`. The names mentioned down the left-hand side of the definition section ("INSTRUMENT", "size" etc) are alternately class and attribute names from an object model. Each block of braces encloses a specification for some particular set of instances that conform to a specific concept, such as 'guitar' or 'neck', defined in terms of constraints on types from a generic class model. The leaf pairs of braces enclose constraints on primitive types such as Integer, String, Boolean and so on.

```
archetype (adl_version=1.5)
    adl-test-instrument.guitar.draft
language
    original_language = <[iso_639-1::en]>
definition
    INSTRUMENT[id1] matches {
        size matches {|60..120|}                      -- size in cm
        date_of_manufacture matches {yyyy-mm-??}      -- year & month ok
        parts cardinality matches {0..*} matches {
            PART[id2] matches {                       -- neck
                material matches {[local::
                                   at3,               -- timber
                                   at4]}              -- timber or nickel alloy
            }
            PART[id3] matches {                       -- body
                material matches {[local::at3}        -- timber
            }
        }
    }
terminology
    id_definitions = <
        ["en"] = <
            ["id1"] = <
                text = <"guitar">;
                description = <"stringed instrument">
            >
            ["id2"] = <
                text = <"neck">;
                description = <"neck of guitar">
            >
            ["id3"] = <
                text = <"body">;
                description = <"body of guitar">
            >
        >
    >
    term_definitions = <
        ["en"] = <
            ["at3"] = <
                text = <"timber">;
                description = <"straight, seasoned timber">
            >
            ["at4"] = <
                text = <"nickel alloy">;
                description = <"frets">
            >
        >
    >
```

2.3 The Role of ADL

While ADL is a normative syntax formalism for archetypes, the Archetype Object Model defines the semantics of an archetype, in particular relationships that must hold true between the parts of an archetype for it to be valid as a whole. Other syntaxes are therefore possible. In particular, XML-schema based XML is used for many common computing purposes relating to archetypes, and may become the dominant syntax in terms of numbers of users. Nevertheless, XML is not used as the normative syntax for archetypes for a number of reasons.

- There is no single XML schema that is likely to endure. Initially published schemas are replaced by more efficient ones, and may also be replaced by alternative XML syntaxes, e.g. the XML-schema based versions may be replaced or augmented by Schematron.
- XML has little value as an explanatory syntax for use in standards, as its own underlying syntax obscures the semantics of archetypes or any other specific purpose for which it is used. Changes in acceptable XML expressions of archetypes described above may also render examples in documents such as this obsolete.

An XML schema for archetypes is available on the *open*EHR website.

3 File Encoding and Character Quoting

3.1 File Encoding

Because ADL files are inherently likely to contain multiple languages due to internationalised authoring and translation, they must be capable of accommodating characters from any language. ADL files do not explicitly indicate an encoding because they are assumed to be in UTF-8 encoding of unicode. For ideographic and script-oriented languuages, this is a necessity.

There are three places in ADL files where non-ASCII characters can occur:

- in string values, demarcated by double quotes, e.g. “xxxx”;
- in regular expression patterns, demarcated by either // or ^^;
- in character values, demarcated by single quotes, e.g. ‘x’;

Note that URIs (a data type in ODIN) are assumed to be ‘percent-encoded’ according to RFC 3986[1], which applies to all characters outside the ‘unreserved set’. The unreserved set is:

```
unreserved  = ALPHA / DIGIT / "-" / "." / "_" / "~"
```

In actual fact, ADL files encoded in latin 1 (ISO-8859-1) or another variant of ISO-8859 - both containing accented characters with unicode codes outside the ASCII 0-127 range - may work perfectly well, for various reasons:

- the contain nothing but ASCII, i.e. unicode code-points 0 - 127; this will be the case in English-language authored archetypes containing no foreign words;
- some layer of the operating system is smart enough to do an on-the-fly conversion of characters above 127 into UTF-8, even if the archetype tool being used is designed for pure UTF-8 only;
- the archetype tool (or the string-processing libraries it uses) might support UTF-8 and ISO-8859 variants.

For situations where binary UTF-8 (and presumably other UTF-* encodings) cannot be supported, ASCII encoding of unicode characters above code-point 127 should only be done using the system supported by many programming languages today, namely \u escaped UTF-16. In this system, unicode codepoints are mapped to either:

- \uHHHH - 4 hex digits which will be the same (possibly 0-filled on the left) as the unicode code-point number expressed in hexadecimal; this applies to unicode codepoints in the range U+0000 - U+FFFF (the ‘base multi-lingual plane’, BMP);
- \uHHHHHHHH - 8 hex digits to encode unicode code-points in the range U+10000 through U+10FFFF (non-BMP planes); the algorithm is described in IETF RFC 2781[2].

It is not expected that the above approach will be commonly needed, and it may not be needed at all; it is preferable to find ways to ensure that native UTF-8 can be supported, since this reduces the burden for ADL parser and tool implementers. The above guidance is therefore provided only to ensure a standard approach is used for ASCII-encoded unicode, if it becomes unavoidable.

Thus, while **the only officially designated encoding for ADL and its constituent syntaxes is UTF-8**, real software systems may be more tolerant. This document therefore specifies that any tool

1. Uniform Resource Identifier (URI): Generic Syntax, Internet proposed standard, January 2005; see http://www.ietf.org/rfc/rfc3986.txt
2. see http://tools.ietf.org/html/rfc2781.

designed to process ADL files need only support UTF-8; supporting other encodings is an optional extra. This could change in the future, if required by the ADL or *open*EHR user community.

3.2 Special Character Sequences

In strings and characters, characters not in the lower ASCII (0-127) range should be UTF-8 encoded, with the exception of quoted single- and double quotes, and some non-printing characters, for which the following customary quoted forms are allowed (but not required):

- \r - carriage return
- \n - linefeed
- \t - tab
- \\ - backslash
- \” - literal double quote
- \’ - literal single quote

Any other character combination starting with a backslash is illiegal; to get the effect of a literal backslash, the \\ sequence should always be used.

Typically in a normal string, including multi-line paragraphs as used in ODIN, only \\ and \” are likely to be necessary, since all of the others can be accommodated in their literal forms; the same goes for single characters - only \\ and \’ are likely to commonly occur. However, some authors may prefer to use \n and \t to make intended formatting clearer, or to allow for text editors that do not react properly to such characters. Parsers should therefore support the above list.

In regular expressions (only used in cADL string constraints), there will typically be backslash-escaped characters from the above list as well as other patterns like \s (whitspace) and \d (decimal digit), according to the PERL regular expression specification[1]. These should not be treated as anything other than literal strings, since they are processed by a regular expression parser.

1. http://www.perldoc.com/perl5.6/pod/perlre.html

4 cADL - Constraint ADL

4.1 Overview

cADL is a block-structured syntax which enables constraints on data defined by object-oriented information models to be expressed in archetypes or other knowledge definition formalisms. It is most useful for defining the specific allowable configurations of data whose instances conform to very general object models. cADL is used both at design time, by authors and/or tools, and at runtime, by computational systems which validate data by comparing it to the appropriate sections of cADL in an archetype. The general appearance of cADL is illustrated by the following example:

```
PERSON matches {                          -- constraint on a PERSON instance
    name matches {                        -- constraint on PERSON.name
        TEXT matches {/.+/}               -- any non-empty string
    }
    addresses cardinality matches {1..*} matches { -- constraint on
        ADDRESS matches {                 -- PERSON.addresses
            -- etc --
        }
    }
}
```

Some of the textual keywords in this example can be more efficiently rendered using common mathematical logic symbols. In the following example, the `matches` keyword have been replaced by an equivalent symbol:

```
PERSON ∈ {                                -- constraint on a PERSON instance
    name ∈ {                              -- constraint on PERSON.name
        TEXT ∈ {/..*/}                    -- any non-empty string
    }
    addresses cardinality ∈ {1..*} ∈ { -- constraint on
        ADDRESS ∈ {                       -- PERSON.addresses
            -- etc --
        }
    }
}
```

The full set of equivalences appears below. Raw cADL is persisted in the text-based form, to remove any difficulties when authoring cADL text in normal text editors, and to aid reading in English. However, the symbolic form might be more widely used for display purposes and in more sophisticated tools, as it is more succinct and less language-dependent. The use of symbols or text is completely a matter of taste, and no meaning whatsoever is lost by completely ignoring one or other format according to one's personal preference. This document uses both conventions.

In the standard cADL documented in this section, literal leaf values (such as the regular expression `/.+/` in the above example) are always constraints on a set of 'standard' widely-accepted primitive types, as described in the section dADL - Data ADL on page 22.

4.2 Basics

4.2.1 Keywords

The following keywords are recognised in cADL:

- `matches, ~matches, is_in, ~is_in`

- occurrences, existence, cardinality
- ordered, unordered, unique
- infinity
- use_node, allow_archetype
- include, exclude
- before, after

Symbol equivalents for some of the above are given in the following table.

Textual Rendering	Symbolic Rendering	Meaning
matches, is_in	∈	Set membership, “p is in P”
not, ~	~	Negation, “not p”
infinitiy	*	Infinity, ‘any number of...’

Keywords are shown in blue in this document.

4.2.2 Block / Node Structure

cADL constraints are written in a block-structured style, similar to block-structured programming languages like C. A typical block resembles the following (the recurring pattern /.+/ is a regular expression meaning “non-empty string”):

```
PERSON ∈ {
    name ∈ {
        PERSON_NAME ∈ {
            forenames cardinality ∈ {1..*} ∈ {/.+/}
            family_name ∈ {/.+/}
            title ∈ {"Dr", "Miss", "Mrs", "Mr", ...}
        }
    }
    addresses cardinality ∈ {1..*} ∈ {
        LOCATION_ADDRESS[id3] ∈ {
            street_number existence ∈ {0..1} ∈ {/.+/}
            street_name ∈ {/.+/}
            locality ∈ {/.+/}
            post_code ∈ {/.+/}
            state ∈ {/.+/}
            country ∈ {/.+/}
        }
    }
}
```

In the above, an identifier (shown in green in this document) followed by the ∈ operator (equivalent text keyword: matches or is_in) followed by an open brace, is the start of a ‘block’, which continues until the closing matching brace (normally visually indented to match the line at the beginning of the block).

Two kinds of identifiers from the underlying information model are used, in alternation: type names (shown in upper case in this document) and attribute names (shown in lower case).

Blocks introduced by a type name are known as *object blocks* or *object nodes*, while those introduced by an attribute name are *attribute blocks* or *attribute nodes* as illustrated below.

FIGURE 3 Object and Attribute Blocks in cADL

An object block or node can be thought of as a constraint matching a set of instances of the type which introduces the block.

The example above expresses a constraint on an instance of the type PERSON; the constraint is expressed by everything inside the PERSON block. The two blocks at the next level define constraints on properties of PERSON, in this case *name* and *addresses*. Each of these constraints is expressed in turn by the next level containing constraints on further types, and so on. The general structure is therefore a recursive nesting of constraints on types, followed by constraints on attributes (of that type), followed by types (being the types of the attribute under which it appears) until leaf nodes are reached.

> A **cADL text** is a structure of alternating object and attribute blocks each introduced respectively by type names and attribute names from an underlying information model.

4.2.3 Comments

In a cADL text, comments are defined as follows:

> **Comments** are indicated by the characters "--". **Multi-line comments** are achieved using the "--" leader on each line where the comment continues.

In this document, comments are shown in brown.

4.2.4 The Underlying Information Model

Identifiers in cADL texts correspond to entities - types and attributes - in an information model. The latter is typically an object-oriented model, but may just as easily be an entity-relationship model or any other typed model of information. A UML model compatible with the example above is shown in FIGURE 4. Note that there can be more than one model compatible with a given fragment of cADL syntax, and in particular, there are usually more properties and classes in the reference model than are mentioned in the cADL constraints. In other words, a cADL text includes constraints *only for those parts of a model that are useful or meaningful to constrain*.

Constraints expressed in cADL cannot be stronger than those from the information model. For example, the PERSON.*family_name* attribute is mandatory in the model in FIGURE 4, so it is not valid to express a constraint allowing the attribute to be optional. In general, a cADL archetype can only further constrain an existing information model. However, it must be remembered that for very generic models consisting of only a few classes and a lot of optionality, this rule is not so much a limitation as a way of adding meaning to information. Thus, for a demographic information model which has only the types PARTY and PERSON, one can write cADL which defines the concepts of entities such as COM-



FIGURE 4 UML Model of PERSON

PANY, EMPLOYEE, PROFESSIONAL, and so on, in terms of constraints on the types available in the information model.

This general approach can be used to express constraints for instances of any information model. The following example shows how to express a constraint on the *value* property of an ELEMENT class to be a QUANTITY with a suitable range for expressing blood pressure.

```
ELEMENT[id10] matches {      -- diastolic blood pressure
    value matches {
       QUANTITY matches {
          magnitude matches {|0..1000|}
          property matches {"pressure"}
          units matches {"mm[Hg]"}
       }
    }
}
```

In this specification, the terms *underlying information model* and *reference model* are equivalent and refer to the information model on which a cADL text is based.

4.2.4.1 Information Model Identifiers

Identifiers from the underlying information model are used to introduce all cADL nodes. Identifiers obey the same rules as in ODIN: type names commence with an upper case letter, while attribute and function names commence with a lower case letter. In cADL, names of types and the name of any property (i.e. attribute or parameterless function) can be used.

> A **type name** is any identifier with an initial upper case letter, followed by any combination of letters, digits and underscores. A **generic type name** (including nested forms) additionally may include commas and angle brackets, but no spaces, and must be syntactically correct as per the OMG UML 2.x specification or higher. An **attribute name** is any identifier with an initial lower case letter, followed by any combination of letters, digits and underscores. Any convention that obeys this rule is allowed.

Type identifiers are shown in this document in all uppercase, e.g. PERSON, while attribute identifiers are shown in all lowercase, e.g. home_address. In both cases, underscores are used to represent word breaks. This convention is used to improve the readability of this document, and other conventions may be used, such as the common programmer's mixed-case convention exemplified by Person and homeAddress. The convention chosen for any particular cADL document should be based on that used in the underlying information model.

4.2.5 The matches Operator

The matches or is_in operator deserves special mention, since it is the key operator in cADL. This operator can be understood mathematically as set membership. When it occurs between an identifier and a block delimited by braces, the meaning is: the set of values allowed for the entity referred to by the name (either an object, or parts of an object - attributes) is specified between the braces. What appears between any matching pair of braces can be thought of as a *specification for a set of values*. Since blocks can be nested, this approach to specifying values can be understood in terms of nested sets, or in terms of a value space for instances of a type. Thus, in the following example, the matches operator links the name of an entity to a linear value space (i.e. a list), consisting of all words ending in “ion”.

```
aaa matches {/[^\s\n\t]+ion[\s\n\t]/}-- the set of words ending in 'ion'
```

The following example links the name of a type XXX with a hierarchical value space.

```
XXX matches {
    aaa matches {
       YYY matches {0..3}
    }
    bbb matches {
       ZZZ matches {>1992-12-01}
    }
}
```

The meaning of the syntax above is: data matching the constraints conssists of an instance of type XXX, or any subtype allowed by the underlying information model, for which the value of attribute *aaa* is of type YYY, or any subtype allowed by the underlying information model, and so on, recursively until leaf level constraints are reached.

Occasionally the matches operator needs to be used in the negative, usually at a leaf block. Any of the following can be used to constrain the value space of the attribute aaa to any number except 5:

```
aaa ~matches {5}
aaa ~is_in {5}
aaa ∉{5}
```

The choice of whether to use matches or is_in is a matter of taste and background; those with a mathematical background will probably prefer is_in, while those with a data processing background may prefer matches.

4.2.6 Natural Language

cADL is completely independent of all natural languages. The only potential exception is where constraints include literal values from some language, and this is easily and routinely avoided by the use of separate language and terminology definitions, as used in ADL archetypes. However, for the purposes of readability, comments in English have been included in this document to aid the reader. In real cADL documents, comments are generated from the archetype terminology in the language of the locale.

4.3 Constraints on Complex types

This section describes the semantics for constraining objects of complex, i.e. non-primitive types. The semantics apply recursively through a constraint structure until leaf nodes constraining primitive types are reached.

4.3.1 Attribute Constraints

In any information model, attributes are either single-valued or multiply-valued, i.e. of a generic container type such as `List<Contact>`. Both have *existence*, while multiply-valued attributes also have *cardinality*.

4.3.1.1 Existence

The existence constraint may be used with any attribute to further constrain the existence defined by the underlying reference model. An existence constraint indicates whether an attribute value is mandatory or optional, and is indicated by "0..1" or "1" markers at line ends in UML diagrams (and often mistakenly referred to as a "cardinality of 1..1"). Attributes defined in the reference model have an effective existence constraint, defined by the invariants (or lack thereof) of the relevant class. For example, the *protocol* attribute in the *open*EHR `OBSERVATION` class[1] is defined in the reference model as being optional, i.e. 0..1. An archetype may redefine this to {1..1}, making the attribute mandatory. Existence constraints are expressed in cADL as follows:

```
OBSERVATION matches {
    protocol existence matches {1..1} matches {
       -- details
    }
}
```

The meaning of an existence constraint is to indicate whether a value - i.e. an object - is mandatory or optional (i.e. obligatory or not) in runtime data for the attribute in question. The above example indicates that a value for the 'units' attribute is optional. The same logic applies whether the attribute is of single or multiple cardinality, i.e. whether it is a container type or not. For container attributes, the existence constraint indicates whether the whole container (usually a list or set) is mandatory or not; a further *cardinality* constraint (described below) indicates how many members in the container are allowed.

> An **existence constraint** may be used directly after any attribute identifier, and indicates whether the object to which the attribute refers is mandatory or optional in the data.

Existence is shown using the same constraint language as the rest of the archetype definition. Existence constraints can take the values `{0}`, `{0..0}`, `{0..1}`, `{1}`, or `{1..1}`. The first two of these constraints may not seem initially obvious, but can be used to indicate that an attribute must not be present in the particular situation modelled by the archetype. This may be reasonable in some cases.

4.3.2 Single-valued Attributes

A single-valued attribute is an attribute whose type as declared in the underlying class model is of a single object type rather than a container type such as a list or set. Single-valued attributes can be constrained with a single object constraint as shown in the following example.

```
value matches {
    QUANTITY matches {
       magnitude matches {|0..55|}
```

1. See http://www.openehr.org/releases/1.0.1/architecture/rm/ehr_im.pdf

```
            property matches {"velocity"}
            units matches {"mph"}
        }
    }
```

Multiple *alternative* object constraints can also be defined, using a number of sibling blocks, as shown in the following example. Each block defines an alternative constraint, only one of which needs to be matched by the data.

```
    value matches {
        DV_QUANTITY[id22] matches {-- miles per hour
            magnitude matches {|0..55|}
            property matches {"velocity"}
            units matches {"mph"}
        }
        DV_QUANTITY[id23] matches {-- km per hour
            magnitude matches {|0..100|}
            property matches {"velocity"}
            units matches {"km/h"}
        }
    }
```

Here the occurrences of both `DV_QUANTITY` constraints is not stated, leading to the result that only one `DV_QUANTITY` instance can appear in runtime data, matching either one of the constraints.

> Two or more object constraints introduced by type names appearing after a single-valued attribute (i.e. one for which there is no cardinality constraint) are taken to be **alternative constraints**, only one of which is matched by the data.

4.3.3 Container Attributes

4.3.3.1 Cardinality

The cardinality of container attributes may be constrained in cADL with the *cardinality* constraint. Cardinality indicates limits on the number of instance members of a container types such as lists and sets. Consider the following example:

```
    HISTORY occurrences ∈ {1} ∈ {
        periodic ∈ {False}
        events cardinality ∈ {*} ∈ {
            EVENT[id2] occurrences ∈ {0..1} ∈ {}      -- 1 min sample
            EVENT[id3] occurrences ∈ {0..1} ∈ {}      -- 2 min sample
            EVENT[id4] occurrences ∈ {0..1} ∈ {}      -- 3 min sample
        }
    }
```

The `cardinality` keyword implies firstly that the property *events* must be of a container type, such as `List<T>`, `Set<T>`, `Bag<T>`. The integer range indicates the valid membership of the container; a single '*' means the range 0..*, i.e. '0 to many'. The type of the container is not explicitly indicated, since it is usually defined by the information model. However, the semantics of a logical set (unique membership, ordering not significant), a logical list (ordered, non-unique membership) or a bag (unordered, non-unique membership) can be constrained using the additional keywords `ordered`, `unordered`, `unique` and `non-unique` within the cardinality constraint, as per the following examples:

```
        events cardinality ∈ {*; ordered} ∈ {               -- logical list
        events cardinality ∈ {*; unordered; unique} ∈ {     -- logical set
        events cardinality ∈ {*; unordered} ∈ {             -- logical bag
```

If no numeric or ordering constraint on the cardinality of a container attribute is required, the keyword is used on its own, and simply indicates that the attribute is a container, as in the following example:

```
events cardinality ∈ {      -- indicates 'events' is a container
```

Although this is not strictly ncessary for the purpose of expressing valid archetypes if the Reference Model can usually be referred to, it enables early stage parsing to generate the correct type of attributes without referring to a Reference Model schema, which in any case may not always be available. This in turn enables more faithful visualisation at an earlier point in the archetype compilation process.

In theory, no cardinality constraint can be stronger than the semantics of the corresponding container in the relevant part of the reference model. However, in practice, developers often use lists to facilitate data integration, when the actual semantics are intended to be of a set; in such cases, they typically ensure set-like semantics in their own code rather than by using an `Set<T>` type. How such constraints are evaluated in practice may depend somewhat on knowledge of the software system.

> A **cardinality constraint** must be used after any Reference Model container attribute name (or after its existence constraint, if there is one) in order to designate it as a container attribute. Additionally, it may constrain the number of member items it may have in the data, and whether it has “list”, “set”, or “bag” semantics, via the use of the keywords ‘ordered’, ‘unordered’, ‘unique’ and ‘non-unique’.

The numeric part of the cardinality contraint can take the values `{0}`, `{0..0}`, `{0..n}`, `{m..n}`, `{0..*}`, or `{*}`, or a syntactic equivalent. The first two of these constraints are unlikely to be useful, but there is no reason to prevent them. There is no default cardinality, since if none is shown, the relevant attribute is assumed to be single-valued (in the interests of uniformity in archetypes, this holds even for smarter parsers that can access the reference model and determine that the attribute is in fact a container.

Cardinality and existence constraints can co-occur, in order to indicate various combinations on a container type property, e.g. that it is optional, but if present, is a container that may be empty, as in the following:

```
events existence ∈ {0..1} cardinality ∈ {0..*} ∈ {-- etc --}
```

4.3.4 Object Constraints

4.3.4.1 Node Identifiers

In cADL, an entity in brackets of the form `[idN]` following a type name is used to identify an object node, i.e. a node constraint delimiting a set of instances of the type as defined by the reference model. Object nodes always commence with a type name. Although any node identifier format could be supported, the current version of ADL assumes that node identifiers are of the form of an archetype term identifier, i.e. `[idN]`, e.g. `[id42]`. Node identifiers are shown in magenta in this document.

The structural function of node identifiers is to allow the formation of paths:

- enable cADL nodes in an archetype definition to be unambiguously referred to within the same archetype;
- enable data created using a given archetype to be matched at runtime;
- to enable cADL nodes in a parent archetype to be unambiguously referred to from a specialised child archetype;
- to enable unique paths to be formed.

All object nodes require a node identifier, guaranteeing the ability to generate unique paths, and to process specialised archetypes with respect to inheritance parents.

> **A Node identifier** is required for every object node in an archetype.

The node identifier can also perform a semantic function, that of giving a design-time *meaning* to the node, by equating the node identifier to some description. **The use of node identifiers in archetypes is the main source of their expressive power**. Each node identifier acts as a 'semantic marker' or 'override' on the node. Thus, in the example shown in section 4.2.4, the `ELEMENT` node is identified by the code `[id10]`, which can be designated elsewhere in an archetype as meaning "diastolic blood pressure". In this way rich meaning is given to data constructed from a limited number of object types.

Not every node identifier needs to be defined in the archetype terminology: it is only mandatory for nodes defined under container attributes. Nodes defined under single-valued attributes may have terminology definitions, but don't typically need them, since the meaning is obvious from the attribute.

4.3.4.2 Occurrences

A constraint on occurrences is used only with cADL object nodes, to indicate how many times in data an instance conforming to the constraint can occur. It is usually only defined on objects that are children of a container attribute, since by definition, the occurrences of an object that is the value of a single-valued attribute can only be 0..1 or 1..1, and this is already defined by the attribute's `existence`. However, it may be used in specialised archetypes to exclude a possibility defined in a parent archetype (see Attribute Redefinition on page 88).

In the example below, three `EVENT` constraints are shown; the first one ("1 minute sample") is shown as mandatory, while the other two are optional.

```
events cardinality ∈ {*} ∈ {
    EVENT[id2] occurrences ∈ {1..1} ∈ {}    -- 1 min sample
    EVENT[id3] occurrences ∈ {0..1} ∈ {}   -- 2 min sample
    EVENT[id4] occurrences ∈ {0..1} ∈ {}    -- 3 min sample
}
```

The following example expresses a constraint on instances of `GROUP` such that for `GROUP`s representing tribes, clubs and families, there can only be one "head", but there may be many members.

```
GROUP[at0103] ∈ {
    kind ∈ {/tribe|family|club/}
    members cardinality ∈ {*} ∈ {
        PERSON[at0104] occurrences ∈ {1} ∈ {
            title ∈ {"head"}
            -- etc --
        }
        PERSON[at0105] occurrences ∈ {0..*} ∈ {
            title ∈ {"member"}
            -- etc --
        }
    }
}
```

The first `occurrences` constraint indicates that a `PERSON` with the title `"head"` is mandatory in the `GROUP`, while the second indicates that at runtime, instances of `PERSON` with the title `"member"` can number from none to many. Occurrences may take the value of any range including `{0..*}`, meaning that any number of instances of the given type may appear in data, each conforming to the one constraint block in the archetype. A single positive integer, or the infinity indicator, may also be used on its own, thus: `{2}`, `{*}`. A range of `{0..0}` or `{0}` indicates that no occurrences of this object are allowed in this archetype. If no occurrences constraint is stated, the occurrences of the object is define by the underlying reference model.

An **occurrences constraint** may appear directly after the type name of any object constraint within a container attribute, in order to indicate how many times data objects conforming to the block may occur in the data.

Where cardinality constraints are used (remembering that occurrences is always there by default, if not explicitly specified), cardinality and occurrences must always be compatible. The rules for this are formally stated in the Archetype Object Model specification. The key elements of these rules are as follows:

- where a cardinality constraint is stated with a finite upper bound:
    - any child object with either stated occurrences with an open upper bound (typically 0..* or 1..*) or else inferred occurrences (0..*) is legal, since the occurrences open upper bound is interpreted to mean the maximum value allowed by the cardinality upper bound.
    - the sum of all child object occurrences lower bounds must be less than the cardinality upper bound;
- no 'orphans': at least instance of one optional child object (occurrences lower bound = 0), and one instance of every mandatory child object (occurrences lower bound > 0) must be includable within the cardinality range.

4.3.5 "Any" Constraints

There are two cases where it is useful to state a completely open, or 'any', constraint. The first is when it is desired to override the existence or cardinality of a property, such as in the following:

```
PERSON[id2] matches {
    name existence matches {1}
    -- etc --
}
```

In the above, no further 'matches {}' part is required in the statement, since no more constraints are to be stated.

The second use of "any" as a constraint value is for types, such as in the following:

```
ELEMENT[id4] matches {          -- speed limit
    value matches {
        QUANTITY[id5]           -- type was 'DATA_VALUE' in RM
    }
}
```

The meaning of this constraint is that in the data at runtime, the *value* property of `ELEMENT` must be of type `QUANTITY`, but can have any value internally. This is most useful for constraining objects to be of a certain type, without further constraining value, and is especially useful where the information model contains subtyping, and there is a need to restrict data to be of certain subtypes in certain contexts.

Deprecated: In ADL 1.4 and transitional ADL 1.5, 'any' constraints were represented with an additional "matches {*}" at the end of the statement. This is deprecated. It is recommended that parsers silently accept this form, but output the modern ADL 1.5 form.

4.3.6 Reference Model Type Matching

All cADL object constraints state a type from an underlying reference model. This may be an abstract type or a concrete type. The part of the data conforming to the constraint can be of any concrete type from the reference model that conforms to the type mentioned in the constraint, i.e. the same type if it is concrete, or any subtype. Correctly evaluating data/archetype conformance is up to tools to implement, and requires access to a formal description of the reference model.

One of the consequences of subtype-based type matching is that semantics are needed for when more than one reference model subtype is declared under the same attribute node in cADL. Consider the reference model inheritance structure shown below, in which the abstract PARTY class has abstract and concrete descendants including ACTOR, ROLE, and so on.



FIGURE 5 Reference model with abstract and concrete subtypes

4.3.6.1 Narrowed Subtype Constraints

The following cADL statement defines an instance space that includes any instance of any of the concrete subtypes of the PARTY class within an instance of the class XXXX in the figure.

```
counter_party ∈ {
     PARTY[id4] ∈ { ... }
}
```

However, in some circumstances, it may be desirable to define a constraint that will match a particular subtype in a specific way, while other subtypes are matched by the more general rule. Under a single-valued attribute, this can be done as follows:

```
counter_party ∈ {
     PARTY[id4] ∈ { ... }
     PERSON[id5] ∈ {
        date_of_birth ∈ { ... }
     }
}
```

This cADL text says that the instance value of the *counter_party* attribute in the data can either be a PERSON object matching the PERSON block, with a *date_of_birth* matching the given range, or else any other kind of PARTY object.

Under a multiply-valued attribute, the alternative subtypes are included as identified child members. The following example illustrates a constraint on the *counter_parties* attribute of instances of the class YYYY in FIGURE 5.

```
counter_parties cardinality ∈ {*} ∈ {
     PERSON[id4] ∈ {
        date_of_birth ∈ { ... }
     }
     ORGANISATION[id5] ∈ {
        date_of_registration ∈ { ... }
     }
     PARTY[id6] ∈ { ... }
```

```
    }
```

The above says that ORGANISATION and PERSON instances in the data can only match the ORGANISATION and PERSON constraints stated above, while an instance any other subtype of PARTY must match the PARTY constraint.

4.3.6.2 Remove Specified Subtypes

In some cases it is required to remove some subtypes altogether. This is achieved by stating a constraint on the specific subtypes with occurrences limited to zero. The following example matches any PARTY instance with the exception of instances of COMPANY or GROUP subtypes.

```
counter_party ∈ {
    PARTY[id4] ∈ { ... }
    COMPANY[id5] occurrences ∈ {0}
    GROUP[id6] occurrences ∈ {0}
}
```

4.3.7 Paths

4.3.7.1 Archetype Path Formation

The use of object nodes allows the formation of *archetype paths*, which can be used to unambiguously reference object nodes within the same archetype or within a specialised child. The syntax of archetype paths is designed to be close to the W3C Xpath syntax, and can be directly converted to it for use in XML.

> **Archetype paths** are paths extracted from the definition section of an archetype, and refer to object nodes within the definition. A path is constructed as a concatenation of '/' characters and attribute names, with the latter including node identifiers where required for disambiguation.

In the following example, the PERSON constraint node is the sole object constraint under the single-valued attribute *manager*:

```
manager ∈ {
    PERSON[id104] ∈ {
        title ∈ {"head of finance", "head of engineering"}
    }
}
```

Two valid paths to the object under the *title* attribute are possible:

```
manager[id104]/title
manager/title
```

Where there are more than one sibling node, node identifiers must be used to ensure distinct paths:

```
employees cardinality ∈ {*} ∈ {
    PERSON[id104] ∈ {
        title ∈ {"head"}
    }
    PERSON[id105] matches {
        title ∈ {"member"}
    }
}
```

The paths to the respective *title* attributes are now:

```
employees[id104]/title
employees[id105]/title
```

The following gives another typical example:

```
HISTORY[id1] occurrences ∈ {1} ∈ {
    periodic ∈ {False}
    events cardinality ∈ {*} ∈ {
        EVENT[id2] occurrences ∈ {0..1} ∈ {}      -- 1 min sample
        EVENT[id3] occurrences ∈ {0..1} ∈ {}      -- 2 min sample
        EVENT[id4] occurrences ∈ {0..1} ∈ {}      -- 3 min sample
    }
}
```

The following paths can be constructed:

```
/                           -- the HISTORY (root) object
/periodic                   -- the HISTORY.periodic attribute
/events[id2]                -- the 1 minute event object
/events[id3]                -- the 2 minute event object
/events[id4]                -- the 3 minute event object
```

The above paths can all be used to reference the relevant nodes within the archetype in which they are defined, or within any specialised child archetype.

Paths used in cADL are expressed in the ADL path syntax, described in detail in section 7 on page 25. ADL paths have the same alternating object/attribute structure implied in the general hierarchical structure of cADL, obeying the pattern `TYPE/attribute/TYPE/attribute/`... .

The examples above are *physical* paths because they refer to object nodes using node identifier codes such as "id4". Physical paths can be converted to *logical* paths by adding the code meanings as annotations for node identifiers, if defined. Thus, the following two paths might be equivalent:

```
/events[id4]                      -- the 3 minute event object
/events[id4|3 minute event|]      -- the 3 minute event object
```

4.3.7.2 External Use of Paths

None of the paths shown above are valid outside the cADL text in which they occur, since they do not include an identifier of the enclosing document, normally an archetype. To reference a cADL node in an archetype from elsewhere (e.g. another archetype or a template), that the identifier of the document itself must be prefixed to the path, as in the following example:

```
[openehr-ehr-entry.apgar-result.v1]/events[id2]
```

This kind of path expression is necessary to form the paths that occur when archetypes are composed to form larger structures.

4.3.7.3 Runtime Paths

Paths for use with runtime data can be constructed in the same way as archetype paths, and are the same except for single-valued attributes. Since in data only a single instance can appear as the value of a single-valued attribute, there is never any ambiguity in referencing it, whereas an archetype path to or through the same attribute may require a node identifier due to he possible presence of multiple alternatives. Consider the example from above:

```
items cardinality matches {*} matches
    ELEMENT[id4] matches {                  -- speed limit
        value matches {
            QUANTITY[id22] matches {        -- miles per hour
                magnitude matches {|0..55|}
                property matches {"velocity"}
                units matches {"mph"}
            }
            QUANTITY[id23] matches {        -- km per hour
                magnitude matches {|0..100|}
                property matches {"velocity"}
```

```
                    units matches {"km/h"}
                }
            }
        }
    }
```

The following archetype paths can be constructed:

```
items[id4]/value[id22]
items[id4]/value[id23]
```

For instance data created according to this archetype, the following runtime path can be used:

```
items[id4]/value       -- since there is only one QUANTITY in the data
```

A query using this path will match the data regardless of which type of QUANTITY object is there. However, in some circumstances, queries may need to be specific, in which case they will use the full archetype path, i.e. `items[id4]/value[id22]` or `items[id4]/value[id23]` to select only 'miles' or 'kilometres' data. This will only work if the node ids (at-codes) are in fact stored in all types of the reference model data. If for example this was not the case with the QUANTITY type, another facet of the QUANTITY objects from the archetype such as 'units = "km/h"' would need to be used in the query to correctly locate only metric QUANTITY objects.

4.3.8 Internal References (Proxy Constraint Objects)

It is possible to define a constraint structure at a certain point to be the same as a structure defined elsewhere in the archetype, rather than copying the desired structure. This is achieved using a *proxy constraint object*, using the following syntax:

```
use_node TYPE[atN] archetype_path
```

This statement defines a node of type TYPE, whose definition is the same as the one found at path archetype_path. The type mentioned in the use_node reference must always be the same type as the referenced type.

The path must not be in the parent path of the proxy object itself, but may be a sibling of the proxy object. The sibling case is a special case, and the meaning of the proxy constraint is that the target object's children should be re-used, but not the target itself (since that would illegally create two siblings with the same identifier). The general case is that the proxy object and target object locations are different, and the meaning is that the proxy object is logically replaced by a deep copy of the target object. (In theory the sibling case could be banned, and proxies defined one level further down with targets of the children of the originally intended target, but this creates inconvenience for the archetype author, and can easily be dealt with in tools).

Occurrences from the target are also assumed, or may be explicitly overridden:

```
use_node TYPE[id4] occurrences ∈ {occ} archetype_path
```

Proxy objects provide an internal reuse mechanism. Specialised archetypes may redefine structures on such nodes as if they had been defined inline. This is described in more detail in Internal Reference (Proxy Object) Redefinition on page 98 in the Specialisation section.

A **proxy constraint object** allows object constraints defined elsewhere to be re-used within the same archetype or a specialised child.

The following example shows the definitions of the ADDRESS nodes for phone, fax and email for a home CONTACT being reused for a work CONTACT.

```
PERSON[id1] ∈ {
    identities ∈ {
```

```
        -- etc --
    }
    contacts cardinality ∈ {0..*} ∈ {
        CONTACT[id2] ∈ {                              -- home address
            purpose ∈ {-- etc --}
            addresses ∈ {-- etc --}
        }
        CONTACT[id3] ∈ {                              -- postal address
            purpose ∈ {-- etc --}
            addresses ∈ {-- etc --}
        }
        CONTACT[id4] ∈ {                              -- home contact
            purpose ∈ {-- etc --}
            addresses cardinality ∈ {0..*} ∈ {
                ADDRESS[id5] ∈ {                      -- phone
                    type ∈ {-- etc --}
                    details ∈ {-- etc --}
                }
                ADDRESS[id6] ∈ {                      -- fax
                    type ∈ {-- etc --}
                    details ∈ {-- etc --}
                }
                ADDRESS[id7] ∈ {                      -- email
                    type ∈ {-- etc --}
                    details ∈ {-- etc --}
                }
            }
        }
        CONTACT[id8] ∈ {                              -- work contact
            purpose ∈ {-- etc --}
            addresses cardinality ∈ {0..*} ∈ {
                use_node ADDRESS[id9] /contacts[id4]/addresses[id5] -- phone
                use_node ADDRESS[id10] /contacts[id4]/addresses[id6] -- fax
                use_node ADDRESS[id11] /contacts[id4]/addresses[id7] -- email
            }
        }
    }
```

The following example shows the occurrences being overridden in the referring node, to enable the specification for 'phone' to be re-used, but with a different occurrences constraint.

```
PERSON[id1] ∈ {
    contacts cardinality ∈ {0..*} ∈ {
        CONTACT[id4] ∈ {                              -- home contact
            addresses cardinality ∈ {0..*} ∈ {
                ADDRESS[id5] occurrences ∈ {1} ∈ {...}   -- phone
            }
        }
        CONTACT[id8] ∈ {                              -- work contact
            addresses cardinality ∈ {0..*} ∈ {
                use_node ADDRESS[id9] occurrences ∈ {0..*}
                    /contacts[id4]/addresses[id5]            -- phone
            }
        }
    }
```

4.3.8.1 Paths and Proxy Objects

In forming paths through the proxy and to nodes below the target, two cases can be identified:

- if the proxy object is a sibling of the target object, the proxy object node identifier is used in paths, and the node id of the target object is not;
- otherwise, paths are formed using the identifier from the proxy target object.

4.3.9 External References

Another kind of reference in an archetype is to another archetype. There are two ways this can be done: using a direct reference, and using an 'archetype slot'. The first is used when the need is to refer to one specific archetype (or to a template from another template), while the second is a constraint that allows for various archetypes matching some criterion to be used. The slot concept is described in the next section.

> An **external reference** defines a fixed compositional connection between two archetypes.

Direct references, or 'external references' as they will be denoted here occur for two main reasons: re-use and templating. In the first case, an archetype has originally been built using inline constraints when it is discovered that another archetype contains the same or very similar inline constraints at a similar point. As would be normal in software design, a refactoring exercise is conducted that results in the common part being created as its own, new archetype, and both original archetypes 'referring' to it. They do this using an external reference, which has syntax of the form:

```
use_archetype TYPE[idN] [occurrences constraint] archetype_id
```

In the above, the `[archetype_id]` replaces the usual archetype node identifier (at-code). The usual occurrence constraints can be optionally applied at the end.

The following example shows sections of two parent archetypes both referring to the same child archetype. The first section is from an *open*EHR `INSTRUCTION` archetype to do with a medication order.

```
INSTRUCTION[id1] ∈ {-- Medication order
    activities cardinality ∈ {0..*; unordered} matches {
        ACTIVITY[id2] ∈ {-- Medication activity
            action_archetype_id ∈ {/openEHR-EHR-ACTION\.medication\.v1/}
            description ∈ {
                use_archetype ITEM_TREE[id3]
                                        openEHR-EHR-ITEM_TREE.medication.v1
            }
        }
    }
}
```

This section is from an *open*EHR `ACTION` archetype defining medication administration actions.

```
ACTION[id1] ∈ {-- Medication action
    ism_transition ∈ {
        ISM_TRANSITION[id2] ∈ {...}
        ...
    }
    description ∈ {
        use_archetype ITEM_TREE[id5] openEHR-EHR-ITEM_TREE.medication.v1
    }
}
```

Each of these archetypes refers to the *open*EHR `ITEM_TREE` archetype `openEHR-EHR-ITEM_TREE.medication.v1`, which is a normal archetype describing medication.

Following the standard object-oriented semantics of type substitutability, and also the ontological subsumption notion, specialisations of the referenced archetype (including templates) are also valid substitutions at design or runtime.

TBD_2: it might be in some circumstances useful to prevent this; the typical OO approach would be to introduce a 'frozen' keyword to prevent dynamic specialisation substitution.

External references can of course also be defined under container attributes.

The second use of external references is typically in templates, to specify an archetype or sub-template of a template for an attribute where no slot has been defined. This use is described in the *open*EHR Template specification.

Paths through nodes specified as external references will include the archetype identifier(s) rather than just at-codes. The following paths are from the `INSTRUCTION` and `ACTION` examples above, respectively.

```
/activities[id2]/description[openEHR-EHR-ITEM_TREE.medication.v1]/...

/description[openEHR-EHR-ITEM_TREE.medication.v1]/...
```

4.3.10 Archetype Slots

At any point in a cADL definition, a constraint can be defined that allows other archetypes to be used, rather than defining the desired constraints inline. This is known as an archetype 'slot', i.e. a connection point whose allowable 'fillers' are constrained by a set of statements, written in the ADL assertion language (defined in section 5 on page 23).

> An **archetype slot** defines a constrained compositional chaining point in an archetype at which other archetypes can be inserted, if they match the constraint defined by the slot.

An archetype slot is introduced with the keyword `allow_archetype` and defined in terms of two lists of assertion statements defining which archetypes are allowed and/or which are excluded from filling that slot, introduced with the keywords `include` and `exclude`, respectively. The following example illustrates the general form of an archetype slot.

```
allow_archetype SECTION[id5] occurrences ∈ {0..*} ∈ {
    include
        -- constraints for inclusion
    exclude
        -- constraints for exclusion
}
```

Since archetype slots are typed (like all other object node types), the reference model type of the allowed archetypes is already constrained. Otherwise, any assertion about a filler archetype can be made within one or other of the `include` and `exclude` constraints. The `include` and `exclude` constraints are used in a mutually exclusive fashion, i.e. only one or other contains a 'substantive' constraint on archetypes to be included or excluded. On its own, the constraint is taken as a *recommendation*, unless the other list contains a constraint matching 'all archetypes', in which case the constraint is definitive. See Formal Semantics of include and exclude Constraints below for details.

The assertions do not constrain data in the way that other archetype statements do, instead they constrain the identities of archetypes, either directly or indirectly. Two kinds of reference may be used in a slot assertion. The first is a reference to an object-oriented property of the filler archetype itself, where the property names are defined by the ARCHETYPE class in the Archetype Object Model. Examples include:

```
archetype_id
parent_archetype_id
short_concept_name
```

This kind of reference is usually used to constrain the allowable archetypes based on *archetype_id* or some other meta-data item (e.g. archetypes written in the same organisation). The second kind of reference is to absolute archetype paths in the definition section of the filler archetype. Both kinds of reference take the form of an Xpath-style path, with the distinction that paths referring to ARCHETYPE attributes *not* in the definition section do not start with a slash (this allows parsers to easily distinguish the two types of reference).

The actual specification of slot fillers, and also the ‘closing’ of slots is done in specialised archetypes, and is described in Slot Redefinition on page 99, in the chapter on specialisation.

4.3.10.1 Slots based on Archetype Identifiers

A basic kind of assertion is on the identifier of archetypes allowed in the slot. This is achieved with statements like the following in the include and exclude lists:

```
archetype_id/value ∈ {/openEHR-EHR-\.SECTION\..*\..*/}
```

It is possible to limit valid slot-fillers to a single archetype simply by stating a full archetype identifier with no wildcards; this has the effect that the choice of archetype in that slot is predetermined by the archetype and cannot be changed later. In general, however, the intention of archetypes is to provide highly re-usable models of real world content with local constraining left to templates, in which case a ‘wide’ slot definition is used (i.e. matches many possible archetypes).

The following example shows how the “Objective” SECTION in a problem/SOAP headings archetype defines two slots, indicating which OBSERVATION and SECTION archetypes are allowed and excluded under the *items* property.

```
SECTION [at2000] occurrences ∈ {0..1} ∈ {         -- objective
    items cardinality ∈ {0..*} ∈ {
        allow_archetype SECTION[at2002] occurrences ∈ {0..*} ∈ {
            include
                archetype_id/value ∈ {/openEHR-EHR-SECTION\..+\..+/}
            exclude
                archetype_id/value ∈
                            {/openEHR-EHR-SECTION\.patient_details\..+/}
        }
    }
}
```

Here, every constraint inside the block starting on an allow_archetype line contains constraints that must be met by archetypes in order to fill the slot. In the examples above, the constraints are in the form of regular expressions on archetype identifiers. In cADL, the PERL regular expression syntax is assumed.

There are two ways in which *archetype_id* regular expressions patterns can be used:

- as a pattern against which to test a particular archetype identifier being proposed for that slot;
- as a pattern to use against a population of archetypes (e.g. all archetypes in a particular repository) in order to generate a list of all possible archetypes for filling the slot.

Due to the second use, it is required that the regular expression pattern always cover a full archetype identifier rather than only sub-parts. As a consequence, a 'meta-pattern' can be defined to check *archetype_id* regular expressions for validity:

- ^.+-.+-.+\..*\..+$

Because identifier matching is an inherently lexical operation, subtypes of mentioned types are not matched unless explicitly stated. Consider the following example:

```
allow_archetype ENTRY[id2] ∈ {-- any kind of ENTRY
    include
        archetype_id/value ∈ {/openEHR-EHR-ENTRY..+\.v1/}
}
```

The intention is to allow any kind of `ENTRY`, but the above constraint won't have the desired effect, because the pattern `openEHR-EHR-ENTRY` is unlikely to match any actual archetypes. Instead the following kind of constraint should be used:

```
allow_archetype ENTRY[id2] ∈ {-- any kind of ENTRY
    include
        archetype_id/value ∈ {
            /openEHR-EHR-EVALUATION\..+\.v1|openEHR-EHR-OBSERVATION\..+\.v1/
        }
}
```

The above would allow any `EVALUATION` and any `OBSERVATION` archetypes to be used in the slot.

4.3.10.2 Slots based on other Constraints

Other constraints are possible as well, including that the allowed archetype must contain a certain keyword, or a certain path. The latter allows archetypes to be linked together on the basis of content. For example, under a "genetic relatives" heading in a Family History Organiser archetype, the following slot constraint might be used:

```
allow_archetype EVALUATION occurrences ∈ {0..*} matches {
    include
        short_concept_name ∈ {"risk_family_history"}
            ∧ ∃ /subject/relationship/defining_code →
                ~/subject/relationship/defining_code/
                    code_list.has([openehr::0]) -- "self"
}
```

This says that the slot allows archetypes on the `EVALUATION` class, which either have as their concept "risk_family_history" or, if there is a constraint on the subject relationship, then it may not include the code `[openehr::0]` (the *open*EHR term for "self") - i.e. it must be an archetype designed for family members rather than the subject of care herself.

4.3.10.3 Formal Semantics of include and exclude Constraints

The semantics of the `include` and `exclude` lists are somewhat subtle. They are as follows:

- The meaning of the 'set of all archetypes' in any given environment is evaluatable (and evaluated) to a finite set consisting of all archetypes *actually* available within that environment, known as the *current archetype set*, not some notional virtual / global set of archetypes, or theoretically possible set.
- Either the `include` or `exclude` constraint, but not both, may be 'substantive', i.e. define a particular set of archetypes that would be matched within a given slot, or 'open', i.e. matching all possible archetypes.
- A slot constraint may consist of a single `include` or `exclude` constraint, or of an `include` / `exclude` pair.

- If an `include` or `exclude` constraint is present on its own, it is understood as a *recommendation*, i.e. it does not constitute a formal constraint for matching or exclusion, but tools and applications may use the recommended match set in an intelligent way. The result set for such an `include` or `exclude` is the whole current archetype set.
- If a substantive `include` or `exclude` constraint is present with a corresponding open `exclude` or `include`, respectively, the substantive constraint is considered formally binding.

The following examples will make this clearer. The first is a slot definition with a substantive `include` constraint which matches archetypes whose identifiers look like `openEHR-EHR-CLUSTER.procedure*.*`, where the '*' characters mean 'anything'. The meaning of the slot overall is that archetypes matching this constraint are recommended, but in fact any CLUSTER archetype is allowed. A smart application might use this to present certain choices to the user in an efficient way.

```
allow_archetype CLUSTER[id2] ∈ {
    include
        archetype_id/value ∈ {/openEHR-EHR-CLUSTER.\.procedure.*\..*/}
}
```

The second example shows the same `include` constraint, and adds the 'any' exclude constraint:

```
allow_archetype CLUSTER[id2] ∈ {
    include
        archetype_id/value ∈ {/openEHR-EHR-CLUSTER.\.procedure.*\..*/}
    exclude
        archetype_id/value ∈ {/.*/}
}
```

The meaning of the slot constraint overall is that only archetypes matching the `include` constraint are allowed, and no others. The same logic applies in the reverse sense when the `exclude` constraint is substantive.

NB: This approach is historical, and is clearly deficient, since it relies on lexical matching. A new kind of slot is proposed that uses a simple grammar consisting of subsumption and logical operators ('<', '<<', and, or etc), which would enable expressions like the following:

```
allow_archetype CLUSTER[id5.1] occurrences ∈ {0..1} ∈ {
    include ∈ {True}
        archetype_id ∈ {
            ARCHETYPE_ID ∈ {
                ...
                concept ∈ {<< investigation_methodology or
                              << investigation_protocol}
                ...
            }
        }
}
```

In the above, the concept part of the archetype name '**`investigation_methodology`**' is used as a concept identifier, in much the same way as a SNOMED CT code. In the final proposal, the latter might be used directly rather than archetype concept identifiers.

4.3.11 Mixed Structures

Four types of structure representing constraints on complex objects have been presented so far:

- *complex object structures*: any node introduced by a type name and followed by {} containing constraints on attributes;

- *internal references*: any node introduced by the keyword `use_node`, followed by a type name; such nodes indicate re-use of a complex object constraint that has already been expressed elsewhere in the archetype;
- *archetype slots*: any node introduced by the keyword `allow_archetype`, followed by a type name; such nodes indicate a complex object constraint which is expressed in some other archetype;
- *placeholder constraints*: any node whose constraint is of the form `[acNNNN]`.

At any given node, any combination of these types can co-exist, as in the following example:

```
SECTION[id2000] ∈ {
    items cardinality ∈ {0..*; ordered} ∈ {
        ENTRY[id2001] ∈ {-- etc --}
        allow_archetype ENTRY[id2] ∈ {-- etc --}
        use_node ENTRY [id2]/some_path[id4]/
        ENTRY[id2003] ∈ {-- etc --}
        use_node ENTRY /[id1002]/some_path[id1012]/
        use_node ENTRY /[id1005]/some_path[id1052]/
        ENTRY[id2004] ∈ {-- etc --}
    }
}
```

Here we have a constraint on an attribute called *items* (of cardinality 0..*), expressed as a series of possible constraints on objects of type `ENTRY`. The 1st, 4th and 7th are described 'in place'; the 3rd, 5th and 6th are expressed in terms of internal references to other nodes earlier in the archetype, while the 2nd is an archetype slot, whose constraints are expressed in other archetypes matching the include/exclude constraints appearing between the braces of this node. Note also that the `ordered` keyword has been used to indicate that the list order is intended to be significant.

4.4 Second-order Constraints

4.4.1 Tuple Constraints

In realistic data, it is not uncommon to need to constrain object properties in a covarying way. A simple example is the need to state range constraints on a temperature, represented as a Quantity type, for both Centigrade and Fahrenheit scales. The default way to do this in ADL is (assuming a simple `QUANTITY` class consisting of *property*, *units* and *magnitude* properties):

```
value ∈ {
    QUANTITY [id14] ∈ {
        property ∈ {[openehr::151|temperature|]}
        units ∈ {"deg F"}
        magnitude ∈ {|32.0..212.0|}
    }
    QUANTITY [id15] ∈ {
        property ∈ {[openehr::151|temperature|]}
        units ∈ {"deg C"}
        magnitude ∈ {|0.0..100.0|}
    }
}
```

What we logically want to do is to state a single constraint on a `QUANTITY` that sets the *magnitude* range constraint dependent on the *units* constraint. Note that we are forced to include at-codes for the two Quantity nodes, to satisfy the path uniqueness rule.

The covarying requirement could be met using rules of the form:

```
.../value/units = "deg F" → magnitude ∈ {|32.0..212.0|}
.../value/units = "deg C" → magnitude ∈ {|0.0..100.0|}
```

However, this seems obscure for what is logically a very simple kind of constraint.

> **NB:** In the *open*EHR ADL 1.4 Archetype Profile, a custom constrainer type `C_DV_QUANTITY` was used to to provide the above constraint. However, this is specific to the Reference Model type, and does not solve similar constraints occurring in other types. This type and also the `C_DV_ORDINAL` type have been removed from ADL 1.5 altogether.

A generic solution involves treating covarying properties formally as tuples, and providing syntax to express 'constraints on tuples'. The following syntax achieves this:

```
value ∈ {
    QUANTITY[id4] ∈ {
        property ∈ {[openehr::151]}   -- temperature
        [units, magnitude] ∈ {
            [{"deg F"}, {|32.0..212.0|}] ,
            [{"deg C"}, {|0.0..100.0|}]
        }
    }
}
```

The above is actually short-hand for the following structure, with the added constraint that only corresponding units and magnitude leaf level constraints can occur together, while other combinations like `"deg F"` and `|0.0..100.0|` would be illegal:

```
value ∈ {
    QUANTITY ∈ {
        property ∈ {[openehr::151]} -- temperature
        units ∈ {
            String ∈ {"deg F"}
            String ∈ {"deg C"}
        }
        magnitude ∈ {
            Integer ∈ {|32.0..212.0|}
            Integer ∈ {|0.0..100.0|}
        }
    }
}
```

TBD_3: need to check path semantics to leaf nodes in tuples.

In the above, the {} surrounding each leaf level constraint are needed because although such constraints are typically atomic, as above, they may also take other standard ADL forms such as a list of strings, list of integers etc. In the latter case, the ',' characters from such lists will be conflated with the ',' separator of the distinct constraints in the tuple. Use of {} is also logically justified: each such entity is indeed a 'constraint' in the ADL sense, and all constraints are delimited by {}.

The above defines constraints on units and magnitude together, as tuples like `[{"deg F"}, {|32.0..212.0|}]` .

This same syntax will work for tuples of 3 or more co-varying properties, and is mathematically clean. It does involve some extra work for compiler implementers, but this only needs to be performed once to support any use of tuple constraints, regardless of Reference Model type.

If we look at the `ORDINAL` data type constraint in the same light. First, doing a typical ordinal constraint (a scale of +, ++, +++) with just standard ADL:

```
ordinal_attr ∈ {
    ORDINAL[id4] ∈ {
        value ∈ {0}
        symbol ∈ {
            CODED_TEXT ∈ {
                terminology_id ∈ {"local"}
                code ∈ {"at1"}                    -- +
            }
        }
    }
    ORDINAL[id5] ∈ {
        value ∈ {1}
        symbol ∈ {
            CODED_TEXT ∈ {
                terminology_id ∈ {"local"}
                code ∈ {"at2"}                    -- ++
                    }
                }
            }
        }
    }
    ORDINAL[id6] ∈ {
        value ∈ {2}
        symbol ∈ {
            CODED_TEXT ∈ {
                terminology_id ∈ {"local"}
                code ∈ {"at3"}     -- +++
            }
        }
    }
}
```

This hides the ORDINAL type altogether, but as for the C_DV_QUANTITY example above, it was a custom solution.

By the use of tuple constraint, almost the same thing can be chieved much more efficiently. We can now write:

```
ordinal_attr ∈ {
    ORDINAL[id4] ∈ {
        [value ,symbol] ∈ {
            [{0}, {[local::at1]}], -- +
            [{1}, {[local::at2]}], -- ++
            [{2}, {[local::at3]}] -- +++
        }
    }
}
```

Deprecated: in the *open*EHR profiled version of ADL 1.4, a custom syntax was used, which is not only smaller, but removes the need for the three at-codes above:

```
ordinal_attr ∈ {
    0|[local::at1],          -- +
    1|[local::at2],          -- ++
    2|[local::at3]           -- +++
}
```

4.4.2 Group Constraints

Within a container attribute, any number of object constraints may be defined. The cardinality and occurrences constraints described above show how to control respectively, the overall container contents, and the occurrence of any particular object constraint within data. However, sometimes finer control is needed on repetition and grouping of members within the container. This can be achieved by the `group` construct, which provides an interior block where a sub-group of the overall container can be treated as a sub-group. The following example shows a typical used of the group construct.

```
ITEM_TREE[id1] ∈ {
    items matches {
      ELEMENT[id2] occurrences ∈ {1} ∈ {...}          -- Investigation type
      ELEMENT[id3] occurrences ∈ {0..1} ∈ {...}       -- reason

      group cardinality ∈ {1} occurrences ∈ {0..1} ∈ {
                                                            -- Methodology
         ELEMENT[id6] occurrences ∈ {0..1} ∈ {...}          -- as Text
         ELEMENT[id7] occurrences ∈ {0..1} ∈ {...}          -- Coded
         CLUSTER[id8] occurrences ∈ {0..1} ∈ {...}          -- structured
      }
      ELEMENT[id11] occurrences ∈ {0..1} ∈ {...}      -- (other details)
      CLUSTER[id12] occurrences ∈ {0..1} ∈ {...}      -- (other details)
    }
}
```

In the above, the group is used to state a logical choice of methodology representations, each defined by one of the three constraints within the group. The group construct includes both cardinality and occurrences qualifier constraints. The former indicates the size and ordering of the group, in the same way as the cardinality constraint does for the overall contents of a container attribute. The latter defines the repeatability of the group. If the group occurrences upper limit is above 1, it means that the sub group may repeat, with each repetition respecting the order and size defined by the group cardinality.

> A **group constraint** may be used to delimit a sub-list of objects within the total list of object constraints defined within a container attribute. A cardinality, defining size, ordering and uniqueness of the sub-list must be defined. An occurrences defining the repeatbility of the sub-list must also be defined. Group constraints can be nested.

The use of group cardinality and occurrences constraints, coupled with the occurrences constraints on each group member provide a means of specifying a number of logical constraint types found in other formalisms, including XML, as follows.

Logical constraint	Group cardinality	Group occurrences	Item occurrences
1 of N choice	1..1	upper = 1	0..1
1 of N choice, repeating	1..1	upper > 1	0..1
N of M choice	N..N	upper = 1	0..1
N of M choice, repeating	N..N	upper > 1	0..1
sequence, repeating	upper > 1, ordered	upper > 1	any
sub-group, repeating	upper > 1, unordered	upper > 1	any

Group blocks can be nested, enabling sub-lists of sub-lists to be defined, as illustrated below.

```
items ∈ {
    ELEMENT[id2] occurrences ∈ {1} ∈ {...}          -- Investigation type
    ELEMENT[id3] occurrences ∈ {0..1} ∈ {...}       -- Investigation reason

    group cardinality ∈ {2} occurrences ∈ {*} ∈ {
                                                    -- pick any 2 & repeat
       ELEMENT[id6] occurrences matches {0..1} ∈ {...}
       ELEMENT[id7] occurrences matches {0..1} ∈ {...}
       CLUSTER[id8] occurrences matches {0..1} ∈ {...}

       group cardinality ∈ {1} occurrences ∈ {0..1} ∈ {
                                                    -- at least one
          ELEMENT[id9] occurrences ∈ {0..1} ∈ {...}
          CLUSTER[id10] occurrences ∈ {0..1} ∈ {...}
       }
    }

    ELEMENT[id11] occurrences ∈ {0..1} ∈ {...}     -- (other details)
    CLUSTER[id12] occurrences ∈ {0..1} ∈ {...}     -- (other details)
}
```

4.4.2.1 Slots and Grouping

The group constraint is often useful with a slot definition, in order to control the ordering and occurrences of items defined by other archetypes, within an overall container. Consider the example of data of the general structure: 'any number of problem and diagnosis Entries, followed by one or more plan & treatment Entries'. An example of data following this structure would be:

- EVALUATION: problem #1
- EVALUATION: diagnosis #1
- EVALUATION: problem #2
- EVALUATION: problem #3
- EVALUATION: plan
- INSTRUCTION: medication #1
- INSTRUCTION: therapy #1

It might be expected that the slot constraints needed to define this are as follows:

```
SECTION[id2] occurrences ∈ {0..1} ∈ {     -- Subjective
    items cardinality ∈ {0..*; ordered} ∈ {
       allow_archetype EVALUATION[id6] occurrences ∈ {*} ∈ { -- Problem
          include
             archetype_id/value ∈ {/openEHR-EHR-EVALUATION\.problem\.v*/}
       }
       allow_archetype EVALUATION[id7] occurrences ∈ {*} ∈ { -- Diagnosis
          include
             archetype_id/value ∈ {/openEHR-EHR-EVALUATION\.problem-
diagnosis\.v*/}
       }
       allow_archetype EVALUATION[id8] occurrences ∈ {1} ∈ { -- Plan
          include
             archetype_id/value ∈ {/openEHR-EHR-EVALUATION\.plan\.v*/}
       }
       allow_archetype INSTRUCTION[id9] occurrences ∈ {*} ∈ {
                                                            -- Intervention
          include
             archetype_id/value ∈ {/openEHR-EHR-INSTRUCTION\.plan\.v*/}
```

```
            }
        }
    }
```

The above says that the SECTION.*items* attribute is an ordered list, and that its contents include multiple EVALUATION objects representing problem, diagnosis and plan, and also multiple INSTRUCTION objects representing interventions. The problem is now apparent. Each slot definition is set of possibilities, but we do not necessarily want to follow the slot ordering for the ordering of the archetypes chosen to fill the slots. To impose the required ordering and occurrences, we can use the group construct as follows.

```
SECTION[id2] occurrences ∈ {0..1} ∈ {    -- Subjective
    items cardinality ∈ {0..*; ordered} ∈ {
       group cardinality ∈ {0..1} occurrences ∈ {0..*} ∈ {
                            -- sub-group of any number of problems & diagnoses
          allow_archetype EVALUATION[id6] occurrences ∈ {1} ∈ {--Problem
             include
               archetype_id/value ∈ {/openEHR-EHR-EVALUATION\.problem\.v*/}
          }
          allow_archetype EVALUATION[id7] occurrences ∈ {1} ∈ {--Diagnosi
             include
               archetype_id/value ∈
                               {/openEHR-EHR-EVALUATION\.diagnosis\.v*/}
          }
       }
       allow_archetype EVALUATION[id8] occurrences ∈ {1} ∈ { -- Plan
          include
             archetype_id/value ∈ {/openEHR-EHR-EVALUATION\.plan\.v*/}
       }
       allow_archetype INSTRUCTION[id9] occurrences ∈ {*} ∈ {
                                                         -- Intervention
          include
             archetype_id/value ∈ {/openEHR-EHR-INSTRUCTION\.plan\.v*/}
       }
    }
}
```

The above has the desired result in data: a group of any number of problems and diagnoses, followed by a plan, followed by one or more Interventions.

4.5 Constraints on Primitive Types

At the leaf nodes in a cADL text, constraints can be expressed on the following primitive types:

- Boolean;
- Character, String;
- Integer, Real;
- Date, Time, Date_time, Duration;
- lists and intervals of some of the above.

While constraints on complex types follow the rules described so far, constraints on attributes of primitive types in cADL are expressed without type names, and omitting one level of braces, as follows:

```
some_attr matches {some_pattern}
```

rather than:

```
some_attr matches {
    PRIMITIVE_TYPE matches {
        some_pattern
    }
}
```

This is made possible because the syntax patterns of all primitive type constraints are mutually distinguishable, i.e. the type can always be inferred from the syntax alone. Since all leaf attributes of all object models are of primitive types, or lists or sets of them, cADL archetypes using the brief form for primitive types are significantly less verbose overall, as well as being more directly comprehensible to human readers. Currently the cADL grammar **only supports the brief form** used in this specification since no practical reason has been identified for supporting the more verbose version. Theoretically however, there is nothing to prevent it being used in the future, or in some specialist application.

4.5.1 Constraints on String

Strings can be constrained in two ways: using a list of fixed strings, and using using a regular expression. All constraints on strings are case-sensitive.

4.5.1.1 List of Strings

A String-valued attribute can be constrained by a list of strings (using the ODIN syntax for string lists), including the simple case of a single string. Examples are as follows:

```
species ∈ {"platypus"}
species ∈ {"platypus", "kangaroo"}
species ∈ {"platypus", "kangaroo", "wombat"}
```

The first example constraints the runtime value of the *species* attribute of some object to take the value "platypus"; the second constrains it be either "platypus" or "kangaroo", and so on. **In almost all cases, this kind of string constraint should be avoided**, since it usually renders the body of the archetype language-dependent. Exceptions are proper names (e.g. "NHS", "Apgar"), product tradenames (but note even these are typically different in different language locales, even if the different names are not literally translations of each other). The preferred way of constraining string attributes in a language independent way is with local [ac] codes. See Local Constraint Codes on page 28.

4.5.1.2 Regular Expression

The second way of constraining strings is with regular expressions, a widely used syntax for expressing patterns for matching strings. The regular expression syntax used in cADL is a proper subset of that used in the Perl language (see [18] for a full specification of the regular expression language of Perl). Three uses of it are accepted in cADL:

```
string_attr matches {/regular expression/}
string_attr matches {=~ /regular expression/}
string_attr matches {!~ /regular expression/}
```

The first two are identical, indicating that the attribute value must match the supplied regular expression. The last indicates that the value must *not* match the expression. If the delimiter character is required in the pattern, it must be quoted with the backslash ('\') character, or else alternative delimiters can be used, enabling more comprehensible patterns. A typical example is regular expressions including units. The following two patterns are equivalent:

```
units ∈ {/km\/h|mi\/h/}
units ∈ {^km/h|mi/h^}
```

The rules for including special characters within strings are described in File Encoding and Character Quoting on page 20.

The regular expression patterns supported in cADL are as follows.

Atomic Items

. match any single character. E.g. `/ ... /` matches any 3 characters which occur with a space before and after;

`[xyz]` match any of the characters in the set `xyz` (case sensitive). E.g. `/[0-9]/` matches any string containing a single decimal digit;

`[a-m]` match any of the characters in the set of characters formed by the continuous range from `a` to `m` (case sensitive). E.g. `/[0-9]/` matches any single character string containing a single decimal digit, /[S-Z]/ matches any single character in the range `S` - `Z`;

`[^a-m]` match any character except those in the set of characters formed by the continuous range from `a` to `m`. E.g. `/[^0-9]/` matches any single character string as long as it does not contain a single decimal digit;

Grouping

`(pattern)` parentheses are used to group items; any pattern appearing within parentheses is treated as an atomic item for the purposes of the occurrences operators. E.g. `/([0-9][0-9])/` matches any 2-digit number.

Occurrences

`*` match 0 or more of the preceding atomic item. E.g. `/.*/` matches any string; `/[a-z]*/` matches any non-empty lower-case alphabetic string;

`+` match 1 or more occurrences of the preceding atomic item. E.g. `/a.+/` matches any string starting with ‘a’, followed by at least one further character;

`?` match 0 or 1 occurrences of the preceding atomic item. E.g. `/ab?/` matches the strings `"a"` and `"ab"`;

`{m,n}` match m to n occurrences of the preceding atomic item. E.g. `/ab{1,3}/` matches the strings `"ab"` and `"abb"` and `"abbb"`; `/[a-z]{1,3}/` matches all lower-case alphabetic strings of one to three characters in length;

`{m,}` match at least m occurrences of the preceding atomic item;

`{,n}` match at most n occurrences of the preceding atomic item;

`{m}` match exactly m occurrences of the preceding atomic item;

Special Character Classes

`\d`, `\D` match a decimal digit character; match a non-digit character;

`\s`, `\S` match a whitespace character; match a non-whitespace character;

Alternatives

`pattern1|pattern2` match either pattern1 or pattern2. E.g. `/lying|sitting|standing/` matches any of the words `"lying"`, `"sitting"` and `"standing"`.

A similar warning should be noted for the use of regular expressions to constrain strings: they should be limited to non-linguistically dependent patterns, such as proper and scientific names. The use of regular expressions for constraints on normal words will render an archetype linguistically dependent, and potentially unusable by others.

4.5.2 Constraints on Integer

Integers can be constrained using a list of integer values, and using an integer interval.

4.5.2.1 List of Integers

Lists of integers expressed in the syntax from ODIN can be used as a constraint, e.g.:

```
length matches {1000}           -- fixed value of 1000
magnitude matches {0, 5, 8}     -- any of 0, 5 or 8
```

The first constraint requires the attribute length to be 1000, while the second limits the value of magnitude to be 0, 5, or 8 only. A list may contain a single integer only:

```
magnitude matches {0}           -- matches 0
```

4.5.2.2 Interval of Integer

Integer intervals are expressed using the interval syntax from ODIN (described in Intervals of Ordered Primitive Types on page 35). Examples of 2-sided intervals include:

```
length matches {|1000|}         -- point interval of 1000 (=fixed value)
length matches {|950..1050|}    -- allow 950 - 1050
length matches {|0..1000|}      -- allow 0 - 1000
length matches {|0..<1000|}     -- allow 0>= x <1000
length matches {|0>..<1000|}    -- allow 0> x <1000
length matches {|100+/-5|}      -- allow 100 +/- 5, i.e. 95 - 105
rate matches {|0..infinity|}    -- allow 0 - infinity, i.e. same as >= 0
```

Examples of one-sided intervals include:

```
length matches {|<10|}          -- allow up to 9
length matches {|>10|}          -- allow 11 or more
length matches {|<=10|}         -- allow up to 10
length matches {|>=10|}         -- allow 10 or more
```

4.5.3 Constraints on Real

Constraints on Real values follow exactly the same syntax as for Integers, in both list and interval forms. The only difference is that the real number values used in the constraints are indicated by the use of the decimal point and at least one succeeding digit, which may be 0. Typical examples are:

```
magnitude ∈ {5.5}                         -- list of one (fixed value)
magnitude ∈ {|5.5|}                       -- point interval (=fixed value)
magnitude ∈ {|5.5..6.0|}                  -- interval
magnitude ∈ {5.5, 6.0, 6.5}               -- list
magnitude ∈ {|0.0..<1000.0|}              -- allow 0>= x <1000.0
magnitude ∈ {|<10.0|}                     -- allow anything less than 10.0
magnitude ∈ {|>10.0|}                     -- allow greater than 10.0
magnitude ∈ {|<=10.0|}                    -- allow up to 10.0
magnitude ∈ {|>=10.0|}                    -- allow 10.0 or more
magnitude ∈ {|80.0+/-12.0|}               -- allow 80 +/- 12
```

4.5.4 Constraints on Boolean

Boolean runtime values can be constrained to be True, False, or either, as follows:

```
some_flag matches {True}
some_flag matches {False}
some_flag matches {True, False}
```

4.5.5 Constraints on Character

Characters can be constrained in two ways: using a list of characters, and using a regular expression.

4.5.5.1 List of Characters

The following examples show how a character value may be constrained using a list of fixed character values. Each character is enclosed in single quotes.

```
color_name matches {'r'}
color_name matches {'r', 'g', 'b'}
```

4.5.5.2 Regular Expression

Character values can also be constrained using single-character regular expression elements, also enclosed in single quotes, as per the following examples:

```
color_name matches {'[rgbcmyk]'}
color_name matches {'[^\s\t\n]'}
```

The only allowed elements of the regular expression syntax in character expressions are the following:

- any item from the Atomic Items list above;
- any item from the Special Character Classes list above;
- the '.' character, standing for "any character";
- an alternative expression whose parts are any item types, e.g. `'a'|'b'|[m-z]`

4.5.6 Constraints on Dates, Times and Durations

Dates, times, date/times and durations may all be constrained in three ways: using a list of values, using intervals, and using patterns. The first two ways allow values to be constrained to actual date, time etc values, while the last allows values to be constrained on the basis of which parts of the date, time etc are present or missing, regardless of value. The pattern method is described first, since patterns can also be used in lists and intervals.

> **NB:** for the date/time constraint type, parser writers should consider allowing the 'T' character to be optional on read but mandatory on save for some time. This is because previous versions of ADL did not include it, with the result that existing tools have created archetypes without the 'T' in date/time constraint patterns.

4.5.6.1 Date, Time and Date/Time

Patterns

Dates, times, and date/times (i.e. timestamps), can be constrained using patterns based on the ISO 8601 date/time syntax, which indicate which parts of the date or time must be supplied. A constraint pattern is formed from the abstract pattern `yyyy-mm-ddThh:mm:ss` (itself formed by translating each field of an ISO 8601 date/time into a letter representing its type), with either `'?'` (meaning optional) or `'X'` (not allowed) characters substituted in appropriate places. A simplified grammar of the pattern is as follows (EBNF; all tokens shown are literals):

```
date_constraint:           yyyy - mm|??|XX - dd|??|XX
time_constraint:           hh : mm|??|XX : ss|??|XX
time_in_date_constraint:   T hh|??|XX : mm|??|XX : ss|??|XX
date_time_constraint:      date_constraint time_in_date_constraint
```

All expressions generated by this grammar must also satisfy the validity rules:

- where '??' appears in a field, only '??' or 'XX' can appear in fields to the right
- where 'XX' appears in a field, only 'XX' can appear in fields to the right

A fuller grammar can be defined to implement both the simplifed grammar and validity rules.

The following table shows the valid patterns that can be used, and the types implied by each pattern.

Implied Type	Pattern	Explanation
Date	yyyy-mm-dd	full date must be specified

Implied Type	Pattern	Explanation
Date	`yyyy-mm-??`	`optional day;` `e.g. day in month forgotten`
Date	`yyyy-??-??`	`optional month, day;` `i.e. any date allowed; e.g. mental health questionnaires which include well known historical dates`
Date	`yyyy-mm-XX`	`mandatory month, no day`
Date	`yyyy-??-XX`	`optional month, no day`
Time	`hh:mm:ss`	`full time must be specified`
Time	`hh:mm:XX`	`no seconds;` `e.g. appointment time`
Time	`hh:??:XX`	`optional minutes, no seconds;` `e.g. normal clock times`
Time	`hh:??:??`	`optional minutes, seconds;` `i.e. any time allowed`
Date/Time	`yyyy-mm-ddThh:mm:ss`	`full date/time must be specified`
Date/Time	`yyyy-mm-ddThh:mm:??`	`optional seconds;` `e.g. appointment date/time`
Date/Time	`yyyy-mm-ddThh:mm:XX`	`no seconds;` `e.g. appointment date/time`
Date/Time	`yyyy-mm-ddThh:??:XX`	`no seconds, minutes optional;` `e.g. in patient-recollected date/times`
Date/Time	`yyyy-??-??T??:??:??`	`minimum valid date/time constraint`

Intervals

Dates, times and date/times can also be constrained using intervals. Each date, time etc in an interval may be a literal date, time etc value, or a value based on a pattern. In the latter case, the limit values are specified using the patterns from the above table, but with numbers in the positions where `'X'` and `'?'` do not appear. For example, the pattern `yyyy-??-XX` could be transformed into `1995-??-XX` to mean any partial date in 1995. Examples of such constraints:

```
|1995-??-XX|                         -- any partial date in 1995
|09:30:00|                           -- exactly 9:30 am
|< 09:30:00|                         -- any time before 9:30 am
|<= 09:30:00|                        -- any time at or before 9:30 am
|> 09:30:00|                         -- any time after 9:30 am
|>= 09:30:00|                        -- any time at or after 9:30 am
|2004-05-20..2004-06-02|             -- a date range
|2004-05-20T00:00:00..2005-05-19T23:59:59|  -- a date/time range
```

4.5.6.2 Duration Constraints

Patterns

Patterns based on ISO 8601 can be used to constraint durations in the same way as for Date/time types. The general form of a pattern is (EBNF; all tokens are literals):

```
P[Y|y][M|m][W|w][D|d][T[H|h][M|m][S|s]]
```

Note that allowing the 'W' designator to be used with the other designators corresponds to a deviation from the published ISO 8601 standard used in *open*EHR, namely:

- durations are supposed to take the form of PnnW or PnnYnnMnnDTnnHnnMnnS, but in *open*EHR, the W (week) designator can be used with the other designators, since it is very common to state durations of pregnancy as some combination of weeks and days.

The use of this pattern indicates which "slots" in an ISO duration string may be filled. Where multiple letters are supplied in a given pattern, the meaning is "or", i.e. any one or more of the slots may be supplied in the data. This syntax allows specifications like the following to be made:

```
Pd          -- a duration containing days only, e.g. P5d
Pm          -- a duration containing months only, e.g. P5m
PTm         -- a duration containing minutes only, e.g. PT5m
Pwd         -- a duration containing weeks and/or days only, e.g. P4w
PThm        -- a duration containing hours and/or minutes only, e.g. PT2h30m
```

List and Intervals

Durations can also be constrained by using absolute ISO 8601 duration values, or ranges of the same, e.g.:

```
PT1m                 -- 1 minute
P1dT8h               -- 1 day 8 hrs
|PT0m..PT1m30s|      -- Reasonable time offset of first apgar sample
```

Mixed Pattern and Interval

In some cases there is a need to be able to limit the allowed units as well as state a duration interval. This is common in obstetrics, where physicians want to be able to set an interval from say 0-50 weeks and limit the units to only weeks and days. This can be done as follows:

```
PWD/|P0W..P50W|      -- 0-50 weeks, expressed only using weeks and days
```

The general form is a pattern followed by a slash ('/') followed by an interval, as follows:

```
duration_pattern '/' duration_interval
```

4.5.7 Terminology Constraints

Coded terms are treated as primitive types in ADL. A coded term is assumed to consist of:

- a terminology identifier
- optionally, a terminology version identifier
- a code or 'code string' (an expression made from codes, permitted in some terminologies)

Coded terms are represented lexically using the forms:

- `[terminology_id::code_string]`
- `[terminology_id(version_id)::code_string]`

Coded terms exist both in the archetype terminology, and in external terminologies. To reference an archetype-local term, the special terminology identifier "local" is used, i.e. as in `[local::at12]`.

Constraints on coded terms are defined using the

Not all constraints can be defined easily within an archetype. One common category of constraint that should be defined externally, and referenced from the archetype is the 'value set' for attributes where the values come from an external authoritative resource. In health, typical examples include 'terminology' resources such as WHO ICDx[1] and SNOMED[2] terminologies and drug databases. The need

1. http://www.who.int/classifications/icd/en/
2. http://www.ihtsdo.org/

within the archetype in this case is to refer to a value set from the resource, rather than defining them inline. The following example shows how this is done in cADL, using the example of an external terminology resource:

```
ENTRY[id1] ∈ {                  -- blood pressure measurement
    name ∈ {                    -- any synonym of BP
        DV_CODED_TEXT ∈ {
            defining_code ∈ {[local::ac1]}
        }
    }
}
```

In the above, the constraint on DV_CODED_TEXT.*defining_code* (of reference model type CODE_PHRASE) is set to an archetype constraint, or 'ac' code of the form [acNNNN], which acts as an internal identifier of a value set that is defined in the external resource. In the terminology section of the archetype, this identifier can be bound to a URI indicating the set of values from the resource.

Placeholder constraints are an alternative to enumeration of some value sets within an archetype. Inline enumeration will work perfectly well in a technical sense, but has at least two limitations. Firstly, the intended set of values allowed for the attribute may change over time (e.g. as has happened with 'types of hepatitis' since 1980), and since the authoritative resource is elsewhere, the archetype has to be continually updated. With a large repository of archetypes, each containing inline coded term constraints, this approach is likely to be unsustainable and error-prone. Secondly, the best means of defining the value set is in general not likely to be via enumeration of the individual terms, but in the form of a semantic expression that can be evaluated against the external resource. This is because the value set is typically logically specified in terms of inclusions, exclusions, conjunctions and disjunctions of general categories.

Discussion: Consider for example the value set logically defined as "any bacterial infection of the lung". The possible values would be codes from a target terminology, corresponding to numerous strains of pneumococcus, staphlycoccus and so on, but not including species that are never found in the lung. Rather than enumerate the list of codes corresponding to this value set (which is likely to be quite large), the archetype author is more likely to rely on semantic links within the external terminology to express the set; a query such as 'is-a bacteria and has-site lung' might be definable against the terminology such as SNOMED-CT or ICD10.

In a similar way, other value sets, including for quantitative values, are likely to be specified by queries or formal expressions, and evaluated by an external knowledge service. Examples include "any unit of pressure" and "normal range values for serum sodium".

In such cases, expressing the placeholder constraint could be done by including the query or other formal expression directly within the archetype itself. However, experience shows that this is problematic in various ways. Firstly, there is little if any standardisation in such formal value set expressions or queries for use with knowledge services - two archetype authors could easily create competing syntactical expressions for the same logical constraint. A second problem is that errors might be made in the query expression itself, or the expression may be correct at the time of authoring, but need subsequent adjustment as the relevant knowledge resource grows and changes. The consequence of this is the same as for a value set enumerated inline - it is unlikely to be sustainable for large numbers of archetyes.

These problems are not accidental: a query with respect to a terminological, ontological or other knowledge resource is most likely to be authored correctly by maintainers or experts of the knowledge resource, rather than archetype authors; it may well be altered over time due to improvements in the query formalism itself.

4.5.8 Constraints on Lists of Primitive types

In many cases, the type in the information model of an attribute to be constrained is a list or set of primitive types, e.g. List<Integer>, Set<String> etc. As for complex types, this is indicated in cADL using the `cardinality` keyword, as follows:

```
some_attr cardinality ∈ {0..*} ∈ {some_constraint}
```

The pattern to match in the final braces will then have the meaning of a list or set of value constraints, rather than a single value constraint. Any constraint described above for single-valued attributes, which is commensurate with the type of the attribute in question, may be used. However, as with complex objects, the meaning is now that every item in the list is constrained to be any one of the values implied by the constraint expression. For example,

```
speed_limits cardinality ∈ {0..*; ordered} ∈ {50, 60, 70, 80, 100, 130}
```

constrains each value in the list corresponding to the value of the attribute *speed_limits* (of type `List<Integer>`), to be any one of the values 50, 60, 70 etc.

4.5.9 Assumed Values

When archetypes are defined to have optional parts, an ability to define 'assumed' values is useful. For example, an archetype for the concept 'blood pressure measurement' might include an optional data point describing the patient position, with choices 'lying', 'sitting' and 'standing'. Since the section is optional, data could be created according to the archetype which does not contain the protocol section. However, a blood pressure cannot be taken without the patient in some position, so clearly there could be an implied or 'assumed' value.

The archetype allows this to be explicitly stated so that all users/systems know what value to assume when optional items are not included in the data. Assumed values are currently definable on primitive types only, and are expressed after the constraint expression, by a semi-colon (';') followed by a value of the same type as that implied by the preceding part of the constraint. The use of assumed values is illustrated here for a number of primitive types:

```
length matches {|0..1000|; 200}            -- allow 0 - 1000, assume 200
some_flag matches {True, False; True}      -- allow T or F, assume T
some_date matches {yyyy-mm-dd hh:mm:XX; 1800-01-01T00:00:00}
```

If no assumed value is stated, no reliable assumption can be made by the receiver of the archetyped data about what the values of removed optional parts might be, from inspecting the archetype. However, this usually corresponds to a situation where the assumed value does not even need to be stated - the same value will be assumed by all users of this data, if its value is not transmitted. In other cases, it may be that it doesn't matter what the assumed value is. For example, an archetype used to capture physical measurements might include a "protocol" section, which in turn can be used to record the "instrument" used to make a given measurement. In a blood pressure specialisation of this archetype it is fairly likely that physicians recording or receiving the data will not care about what instrument was used.

4.6 Syntax Validity Rules

The following syntax validity rules apply to the cADL syntax.

> **SCAS: attribute structure validity**. an attribute constraint must either contain object constraints, or the 'any' ('*') constraint, but not be empty.

> **SEXL: existence limits validity**. an existence constraint must define a range which can only be one of the following: 0..0, 0..1, 1..1.

5 Assertions

5.1 Overview

This section describes the assertion sub-language of archetypes. Assertions are used in archetype “slot” clauses in the cADL `definition` section, and in the `rules` section. The following simple assertion in the `rules` section of an archetype says that the speed in kilometres of some node is related to the speed-in-miles by a factor of 1.6:

```
validity: /speed[id2]/kilometres[id14]/magnitude =
          /speed[id4]/miles[id14]/magnitude * 1.6
```

5.1.1 Requirements

Assertions are needed in archetypes to express rules in two locations in an archetype. In an archetype slot, assertions can be stated to control what archetypes are allowed in the slot, as shown in the following example:

```
CLUSTER[id3] occurrences matches {0..1} matches {-- Detail
    items cardinality matches {0..*; unordered} matches {
       allow_archetype CLUSTER[id9] occurrences matches {0..1} matches {
          include
             archetype_id/value matches {/openEHR-EHR-CLUSTER.exam-.+\.v1/}
       }
    }
}
```

In the above, the statement following the `include` keyword expresses a condition on the value found at the path `archetype_id/value`, using the familiar ADL `matches` operator, and a regular expression on archetype identifiers. Most slot statements are of this kind, with some requiring slightly more complex expressions. See section 4.3.10 on page 38 for more details.

The main requirement for assertions in archetypes is for expressing rules that cannot be expressed uding the standard cADL syntax. Types of rules include:

- constraints involving more than one node in an archetype, such as a rule stating that the sum of the five 0-2 value scores in an Apgar test (heartrate, breathing, muscle tone, reflex, colour) correspond to the Apgar total, recorded in a sixth node;
- rules involving predefined variables such as ‘current date’;
- rules involving query results from a data or knowledge context, allowing values such as ‘patient date of birth’ to be referenced.

The semantic requirements are for expressions including arithmetic, boolean, and relational operators, some functions, quantifier operators, a notion of operator precedence, parentheses, constant values, and certain kinds of variables. However, there is no requirement for procedural semantics, type declarations or many of the other complexities of full-blown programming languages.

5.1.2 Design Basis

The archetype assertion language is a small language of its own. Formally it is a reduced first-order predicate logic language with various operators. It has similarities with OMG’s OCL (Object Constraint Language) syntax, and is also similar to the assertion syntax which has been used in the Object-Z [14] and Eiffel [12] languages and tools for over a decade (see Sowa [15], Hein [8], Kilov &

Ross [9] for an explanation of predicate logic in information modelling). None of these languages has been used directly, for reasons including:

- OCL has a complex type system, and includes some undecidable procedural semantics;
- none have adequate variable referencing mechanisms, such as to paths and external queries;
- they are too powerful, and would introduce unnecessary complexity into archetypes and templates.

There are also similarities with other languages developed in the health arena for expressing 'medical logic' (Arden), guidelines (GLIF and many others) and decision support (GELLO and many others). These languages were not directly used either, for reasons including:

- none have a path referencing mechanism;
- some are too procedural (Arden, GLIF);
- current versions of some of these languages have been made specific to the HL7v3 RIM, a particular model of health information designed for message representation (GLIF 3.x, GELLO);
- all in their published form are too powerful for the needs identified here.

The design approach used here was to create a small concrete syntax allowing for a core subset of first-order predicate logic, which could easily be parsed into a typical parse-tree form, defined in the *open*EHR Archetype Object Model. Many different variations on syntax elements are possible (as evidenced by the many formal logic syntaxes used in mathematics and computing theory); the elements used here were chosen for ease of expression using normal kebyoard characters and intuitiveness.

5.2 Keywords

The syntax of the invariant section is a subset of first-order predicate logic. In it, the following keywords can be used:

- `exists`, `for_all`,
- `and`, `or`, `xor`, `not`, `implies`
- `true`, `false`

Symbol equivalents for some of the above are given in the following table.

Textual Rendering	Symbolic Rendering	Meaning
matches, is_in	∈	Set membership, "p is in P"
exists	∃	Existential quantifier, "there exists ..."
for_all	∀	Universal quantifier, "for all x..."
implies	→	Material implication, "p implies q", or "if p then q"
and	∧	Logical conjunction, "p and q"
or	∨	Logical disjunction, "p or q"
xor	⊻	Exclusive or, "only one of p or q"
not, ~	~, ¬	Negation, "not p"

5.3 Typing

The assertion language is fully typed. All operators, variables and constants have either assumed or declared type signatures.

5.4 Operators

Assertion expressions can include arithmetic, relational and boolean operators, plus the existential and universal quantifiers.

5.4.1 Arithmetic Operators

The supported arithmetic operators are as follows:

addition: +

subtraction: -

multiplication: *

division: /

exponent: ^

modulo division: % -- remainder after integer division

5.4.2 Equality Operators

The supported equality operators are as follows:

equality: =

inequality: !=

The semantics of these operators are of value comparison.

5.4.3 Relational Operators

The supported relational operators are as follows:

less than: <

less than or equal: <=

greater than: >

greater than or equal: >=

The semantics of these operators are of value comparison on entities of Comparable types (see *open*EHR Support IM, Assumed Types section). All generate a Boolean result.

5.4.4 Boolean Operators

The supported boolean operators are as follows:

not: **not**

and: **and**

xor: **xor**

implies: **implies**

set membership: **matches**, **is_in**

The boolean operators also have symbolic equivalents shown earlier. All boolean operators take Boolean operands and generate a Boolean result. The `not` operator can be applied as a prefix operator to all operators returning a boolean result.

5.4.5 Quantifiers

The two standard logical quantifier operators are supported:

existential quantifier: **exists**

universal quantifier: **for_all**

These operators also have the usual symbolic equivalents shown earlier. The `exists` operator can be used on an variable, including paths referring to a node or value within an archetype. The `for_all` operator can be applied to sets and lists, such as referred to by a path to a multiply-valued attribute.

5.4.6 Functions

The following functions are supported:

sum (x, y, ...): equivalent to x + y + ...

mean (x, y, ...): the mean (average) value of x, y, ...

max (x, y, ...): the maximum value among x, y, ...

min (x, y, ...): the minimum value among x, y, ...

All of the above functions have the signature `func(Real, ...):Real`, but will also perform as though having the signature `func(Integer, ...):Integer`, due to automatic numeric type promotion/demotion rules.

Other functions may be added in the future.

5.5 Operands

Operands in an assertion expression are typed and are of four kinds, as described in the following subsections.

5.5.1 Constants

Constant values are of any primitive type defined in the *open*EHR Support IM Assumed Types, and expressed according in the ODIN syntax (see section 4.5 on page 32), i.e.:

- Character, e.g. `'x'`;
- String, e.g. `"this is a string"`;
- Boolean, e.g. `True`, `False`;
- Integer, e.g. `5`;
- Real, e.g. `5.2`;
- ISO8601_DATE, e.g. `2004-08-12`;
- ISO8601_TIME, e.g. `12:00:59`;
- ISO8601_DATE_TIME, e.g. `2004-08-12T12:00:59`;
- ISO8601_DURATION, e.g. `P39W`;
- URI, e.g. `http://en.wikipedia.org/wiki/Everest`;
- coded term, e.g. `[snomed_ct::2004950]`;
- Intervals of any numeric type, according to ODIN syntax e.g. `|70..130|`;

- List of any primitive type, e.g. "string1", "string2", "string3";

5.5.2 Object References

A reference to an object in data, including a leaf value, is expressed using an archetype path. All such paths are absolute (i.e. contain a leading '/') and are understood to be with respect to the root of the current archetype. References to archetype nodes have the type defined at the relevant point in the underlying reference model. Examples include:

```
/data[id2]/items[id3]/value[id35]/value -- Date of initial onset; type
ISO8601_DATE
```

To Be Continued:

5.5.3 Built-in Variables

A small number of built-in variables are available for use in assertions, and are referred to using a '$' symbol, for example `$current_date`. Built-in variables defined include:

```
$current_date: ISO8601_DATE
$current_time: ISO8601_TIME
$current_date_time: ISO8601_DATE_TIME
$current_year: Integer
$current_month: Integer
```

5.5.4 Archetype-defined Variables

Variables may be declared within the rules section of an archetype. This is done using the following syntax:

```
$var_name:Type ::= expression
```

This facility can be used to equate a variable name to a path, e.g. the following equates the variable `$diagnosis` to the code at the path containing the diagnosis (e.g. in the `openEHR-EHR-EVALUATION.problem-diagnosis.v1` archetype):

```
$diagnosis:CODE_PHRASE ::= /data/items[id2.1]/value/defining_code
```

The variable can then be used instead of the path in subsequent expressions.

5.5.5 External Queries

An expression referring to an externally defined query, possibly including arguments, may be defined using the variable declaration syntax. The general pattern is as follows:

```
$varname:Type ::= query(context, query_name, arg1, arg2, ...)
```

Examples include:

```
$date_of_birth:ISO8601_DATE ::=    query("ehr", "date_of_birth")
$has_diabetes:Boolean ::=          query("ehr", "has_diagnosis",
                                      "snomed_ct::1234567")
$is_female:Boolean ::=             query("ehr", "is_female")
```

Any number of arguments can be included.

Query Contexts

Query Names

5.6 Precedence and Parentheses

To Be Continued:

5.7 Conditions

Example....

```
$is_female implies exists /path/to/xxx
```

5.8 Natural Language Issues

xx

6 ADL Paths

6.1 Overview

The notion of paths is integral to ADL, and a common path syntax is used to reference nodes in both ODIN and cADL sections of an archetype. The same path syntax works for both, because both ODIN and cADL have an alternating object/attribute structure. However, the interpretation of path expressions in ODIN and cADL differs slightly; the differences are explained in the ODIN and cADL sections of this document. This section describes only the common syntax and semantics.

The general form of the path syntax is as follows (see syntax section below for full specification):

```
path: ['/'] path_segment { '/' path_segment }+
path_segment: attr_name [ '[' object_id ']' ]
```

Essentially, ADL paths consist of segments separated by slashes ('/'), where each segment is an attribute name with optional object identifier predicate, indicated by brackets ('[]').

> **ADL Paths** are formed from an alternation of segments made up of an attribute name and optional object node identifier predicate, separated by slash ('/') characters. Node identifiers are delimited by brackets (i.e. []).

Similarly to paths used in file systems, ADL paths are either absolute or relative, with the former being indicated by a leading slash.

> Paths are **absolute** or **relative** with respect to the document in which they are mentioned. Absolute paths commence with an initial slash ('/') character.

The ADL path syntax also supports the concept of "movable" path patterns, i.e. paths that can be used to find a section anywhere in a hierarchy that matches the path pattern. Path patterns are indicated with a leading double slash ("//") as in Xpath.

> Path **patterns** are absolute or relative with respect to the document in which they are mentioned. Absolute paths commence with an initial slash ('/') character.

6.2 Relationship with W3C Xpath

The ADL path syntax is semantically a subset of the Xpath query language, with a few syntactic shortcuts to reduce the verbosity of the most common cases. Xpath differentiates between "children" and "attributes" sub-items of an object due to the difference in XML between Elements (true sub-objects) and Attributes (tag-embedded primitive values). In ADL, as with any pure object formalism, there is no such distinction, and all subparts of any object are referenced in the manner of Xpath children; in particular, in the Xpath abbreviated syntax, the key `child::` does not need to be used.

ADL does not distinguish attributes from children, and also assumes the `node_id` attribute. Thus, the following expressions are legal for cADL structures:

```
items[1]         -- the first member of 'items'
items[systolic]  -- the member of 'items' with meaning 'systolic'
items[id1]       -- the member of 'items' with node id 'id1'
```

The Xpath equivalents are:

```
items[1]         -- the first member of 'items'
items[meaning() = 'systolic']-- the member of 'items' for which the meaning()
function evaluates to "systolic"
items[@archetype_node_id = 'id1']-- the member of 'items' with key 'id1'
```

In the above, `meaning()` is a notional function is defined for Xpath in *open*EHR, which returns the rubric for the `node_id` of the current node. Such paths are only for display purposes, and paths used for computing always use the 'at' codes, e.g. `items[id1]`, for which the Xpath equivalent is `items[@node_id = 'id1']`.

The ADL movable path pattern is a direct analogue of the Xpath syntax abbreviation for the 'descendant' axis.

7 Default Values

7.1 Overview

In ADL 1.5, it is possible to specify a default value for any object node. This almost always limited to use in templates, since default values are usually specific to local contexts or use cases. However they may validly be used in any archetype.

Within a template, a default value can be defined to support the situation where only one value is possible for a data item due to the specific nature of the template. For example, a blood pressure archetype may allow a number of possible values for 'patient position', such as 'lying', and 'sitting', 'standing'. When used in a hospital, the patient will usually be lying so a default value for this can be set, as shown in the following example:

```
/data/events[id6]/state/items[id8]/value ∈ {
    DV_CODED_TEXT ∈ {
        _default ∈ {
            (DV_CODED_TEXT) <
                defining_code = <[snomedct::163033001]> -- lying BP
            >
        }
    }
}
```

Default values are expressed in ODIN syntax, since they are instances of objects, rather than being constraints. They are introduced using a pseudo-attribute '_default', which is detected by the compiler as being a meta-attribute. The example above only sets the default value, but it could have also modified the constraint on the value object as well, as in the following version (where the standing blood pressure possibility from the archetype has been removed):

```
/data/events[id6]/state/items[id8]/value ∈ {
    DV_CODED_TEXT ∈ {
        defining_code ∈ {
            [snomedct::
            163033001, -- lying blood pressure
            163035008] -- sitting blood pressure
        }
        _default ∈ {
            (DV_CODED_TEXT) <
                defining_code = <[snomed::163033001]> -- lying BP
            >
        }
    }
}
```

Default values can be set in the same way on container objects, such that one or more container objects distinguished by node identifier or name (if renaming has been used in the template) within the same container can have a default value assigned to them.

To Be Continued: example

A default value is either of the same type as specified by the corresponding archetype node (*rm_type_name* attribute) or any subtype allowed by the reference model.

8 ADL - Archetype Definition Language

8.1 Introduction

This section describes whole ADL artefacts. The relationship of the cADL-encoded `definition` section and the ODIN-encoded `terminology` section is discussed in detail. In this section, only standard ADL (i.e. the standard cADL constructs and types described so far) is assumed.

TBD_4: Archetypes for use in particular domains can also be built with more efficient cADL syntax and domain-specific types, as described in Customising ADL on page 105, and the succeeding sections.

Some syntax valdity rules are defined, but validity in general is defined by the rules stated in the AOM specification, which can be checked by a compiler as soon as an AOM structure is parsed from an ADL document (or other serialsiation format).

The general structure of ADL artefacts is as follows:

```
(([flat] archetype | template | template_overlay) |
        operational_template) (qualifiers)
    id
[specialize
    parent_id]
[concept                    -- deprecated
    coded_concept_name]
language
    ODIN
description
    ODIN
definition
    cADL
[rules
    assertions]
terminology
    ODIN
[annotations
    ODIN]
[revision_history
    ODIN]
```

An ADL source template has the structure (‘template’ keyword; must be specialised):

```
[flat] template (qualifiers)
    id
specialize
    parent_id
language
    ODIN
description
    ODIN
definition
    cADL
[rules
    assertions]
terminology
    ODIN
[annotations
```

```
        ODIN]
    [revision_history
        ODIN section]
```

An ADL template overlay has the structure shown below (‘template_overlay’ keyword, must be specialised; minimal sections):

```
    template_overlay (flags)
        id
    specialize
        parent_id
    definition
        cADL
    terminology
        ODIN
```

The structure of an operational template is as follows (‘operational_template’ keyword; full flattened structure, including component_ontologies section):

```
    operational_template (qualifiers)
        template_id
    language
        ODIN
    description
        ODIN
    definition
        cADL
    [rules
        assertions]
    terminology
        ODIN
    [annotations
        ODIN]
    component_ontologies
        ODIN
```

8.2 File-naming Convention

Up until ADL 1.4, archetypes were expressed in ‘flat’ form ADL and were saved in files with the extension ‘.adl’. These are now treated as legacy flat format files. Beginning with ADL 1.5, the source file format of an archetype is the ‘differential’ form, with the extension ‘.adls’. The flat format being generated by tools, and now has the extension ‘.adlf’.

For specialised archetypes, differential form follows the object-oriented convention and only includes overridden or new elements but not unchanged inherited elements. Inherited elements are determined by compiling and ‘flattening’ a differential archetype with respect to the parent archetypes in its *inheritance lineage*.

For top-level archetypes the full structure is included in the differential file, but internal references, where they exist, are expanded out in the flat form.

8.3 Basics

8.3.1 Keywords

ADL has a small number of keywords which are reserved for use in archetype declarations, as follows:

- `archetype`, `template`, `template_overlay`, `operational_template`,
- `specialise`/`specialize`,
- `concept`,
- `language`,
- `description`, `definition`, `rules`, `terminology`

All of these words can safely appear as identifiers in the `definition` and `terminology` sections.

Deprecated keywords include:

- `invariant -- replaced by 'rules'`

8.3.2 Artefact declaration

The first word in a source ADL archetype declares the artefact type, and is one of the following keywords:

‘archetype’: signifies an archetype;

‘template’: signifies a template;

‘template_overlay’: signifies an overlay component of a template.

The flattened form of any of the above types starts with the keyword ‘flat’ followed by the artefact type.

A fourth artefact type is also possible.

‘operational_archetype’: signifies an operational archetype, generated by flattening a template.

8.3.3 Node Identifier Codes

In the `definition` section of an ADL archetype, a specific set of codes is used as node identifiers. Identifier codes always appear in brackets (`[]`), and begin with the ‘id’ prefix. Specialisations of locally coded concepts have the same root, followed by ‘dot’ extensions, e.g. `[id10.2]`. From a terminology point of view, these codes have no implied semantics - the ‘dot’ structuring is used as an optimisation on node identification.

> **NB:** In ADL 1.4 and transitional forms of ADL 1.5, ‘at’ codes were used as node identifiers. Within top-level archetypes, the code numbers were 0-padded to make up a 4-digit code, e.g. ‘at0004’. In ADL 1.5, all such codes are replaced by an ‘id’ code whose number is one greater than the original at-code number. This numeric change is to accommodate the conversion of the root identifier code at0000 to id1 rather than id0.

8.3.4 Local Term Codes

In the `definition` section of an ADL archetype, a second set of codes is used for terms denoting constraints on coded items. Term codes are either local to the archetype, or from an external lexicon. This means that the archetype description is the same in all languages, and is available in any language that the codes have been translated to. All term codes are shown in brackets (`[]`) and are prefixed with “at”, e.g. `at10`. Codes of any length are acceptable in ADL archetypes. Specialisations of locally coded concepts have the same root, followed by ‘dot’ extensions, e.g. `at10.2`. From a terminology point of view, these codes have no implied semantics - the ‘dot’ structuring is used as an optimisation on node identification.

> **NB:** In ADL 1.4 and transitional forms of ADL 1.5, 0-padded ‘at’ codes were used within top-level archetypes. In ADL 1.5, all such codes are reformatted to remove the 0-padding.

8.3.5 Local Constraint Codes

A third kind of local code is used to stand for constraints on code text items in the body of the archetype. Although these could be included in the main archetype body, because they are language- and/or terminology-sensitive, they are defined in the terminology section, and referenced by codes prefixed by "ac", e.g. `[ac9]`.

> *Deprecated:* In ADL 1.4 and transitional forms of ADL 1.5, 0-padded 'ac' codes were used within top-level archetypes. In ADL 1.5, all such codes are reformatted to remove the 0-padding.

8.4 Header Sections

8.4.1 Archetype Identification Section

This section introduces the archetype with the 'archetype' keyword, followed by a small number of items of meta-data in parentheses, and an archetype identifier. The archetype identifier may include a namespace, in the form of a reverse domain name, which denotes the *original authoring organisation*. The lack of a namespace in the identifier indicates an *ad hoc*, uncontrolled artefact, not formally associated with any organisation, typical for experimental archetypes, and pre-ADL 1.5 archetypes not yet upgraded to have a namespace. The main part of the identifier is multi-axial concept identifier.

A typical identification sectionfor an *ad hoc* archetype is as follows:

```
archetype (adl_version=1.5)
    openEHR-EHR-OBSERVATION.haematology_result.v1
```

8.4.1.1 Namespaces

A namespaced archetype will have an identification section like the following examples:

```
archetype (adl_version=1.5)
    br.gov.saude::openEHR-EHR-OBSERVATION.haematology_result.v1

template (adl_version=1.5)
    uk.org.primary_care::openEHR-EHR-OBSERVATION.haematology_result.v1

archetype (adl_version=1.5)
    org.openehr::openEHR-EHR-OBSERVATION.haematology_result.v1
```

Namespaces are used to distinguish locally created artefacts representing a given concept (such as 'haematology result') from an artefact created elsewhere intended to represent the same concept.

Once a namespace is attached to an archetype, it is considered a part of the identifier, and never changed, even if the archetype moves to a new publishing organisation. This ensures the constant relationship between archetypes and the data created using them.

8.4.1.2 Ontological Archetype Identifier

The multi-axial archetype identifier identifies archetypes in a global concept space within a given namespace. It is also known as an 'ontological' identifier, since the concept space can be understood as an ontology of archetypes. The syntax of the identifier is described in the Identification section of the *open*EHR Support IM specification. The structure of the concept space is essentially two-level, with the first level being a reference model class (e.g. *open*EHR `OBSERVATION` class) and the second being a domain concept (e.g. 'haematology result').

Because namespaces are usually treated hierarchically, higher level namespaces (e.g. '.org' domains) are assumed to be includable by more local namespaces, with the result that the concept definition space is inherited as well.

The semantics of the ontological identifier are described in detail in the Knowledge Artefact Identification specification.

8.4.1.3 Specialised Archetype Identification

The archetype identifier of any specialised archetype, including all templates, follows the same rules as for non-specialised archetypes.

> **Note**: in previous versions of ADL, the archetype identifier of a specialised archetype had a concept part that consisted of the concept part of the parent followed by '-' and a further specialised concept. For example, `openEHR-EHR-OBSERVATION.haematology-cbc.v1` was a valid child of `openEHR-EHR-OBSERVATION.haematology.v1`. This restriction is no longer the case. The previous style of identifier is still legal, but the '-' no longer has any significance.

8.4.1.4 Validity

The following syntax validity rule applies in the identification section:

> **SARID: archetype identifier validity**. the identifier of the artefact must conform to the ARCHETYPE_ID identifier syntax defined in the *open*EHR.Support IM Specification.

8.4.1.5 ADL Version Indicator

An ADL version identifier is mandatory in all archetypes, and is expressed as a string of the form adl_version=N.M, where N.M is the ADL version identifier.

8.4.1.6 Controlled Indicator

A flag indicating whether the archetype is change-controlled or not can be included after the version, as follows:

```
archetype (adl_version=1.5; controlled)
    org.openehr::openEHR-EHR-OBSERVATION.haematology.v1
```

This flag may have the two values "controlled" and "uncontrolled" only, and is an aid to software. Archetypes that include the "controlled" flag should have the revision history section included, while those with the "uncontrolled" flag, or no flag at all, may omit the revision history. This enables archetypes to be privately edited in an early development phase without generating large revision histories of little or no value.

8.4.1.7 Generated Indicator

A flag indicating whether the archetype was generated or authored can be included after the version, as follows:

```
archetype (adl_version=1.5; generated)
    org.openehr::openEHR-EHR-OBSERVATION.haematology.v1
```

This marker is used to support the migration to differential archetype representation introduced in ADL 1.5, to enable proper representation of specialised archetypes. The 'generated' marker can be used on specialised archetypes - i.e. ADL 1.5 style .adls files - generated from flat archetypes - ADL 1.4 .adl files - and also in flat archetypes generated from differential files, by an inheritance-flattening process.

8.4.1.8 Uid

A unique identifier for the archetype in the form of a GUID can be specified using the syntax below:

```
archetype (adl_version=1.5; uid=15E82D77-7DB7-4F70-8D8E-EED6FF241B2D)
```

This identifier is set at initial creation or at any time later, and nevre subsequently changes. It acts as an identifier for the physical artefact, regardless of what semantics are changed, including changes to the constituent parts of the multi-axial identifier.

8.4.2 Specialise Section

This optional section indicates that the archetype is a specialisation of some other archetype, whose identity must be given. Only one specialisation parent is allowed, i.e. an archetype cannot 'multiply-inherit' from other archetypes. An example of declaring specialisation is as follows:

```
archetype (adl_version=1.5)
    openEHR-EHR-OBSERVATION.cbc.v1
specialise
    openEHR-EHR-OBSERVATION.haematology.v1
```

Here the identifier of the new archetype is derived from that of the parent by adding a new section to its domain concept section. See the `ARCHETYPE_ID` definition in the identification package in the *open*EHR Support IM specification.

Note that both the US and British English versions of the word "specialise" are valid in ADL.

The following syntax validity rule applies in the specialisation section:

> **SASID: archetype specialisation parent identifier validity**. for specialised artefacts, the identifier of the specialisation parent must conform to the ARCHETYPE_ID identifier syntax defined in the *open*EHR Support IM Specification.

8.4.3 Language Section and Language Translation

The `language` section includes meta-data describing the original language in which the archetype was authored (essential for evaluating natural language quality), and the total list of languages available in the archetype. There can be only one `original_language`. The `translations` list must be updated every time a translation of the archetype is undertaken. The following shows a typical example.

```
language
    original_language = <[iso_639-1::en]>
    translations = <
        ["de"] = <
            language = <[iso_639-1::de]>
            author = <
                ["name"] = <"Frederik Tyler">
                ["email"] = <"freddy@something.somewhere.co.uk">
            >
            accreditation = <"British Medical Translator id 00400595">
        >
        ["ru"] = <
            language = <[iso_639-1::ru]>
            author = <
                ["name"] = <"Nina Alexandrovna">
                ["organisation"] = <"Dostoevsky Media Services">
                ["email"] = <"nina@translation.dms.ru">
            >
            accreditation = <"Russian Translator id 892230-3A">
        >
    >
```

Archetypes must always be translated completely, or not at all, to be valid. This means that when a new translation is made, every language dependent section of the `description` and `terminology` sections has to be translated into the new language, and an appropriate addition made to the `translations` list in the language section.

> **Note**: some non-conforming ADL tools in the past created archetypes without a language section, relying on the terminology section to provide the original_language (there called primary_language) and list of languages (languages_available). In the interests of backward compatibility, tool builders should consider accepting archetypes of the old form and upgrading them when parsing to the correct form, which should then be used for serialising/saving.

8.4.4 Description Section

The `description` section of an archetype contains descriptive information, or what some people think of as document "meta-data", i.e. items that can be used in repository indexes and for searching. The ODIN syntax is used for the description, as in the following example.

```
description
    original_author = <
        ["name"] = <"Dr J Joyce">
        ["organisation"] = <"NT Health Service">
        ["date"] = <2003-08-03>
    >
    lifecycle_state = <"initial">
    resource_package_uri =
        <"www.aihw.org.au/data_sets/diabetic_archetypes.html">

    details = <
        ["en"] = <
            language = <[iso_639-1::en]>
            purpose = <"archetype for diabetic patient review">
            use = <"used for all hospital or clinic-based diabetic reviews,
                including first time. Optional sections are removed according
                to the particular review"
            >
            misuse = <"not appropriate for pre-diagnosis use">
            original_resource_uri =
                <"www.healthdata.org.au/data_sets/
                               diabetic_review_data_set_1.html">
            other_details = <...>
        >
        ["de"] = <
            language = <[iso_639-1::de]>
            purpose = <"Archetyp für die Untersuchung von Patienten
                     mit Diabetes">
            use = <"wird benutzt für alle Diabetes-Untersuchungen im
                   Krankenhaus, inklusive der ersten Vorstellung. Optionale
                   Abschnitte werden in Abhängigkeit von der speziellen
                   Vorstellung entfernt."
            >
            misuse = <"nicht geeignet für Benutzung vor Diagnosestellung">
            original_resource_uri =
                <"www.healthdata.org.au/data_sets/
                               diabetic_review_data_set_1.html">
            other_details = <...>
        >
```

```
        >
```

A number of details are worth noting here. Firstly, the free hierarchical structuring capability of ODIN is exploited for expressing the 'deep' structure of the `details` section and its subsections. Secondly, the ODIN qualified list form is used to allow multiple translations of the `purpose` and `use` to be shown. Lastly, empty items such as `misuse` (structured if there is data) are shown with just one level of empty brackets. The above example shows meta-data based on the *open*EHR Archetype Object Model (AOM).

The `description` section is technically optional according to the AOM, but in any realistic use of ADL for archetypes, it will be required. A minimal description section satisfying to the AOM is as follows:

```
description
    original_author = <
        ["name"] = <"Dr J Joyce">
        ["organisation"] = <"NT Health Service">
        ["date"] = <2003-08-03>
    >
    lifecycle_state = <"initial">
    details = <
        ["en"] = <
            language = <[iso_639-1::en]>
            purpose = <"archetype for diabetic patient review">
        >
    >
```

8.4.5 Deprecated Sections

8.4.5.1 Concept Section

A 'concept' section was required up until ADL 1.4. In ADL 1.5, the concept section is deprecated, but allowed, enabling ADL 1.4 archetypes to be treated as valid. It will be removed in a future version of ADL, since it is completely redundant.

All archetypes represent some real world concept, such as a "patient", a "blood pressure", or an "antenatal examination". The concept is always coded, ensuring that it can be displayed in any language the archetype has been translated to. A typical `concept` section is as follows:

```
concept
    [at0000]     -- haematology result
```

In this concept definition, the term definition of `[at0000]` is the proper description corresponding to the "`haematology-cbc`" section of the archetype identifier above.

The following syntax validity rule applies to the concept section, if present, allowing parsers to correctly ignore it:

> **SACO: archetype concept validity**: if a concept section is present, it must consist of the 'concept' keyword and a single local term.

8.5 Definition Section

The `definition` section contains the main formal definition of the archetype, and is written in the Constraint Definition Language (cADL). A typical `definition` section is as follows:

```
definition
    OBSERVATION[id1] ∈ {         -- blood pressure measurement
        name ∈ {                 -- any synonym of BP
```

```
        DV_CODED_TEXT[id2] ∈ {
            defining_code ∈ {
                CODE_PHRASE[id3] ∈ {[ac1]}
            }
        }
    }
    data ∈ {
        HISTORY[id4] ∈ {                                  -- history
            events cardinality ∈ {1..*} ∈ {
                POINT_EVENT[id5] occurrences ∈ {0..1} ∈ {-- baseline
                    name ∈ {
                        DV_CODED_TEXT[id6] ∈ {
                            defining_code ∈ {
                                CODE_PHRASE[id7] ∈ {[ac2]}
                            }
                        }
                    }
                    data ∈ {
                        ITEM_LIST[id8] ∈ {      -- systemic arterial BP
                            items cardinality ∈ {2..*} ∈ {
                                ELEMENT[id9] ∈ {               -- systolic BP
                                    name ∈ {        -- any synonym of 'systolic'
                                        DV_CODED_TEXT[id10] ∈ {
                                            defining_code ∈ {
                                                CODE_PHRASE[id11] ∈ {[ac2]}
                                            }
                                        }
                                    }
                                    value ∈ {
                                        DV_QUANTITY[id12] ∈ {
                                            magnitude ∈ {0..1000}
                                            property ∈ {[properties::944]}
                                                                -- "pressure"
                                            units ∈ {[units::387]}  -- "mm[Hg]"
                                        }
                                    }
                                }
                                ELEMENT[id13] ∈ {          -- diastolic BP
                                    name ∈ {        -- any synonym of 'diastolic'
                                        DV_CODED_TEXT[id14] ∈ {
                                            defining_code ∈ {
                                                CODE_PHRASE[id15] ∈ {[ac3]}
                                            }
                                        }
                                    }
                                    value ∈ {
                                        DV_QUANTITY[id16] ∈ {
                                            magnitude ∈ {0..1000}
                                            property ∈ {[properties::944]}
                                                                -- "pressure"
                                            units ∈ {[units::387]}  -- "mm[Hg]"
                                        }
                                    }
                                }
                                ELEMENT[id17] occurrences ∈ {0..*} ∈ {*}
                                                          -- unknown new item
                            }
                        ...
```

This definition expresses constraints on instances of the types ENTRY, HISTORY, EVENT, ITEM_LIST, ELEMENT, QUANTITY, and CODED_TEXT so as to allow them to represent a blood pressure measurement, consisting of a history of measurement events, each consisting of at least systolic and diastolic pressures, as well as any number of other items (expressed by the [at9000] "any" node near the bottom).

8.5.1 Design-time and Run-time paths

All archetype object constraint nodes require a node identifier. When data are created according to the definition section of an archetype, the archetype node identifiers can be written into the data, providing a reliable way of finding data nodes, regardless of what other runtime names might have been chosen by the user for the node in question. There are two reasons for doing this. Firstly, querying cannot rely on runtime names of nodes (e.g. names like "sys BP", "systolic bp", "sys blood press." entered by a doctor are unreliable for querying); secondly, it allows runtime data retrieved from a persistence mechanism to be re-associated with the cADL structure which was used to create it.

An example which shows the difference between design-time meanings associated with node identifiers and runtime names is the following, from a SECTION archetype representing the problem/SOAP headings (a simple heading structure commonly used by clinicians to record patient contacts under top-level headings corresponding to the patient's problem(s), and under each problem heading, the headings "subjective", "objective", "assessment", and "plan").

```
SECTION[id1] matches {                    -- problem
   name matches {
      DV_CODED_TEXT[id2] matches {
         defining_code matches {[ac1]} -- any clinical problem type
      }
   }
}
```

In the above, the node identifier [id1] is assigned a meaning such as "clinical problem" in the archetype terminology section. The subsequent lines express a constraint on the runtime *name* attribute, using the internal code [ac1]. The constraint [ac1] is also defined in the archetype terminology section with a formal statement meaning "any clinical problem type", which could clearly evaluate to thousands of possible values, such as "diabetes", "arthritis" and so on. As a result, in the runtime data, the node identifier corresponding to "clinical problem" and the actual problem type chosen at runtime by a user, e.g. "diabetes", can both be found. This enables querying to find all nodes with meaning "problem", or all nodes describing the problem "diabetes". Internal [acNNNN] codes are described in Local Constraint Codes on page 69.

8.6 Rules Section

The rules section in an ADL archetype introduces assertions which relate to the entire archetype, and can be used to make statements which are not possible within the block structure of the definition section. Any constraint which relates more than one property to another is in this category, as are most constraints containing mathematical or logical formulae. Rules are expressed in the archetype assertion language, described in section 5 on page 57.

An assertion is a first order predicate logic statement which can be evaluated to a boolean result at runtime. Objects and properties are referred to using paths.

The following simple example says that the speed in kilometres of some node is related to the speed-in-miles by a factor of 1.6:

```
rules
    validity: /speed[id2]/kilometres[id13]/magnitude =
```

```
/speed[id4]/miles[id14]/magnitude * 1.6
```

8.7 Terminology Section

8.7.1 ADL 1.4 / 1.5 Version Compatibility

Deprecated: In ADL 1.4 archetypes, the terminology section used the following kind of nesting of containers:

```
term_definitions = <
    ["en"] = <
        items = <
            ["at0001"] = <...>
            ["at0002"] = <...>
        >
    >
>
```

Deprecated: This has been replaced in ADL 1.5 archetypes by the proper ODIN nested structures described below, which correspond to the nested Hash<> representational structures defined in the AOM. Tools that generate ADL1.4 style archetypes via archetype 'flattening' should output the ADL 1.4 style nesting structure to enable older tools to deal with the output.

8.7.2 Overview

The `terminology` section of an archetype is expressed in ODIN, and is where codes representing node IDs, constraints on text or terms, and bindings to terminologies are defined. Linguistic language translations are added in the form of extra blocks keyed by the relevant language. The following example shows the layout of this section.

```
terminology
    term_definitions = <
        ["en"] = <
            ["id1"] = <...>
            ["at1"] = <...>
            ["ac1"] = <...>
        >
        ["de"] = <
            ["id1"] = <...>
            ["at1"] = <...>
            ["ac1"] = <...>
        >
    >
    term_bindings = <
        ["snomed_ct"] = <
            ["id4"] = <...>
            ["ac1"] = <...>
            ...
        >
    >
```

The `term_definitions` section is mandatory, and must be defined for each translation carried out.

Deprecated: In ADL 1.4 the terminology section consisted of potentially 4 parts, i.e. term_definitions, constraint_definitions, term_bindings and constraint_bindings. The former two are now merged into one list, term_definitions, and the latter two into one list, term_bindings.

8.7.3 Term_definitions Section

This section is where all archetype local terms (including all at-codes, ac-codes, and at least the id-codes attached to container attribute children) are defined. The following example shows an extract from the English and German term definitions for the archetype local terms in a problem/SOAP headings archetype. Each term is defined using a structure of name/value pairs, and mustat least include the names "text" and "description", which are akin to the usual rubric, and full definition found in terminologies like SNOMED-CT. Each term object is then included in the appropriate language list of term definitions, as shown in the example below.

```
term_definitions = <
    ["en"] = <
        ["id1"] = <
            text = <"problem">
            description = <"The problem experienced by the subject
                of care to which the contained information relates">
        >
        ["id2"] = <
            text = <"problem/SOAP headings">
            description = <"SOAP heading structure for multiple problems">
        >
        ...
        ["id3"] = <
            text = <"plan">
            description = <"The clinician's professional advice">
        >
    >
["de"] = <
    ["id1"] = <
            text = <"klinisches Problem">
            description = <"Das Problem des Patienten worauf sich diese \
                    Informationen beziehen">
    >
    ["id2"] = <
            text = <"Problem/SOAP Schema">
            description = <"SOAP-Schlagwort-Gruppierungsschema fuer
                mehrfache Probleme">
    >
    ["id3"] = <
            text = <"Plan">
            description = <"Klinisch-professionelle Beratung des
                    Pflegenden">
    >
>
```

In some cases, term definitions may have been lifted from existing terminologies (only a safe thing to do if the definitions *exactly* match the need in the archetype). To indicate where definitions come from, a "provenance" tag can be used, as follows:

```
["id3"] = <
    text = <"plan">;
    description = <"The clinician's professional advice">;
    provenance = <"ACME_terminology(v3.9a)">
>
```

Note that this does not indicate a *binding* to any term, only its origin. Bindings are described in section 8.7.4.

The `term_definitions` section also includes definitions for archetype-local constraint codes, which are of the form `[acN]`. Each such code refers to some constraint such as “any term which is a subtype of ‘hepatitis’ in the ICD9AM terminology”; the constraint definitions do not provide the constraints themselves, but define the *meanings* of such constraints, in a manner comprehensible to human beings, and usable in GUI applications. This may seem a superfluous thing to do, but in fact it is quite important. Firstly, term constraints can only be expressed with respect to particular terminologies - a constraint for “kind of hepatitis” would be expressed in different ways for each terminology which the archetype is bound to. For this reason, the actual constraints are defined in the `term_bindings` section. An example of a constraint term definition for the hepatitis constraint is as follows:

```
[“ac1015”] = <
    text = <"type of hepatitis">
    description = <"any term which means a kind of viral hepatitis">
>
```

Note that while it often seems tempting to use classification codes, e.g. from the ICD vocabularies, these will rarely be much use in terminology or constraint definitions, because it is nearly always *descriptive*, not classificatory terms which are needed.

8.7.4 Term_bindings Section

This section is used to describe the equivalences between archetype local terms and terms and value sets found in external terminologies. Bindings are expressed as URIs.

TBD_5: note that the actual URIs here are probably not yet correct; refer to IHTSDO / Mayo / CTS2 etc

The main purpose for allowing query engines to search for an instance of some external term to determine what equivalent to use in the archetype.

Global Term Bindings

There are two types of term bindings that can be used, ‘global’ and path-based. The former is where an external term is bound directly to an archetype local term, and the binding holds globally throughout the archetype. In many cases, archetype terms only appear once in an archetype, but in some archetypes, at-codes are reused throughout the archetype. In such cases, a global binding asserts that the correspondence is true in all locations. A typical global term binding section resembles the following:

```
term_bindings = <
    ["umls"] = <
        ["id1"] = <http://umls.nlm.edu/id/C124305> -- apgar result
        ["id2"] = <.../0000000> -- 1-minute event
        ["id4"] = <.../C234305> -- cardiac score
        ["id5"] = <.../C232405> -- respiratory score
        ["id6"] = <.../C254305> -- muscle tone score
        ["id7"] = <.../C987305> -- reflex response score
        ["id8"] = <.../C189305> -- color score
        ["id9"] = <.../C187305> -- apgar score
        ["id10"] = <.../C325305> -- 2-minute apgar
        ["id11"] = <.../C725354> -- 5-minute apgar
        ["id12"] = <.../C224305> -- 10-minute apgar
    >
>
```

Each entry indicates which term in an external terminology is equivalent to the archetype internal codes. Note that not all internal codes necessarily have equivalents: for this reason, a terminology binding is assumed to be valid even if it does not contain all of the internal codes.

Path-based Bindings

The second kind of binding is one between an archetype path and an external code. This occurs commonly for archetypes where a term us re-used at the leaf level. For example, in the binding example below, the id4 code represents 'temperature' and the codes id3, id5, id6 etc correspond to various times such as 'any', 1-hour average, 1-hour maximum and so on. Some terminologies (notably LOINC, the laboratory terminology in this example) define 'pre-coordinated' codes, such as '1 hour body temperature'; these clearly correspond not to single codes such as id4 in the archetype, but to whole paths. In such cases, the key in each term binding row is a full path rather than a single term.

```
["LNC205"] = <
    ["/data[id2]/events[id3]/data[id1]/item[id4]"] =
<http://loinc.org/ver=LNC205/id=8310-5>
    ["/data[id2]/events[id5]/data[id1]/item[id4]"] = <.../8321-2>
    ["/data[id2]/events[id6]/data[id1]/item[id4]"] = <.../8311-3>
    ["/data[id2]/events[id7]/data[id1]/item[id4]"] = <.../8316-2>
    ["/data[id2]/events[id8]/data[id1]/item[id4]"] = <.../8332-0>
    ["/data[id2]/events[id9]/data[id1]/item[id4]"] = <.../8312-1>
    ["/data[id2]/events[id17]/data[id1]/item[id4]"] = <.../8325-3>
    ["/data[id2]/events[id19]/data[id1]/item[id4]"] = <.../8320-4>
>
```

Bindings to external value sets are also included in the bindings section, also as URIs:

```
term_bindings = <
    [“snomed_ct”] = <
        [“ac1”] = < http://terminology.org?query_id=12345>
        [“ac2”] = <http://terminology.org?query_id=678910>
    >
>
```

In this example, each local constraint code is formally defined to refer to a query defined in a terminology service, in this case, a terminology service that can interrogate the Snomed-CT terminology.

8.7.5 Deprecated Elements

In the 1.4 release of ADL, there were two separate bindings sections, `term_bindings` and `constraint_bindings`. These have been merged into one section, `bindings`.

Additionally in ADL 1.4, a `terminologies_available` header statement was required to identify all terminologies for which `term_bindings` sections have been written. For example:

```
terminologies_available = <“snomed_ct”, “loinc”>
```

This is no longer required. In archetypes that have it, it is ignored, and should not be included in ADL 1.5 syntax output serialisation.

8.8 Annotations Section

The annotations section of an archetype or template provides a place for node-level meta-data to be added to the archetype. This can be used during the design phase to track dependencies, design decisions, and specific resource references. Each annotation is keyed by the path of the node being annotated, or a pure RM path, and may have any number of tagged elements. A typical annotations section looks as follows.

```
annotations
```

```
        items = <
            [“en”] = <
                items = <
                    ["/data/items[at0.37]/items[at0.38]/value"] = < -- Clin st. /
stage
                        items = <
                            ["messaging requirement"] =
                                        <"= ‘staging’ field in msg type 2345">
                            ["guideline"] =
                                        <"http://guidelines.org/gl24.html#staging">
                            ["data dict equivalent"] = <"NHS data item aaa.1">
                        >
                    >
                    ["/data/items[at0.37]/items[at0.39]/value"] = < -- Clin st. /
Tumour
                        items = <
                            ["message requirement"] = <"tumour field in msg type
2345">
                            ["guideline"] = <"http://guidelines.org/gl24.html#mass">
                            ["data dict equivalent"] = <"NHS data item aaa.2">
                        >
                    >
                >
            >
        >
```

Annotations are defined as a separate section that can be easily removed in production versions of an archetype or template, and/or ignored in the generation of digital signatures.

Typically annotations are used to document a particular node within an archetype, specified by its (unique) archetype path. In some cases, the archetype (or tempate) author wants to document the *use of a reference model attribute* that is not constrained in the archetype (and therefore does not have an archetype path as such). In this case, the path will just be a valid RM path, i.e. a path relative to the top-level object of the archetype, and containing no at-codes.

8.9 Revision History Section

The revision history section of an archetype shows the audit history of changes to the archetype, and is expressed in ODIN syntax. It is optional, and is included at the end of the archetype, since it does not contain content of direct interest to archetype authors, and will monotonically grow in size. Where archetypes are stored in a version-controlled repository such as CVS or some commercial product, the revision history section would normally be regenerated each time by the authoring software, e.g. via processing of the output of the ‘prs’ command used with SCCS files, or ‘rlog’ for RCS files. The following shows a typical example, with entries in most-recent-first order (although technically speaking, the order is irrelevant to ADL).

```
    revision_history
        revision_history = <
            ["1.57"] = <
                committer = <"Miriam Hanoosh">
                committer_organisation = <"AIHW.org.au">
                time_committed = <2004-11-02 09:31:04+1000>
                revision = <"1.2">
                reason = <"Added social history section">
                change_type = <"Modification">
            >
            -- etc
```

```
        ["1.1"] = <
            committer = <"Enrico Barrios">
            committer_organisation = <"AIHW.org.au">
            time_committed = <2004-09-24 11:57:00+1000>
            revision = <"1.1">
            reason = <"Updated HbA1C test result reference">
            change_type = <"Modification">
        >
        ["1.0"] = <
            committer = <"Enrico Barrios">
            committer_organisation = <"AIHW.org.au">
            time_committed = <2004-09-14 16:05:00+1000>
            revision = <"1.0">
            reason = <"Initial Writing">
            change_type = <"Creation">
        >
    >
```

9 Specialisation

9.1 Overview

Archetypes can be specialised in a similar way to classes in object-oriented programming languages. Common to both situations is the use of a *differential* style of declaration, i.e. the contents of a specialised entity are expressed as differences with respect to the parent - previously defined elements from the parent that are not changed are not repeated in the descendant. Two extra constructs are included in the ADL syntax to support redefinition in specialised archetypes.

The basic test that must be satisfied by a specialised archetype is as follows:

- All possible data instance arrangements that conform to the specialised archetype must also conform to all of its parents, recursively to the ultimate parent.

This condition ensures that data created by a specialised archetype that is not itself shared by two systems can be processed by the use of a more general parent that is shared.

The semantics that allow this are similar to the 'covariant redefinition'[1] notion used in some object-oriented programming languages, and can be summarised as follows.

- A non-specialised (i.e. top-level) archetype defines an instance space that is a subset of the space defined by the class in the reference information model on which the archetype is based.
- A specialised archetype can specialise only one parent archetype, i.e. single inheritance.
- A specialised archetype defines an instance space defining the following elements:
  - unchanged object and attribute constraints *inherited* from the parent archetype;
  - and one or more:
    - *redefined* object constraints, that are proper subsets of the corresponding parent object constraints;
    - *redefined* attribute constraints, that are proper subsets of the corresponding parent attribute constraints;
    - *extensions*, i.e. object constraints added to a container attribute with respect to the corresponding attribute in the parent archetype, but only as allowed by the underlying reference model.
- All elements defined in a parent archetype are either inherited unchanged or redefined in a specialised child.
- Specialised archetypes are expressed *differentially* with respect to the parent, i.e. they do not mention purely inherited elements, only redefinitions and extensions.
- Extensions always define an additional subset of the instance space defined by the reference model element being extended (i.e. to which the 'new' objects belong). The extension capability allows archetypes to remain extensible without having to know in advance how or if they will be extended.

The following sections describe the details of specialisation. The term 'object' is used synonymously with 'object constraint' since all elements in ADL are constraints.

1. see http://en.wikipedia.org/wiki/Covariance_and_contravariance_(computer_science)

9.2 Examples

The examples below provide a basis for understanding most of the semantics discussed in the subsequent sections.

9.2.1 Redefinition for Specialisation

The example shown in FIGURE 6 illustrates redefinition in a specialised archetype. The first text is taken from the definition section of the 'laboratory result' `OBSERVATION` archetype on *open*EHR.org[1], and contains an `ELEMENT` node whose identifier is `[id13]`, defined as 'panel item' in the archetype ontology (sibling nodes are not shown here). The intention is that the id13 node be specialised into particular 'panel items' or analytes according to particular types of test result. Accordingly, the id13 node has occurrences of 0..* and its value is not constrained with respect to the reference model, meaning that the type of the *value* attribute can be any descendant of `DATA_VALUE`.

The second text is a specialised version of the laboratory result archetype, defining 'thyroid function test result'. The redefinitions include:

- a redefinition of the top-level object node identifier `[id1]`, with the specialised node identifier `[id1.1]`;
- eight nodes redefining the `[id13]` node are shown, with overridden node identifiers `[id13.2]` - `[id13.9]`;
- reduced occurrences (0..1 in each case);
- redefinition of the *value* attribute of each `ELEMENT` type to `DV_QUANTITY`, shown in expanded form for node `[id13.2]` (achieved by an inline ODIN section; see section 10.1.2 on page 106).

This archetype is typical of a class of specialisations that use only redefinition, due to the fact that all objects in the redefined part of the specialised version are semantically specific kinds of a general object, in this case, 'panel item'.

9.2.2 Redefinition for Refinement

The example shown in FIGURE 7 is taken from the *open*EHR 'Problem' archetype[2] lineage and illustrates the use of redefinition and extension. The first text is the the definition section of the top-level 'Problem' archetype, and shows one `ELEMENT` node in expanded form, with the remaining nodes in an elided form.

The second text is from the 'problem-diagnosis' archetype, i.e. a 'diagnosis' specialisation of the general notion of 'problem'. In this situation, the node `[id2]`, with occurrences of 1, i.e. mandatory non-multiple, has its meaning narrowed to `[id2.1]` 'diagnosis' (diagnosed problems are seen as a subset of all problems in medicine), while new sibling nodes are added to the *items* attribute to define details particular to recording a diagnosis. The extension nodes are identified by the codes `[at0.32]`, `[at0.35]` and `[at0.37]`, with the latter two shown in elided form.

1. http://www.openehr.org/svn/knowledge/archetypes/dev/html/en/openEHR-EHR-OBSERVATION.laboratory.v1.html
2. http://svn.openehr.org/knowledge/archetypes/dev/html/en/openEHR-EHR-EVALUATION.problem.v1.html

```
------ openEHR-EHR-OBSERVATION.laboratory.v1 ------
OBSERVATION[id1] ∈ {-- Laboratory Result
     data ∈ {
         HISTORY[id2] ∈ {
             events ∈ {
                 EVENT[id3] ∈ {-- Any event
                     data ∈ {
                         ITEM_TREE[id4] ∈ {
                             items cardinality ∈ {0..*; unordered} ∈ {
                                 CLUSTER[id5] occurrences ∈ {1} ∈ {..}          -- Specimen
                                 ELEMENT[id8] occurrences ∈ {0..1} ∈ {..}       -- Diagnostic services
                                 CLUSTER[id11] occurrences ∈ {0..*} ∈ {..}      -- level 1
                                 ELEMENT[id13] occurrences ∈ {0..*} ∈ {         -- panel item
                                     value ∈ {*}
                                 }
                                 ELEMENT[id17] occurrences ∈ {0..1} ∈ {..}      -- Overall Comment
                                 CLUSTER[id18] occurrences ∈ {0..1} ∈ {..}      -- Quality
                                 ELEMENT[id37] occurrences ∈ {0..1} ∈ {..}      -- Multimedia rep.
redefinition                     }
                             }
}    }   }   }   }   }   }

------ openEHR-EHR-OBSERVATION.laboratory-thyroid.v1 ------
OBSERVATION[id1.1]          -- Thyroid function tests
     /data[id2]/events[id3]/data[id4]/items ∈ {
         ELEMENT[id13.2] occurrences ∈ {0..1} ∈ {-- TSH
             value ∈ {
                 DV_QUANTITY[id0.7] ∈ {
                     property ∈ {[openehr::119]}
                     magnitude ∈ {|0.0..100.0|}
                     units ∈ {"mIU/l"}
                 }
             }
         }
         ELEMENT[id13.7] occurrences ∈ {0..1} ∈ {..}          -- Free Triiodothyronine (Free T3)
         ELEMENT[id13.8] occurrences ∈ {0..1} ∈ {..}          -- Total Triiodothyronine (Total T3)
         ELEMENT[id13.3] occurrences ∈ {0..1} ∈ {..}          -- Free thyroxine (Free T4)
         ELEMENT[id13.4] occurrences ∈ {0..1} ∈ {..}          -- Total Thyroxine (Total T4)
         ELEMENT[id13.5] occurrences ∈ {0..1} ∈ {..}          -- T4 loaded uptake
         ELEMENT[id13.9] occurrences ∈ {0..1} ∈ {..}          -- Free Triiodothyronine index (Free T3 index)
         ELEMENT[id13.6] occurrences ∈ {0..1} ∈ {..}          -- Free thyroxine index (FTI)
     }
}
```

FIGURE 6 Specialised archetype showing redefinition to multiple children

```
------ openEHR-EHR-EVALUATION.problem.v1 ------
EVALUATION[id1] ∈ {-- Problem
     data ∈ {
        ITEM_TREE[id2] ∈ {
           items cardinality ∈ {0..*; ordered} ∈ {
              ELEMENT[id3] occurrences ∈ {1} ∈ {-- Problem
                 value ∈ {
                    DV_TEXT[id4] ∈ {*}
redefinition     }
              }
              ELEMENT[id5] occurrences ∈ {0..1} ∈ {..}-- Date of initial onset
              ELEMENT[id6] occurrences ∈ {0..1} ∈ {..}-- Age at initial onset
              ELEMENT[id7] occurrences ∈ {0..1} ∈ {..}-- Severity
              ELEMENT[id8] occurrences ∈ {0..1} ∈ {..}-- Clinical description
              ELEMENT[id10] occurrences ∈ {0..1} ∈ {..}-- Date clinically received
              CLUSTER[id11] occurrences ∈ {0..*} ∈ {..}-- Location
              CLUSTER[id14] occurrences ∈ {0..1} ∈ {..}-- Aetiology
              -- etc
}    }   }   }
```

```
------ openEHR-EHR-EVALUATION.problem-diagnosis.v1 ------
EVALUATION[id1.1] ∈ {-- Recording of diagnosis
     /data[id2]/items[id3]/value ∈ {
        DV_CODED_TEXT[id4] ∈ {
           defining_code ∈ {[ac0.1]} -- X 'is_a' diagnosis
        }
     }

     /data/items cardinality ∈ {0..*; ordered} ∈ {
        before [id5]
        ELEMENT[at0.32] occurrences ∈ {0..1} ∈ { -- Status
           value ∈ {
              DV_CODED_TEXT[at0.33] ∈ {
                 defining_code ∈ {
                    [local::at0.33, at0.34] -- provisional / working     extension
                 }
              }
           }
        }
        after [id31]
        CLUSTER[at0.35] occurrences ∈ {0..1} ∈ {..} -- Diag. criteria
        CLUSTER[at0.37] occurrences ∈ {0..1} ∈ {..} -- Clin. staging
     }
}
```

FIGURE 7 Specialised archetype showing redefinition and extension

9.3 Specialisation Concepts

9.3.1 Differential and Flat Forms

Specialised archetypes in their authored form are represented in ‘differential’ form. The syntax is the same as for non-specialised archetypes, with two additions: specialisation paths (see section 9.3.3) and ordering indicators (see Ordering of Sibling Nodes on page 89). For a specialised archetype therefore, the lineage of archetypes back to the ultimate parent must be taken into account in order to obtain its complete semantics.

Differential form means that the only attributes or objects mentioned are those that redefine corresponding elements in the parent and those that introduce new elements. The differential approach to representation of specialised archetypes give rise to the need for a *flat form* of a specialised archetype: the equivalent archetype defined by the sum of the (differential) child and its parent, as if the child archetype had been defined standalone. The flat form of archetypes is used for building templates, and subsequently at runtime. It is generated by ‘compressing’ the effects of inheritance of the parent to the specialised child into a single archetype, and applies recursively all the way up an *archetype lineage* to the ultimate parent, which must be a top-level (non-specialised) archetype. For a top-level archetype, the flat-form is the same as its differential form (i.e. in a top-level archetype, every node is considered to be an extension node).

9.3.2 Specialisation Levels

In order to talk about archetypes at different levels of specialisation, a standard way of identifying the levels of specialisation is used, as follows:

- level 0: top-level, non-specialised archetypes
- level 1: specialisations of level 0 archetypes
- level 2: specialisations of level 1 archetypes
- etc.

For nodes carrying a node identifier, the specialisation level is always equal to the number of ‘.’ characters found in the identifier.

9.3.3 Specialisation Paths

Because ADL is a block-structured language, the redefinition of nodes deep in the parent structure normally requires descending into the structure. Since it is common to want to further constrain only nodes deep within a structure in specialised archetype, a more convenient way is provided in ADL to do this using a *specialisation path*, illustrated by the following example:

```
OBSERVATION[id1.1] ∈ {-- Thyroid function tests
    /data[id2]/events[id3]/data[id4]/items ∈ {
        ELEMENT[id13.2] occurrences ∈ {0..1} ∈ {-- TSH
            value ∈ {
                DV_QUANTITY[id0.7] ∈ { ... }
            }
        }
        ELEMENT[id13.7] occurrences ∈ {0..1} ∈ {..}-- Free T3
        ....
    }
  }
}
```

In this fragment, a path is used rather than an attribute name. A path can be used in this manner only if no further constraints are required 'on the way' into the deep structure.

The rules for specialisation paths are as follows.

- A specialisation path is constructed down to the first attribute having any child objects to be further constrained in the present archetype.
- All path segments must carry an id-code predicate.
- The shortest useful path that can be used is '/' followed by an attribute name from the top level class being constrained by the archetype.

9.3.4 Path Congruence

Any node in an archetype can unambiguously be located by its archetype path. For example, the text value of the 'problem' node of the `openEHR-EHR-EVALUATION.problem.v1` archetype shown at the top of FIGURE 7 is:

```
/data[id2]/items[id3]/value
```

Similarly the path to the redefined version of the same node in the `openEHR-EHR-EVALUATION.problem-diagnosis.v1` archetype at the bottom of the same figure is:

```
/data[id2]/items[id3.1]/value
```

By inspection, it can be seen that this path is a variant of the corresponding path in the parent archetype, where a particular object node identifier has been specialised.

In general, the path of every redefined node in a specialised archetype will have a direct equivalent in the parent archetype, which can be determined by removing one level of specialisation from any node identifiers within the specialised path *that are at the level of specialisation of the specialised archetype* (i.e. node identifiers corresponding to higher specialisation levels are not changed). In this way, the nodes in a specialised archetype source can be connected to their counterparts in parent archetypes, for purposes of validation and flattening.

Conversely, any given path in an archetype that has children will have congruent paths in the children wherever nodes have been specialised.

9.3.5 Redefinition Concepts

A specialised archetype definition section is expressed in terms of:

- *redefined* object constraints;
- *redefined* attribute constraints;
- *extensions*, being object constraints added to container attributes.

Apart from *existence* and *cardinality* redefinition, all specialisation relates to object constraints. In the ADL syntax, objects can be specified in two places: under single-value attributes and under multiply-valued (container) attributes.

Each object under a single-valued attribute defines an alternative that may be used to constrain data at that attribute position. An example is the `OBSERVATION`.*protocol* attribute from the *open*EHR reference model: if multiple objects appear under this attribute, only one can be used at runtime to constrain data.

Within a container attribute, the meaning of multiple objects is that each child object defines constraints on one or more members of the container in the data. The *occurrences* constraint on each one determines how many objects in the data match a given object constraint in the attribute.

Object constraints can be specialised in both places by redefinition, refinement and exclusion. In addition, extension can be used under container attributes. The actual semantics are described in terms of object node identification, type redefinition, and structural constraints (existence, cardinality and occurrences), and are mostly the same for objects under single- and multiply-valued attributes. The following sections describe the details.

9.4 Attribute Redefinition

A small number of things can be redefined on attributes, including existence and cardinality. A basic rule of redefinition is that a specialised archetype cannot change the multiplicity type of an attribute.

9.4.1 Existence Redefinition and Exclusion

All attributes mentioned in an archetype have an *existence* constraint, indicating whether a value is required or not. The constraint is either stated explicitly - typically done for single-valued attirbutes - or it is the value from the reference model - typical for multiply-valued attributes. In both cases, the existence of an attribute in a parent archetype can be redefined in a specialised archetype using the standard cADL syntax. In the following example, an implicit existence constraint picked up from the reference model of {0..1} is redefined in a child archetype to {1}, i.e. mandatory.

Parent archetype:

```
OBSERVATION[id1] ∈ { -- blood pressure measurement
    protocol ∈ { -- existence not changed from reference model
       -- etc
    }
}
```

Child archetype:

```
OBSERVATION[id1.1] ∈ { -- paediatric blood pressure measurement
    /protocol existence ∈ {1} ∈ {
       -- etc
    }
}
```

Redefinition of existence to {0} by this method denotes exclusion, i.e. removal of the entire attribute (including all sub-structure) from the resulting structure. In an archetype, it is likely to indicate poor design, given that the decision to remove optional attributes is much more likely to be local, and therefore more appropriate in templates rather than archetypes; within a template it would be perfectly normal. The following example shows the *protocol* attribute in the above OBSERVATION archetype being excluded in this way:

```
OBSERVATION[id1] ∈ { -- paediatric blood pressure measurement
    protocol existence ∈ {0}
}
```

9.4.2 Container Attributes

The following sub-sections describe specialisation semantics specific to container attributes.

9.4.2.1 Cardinality

The *cardinality* constraint defines how many object instances can be in the container within the data (not the archetype). In a specialised archetype, cardinality can be redefined to be a narrower range than in the parent, further limiting the valid ranges of items in the data that may occur within the container. This would normally only make sense if refinements were made to the *occurrences* of the contained items, i.e.:

- narrowing the occurrences range of an object;
- excluding an object by setting its occurrences to {0};
- adding new objects, which themselves will have occurrences constraints;
- setting some object occurrences to mandatory, and the enclosing cardinality lower limit to some non-zero value.

As long as the relationship between the enclosing attribute's cardinality constraint and the occurrences constraints defined on all the contained items (including those inherited unchanged, and therefore not mentioned in the specialised archetype) is respected (see VCOC validity rule, AOM specification), any of the above specialisations can occur.

The following provides an example of cardinality redefinition.

Parent archetype:

```
ITEM_LIST[id3] ∈ { -- general check list
    items cardinality ∈ {0..*} ∈ { -- any number of items
        ELEMENT[id12] occurrences ∈ {0..*} ∈ {} -- generic checklist item
    }
}
```

Child archetype:

```
ITEM_LIST[id3] ∈ { -- pre-operative check list
    /items cardinality ∈ {3..10} ∈ { -- at least 3 mandatory items
        ELEMENT[id12.1] occurrences ∈ {1} ∈ {} -- item #1
        ELEMENT[id12.2] occurrences ∈ {1} ∈ {} -- item #2
        ELEMENT[id12.3] occurrences ∈ {1} ∈ {} -- item #3
        ELEMENT[id12.4] occurrences ∈ {0..1} ∈ {} -- item #4
        ...
        ELEMENT[id12.10] occurrences ∈ {0..1} ∈ {} -- item #10
    }
}
```

9.4.2.2 Ordering of Sibling Nodes

Within container attributes, the order of objects may be significant from the point of view of domain users, i.e. the container may be considered as an ordered list. This is easy to achieve in top-level archetype, using the 'ordered' qualifier on a cardinality constraint. However when particular node(s) are redefined into multiple specialised nodes, or new nodes added by extension, the desired order of the new nodes may be such that they should occur interspersed at particular locations among nodes defined in the parent archetype. The following text is a slightly summarised view of the *items* attribute from the problem archetype shown in FIGURE 7:

```
items cardinality ∈ {0..*; ordered} ∈ {
    ELEMENT[id2] occurrences ∈ {1} ∈ {..}        -- Problem
    ELEMENT[id3] occurrences ∈ {0..1} ∈ {..}     -- Date of initial onset
    ELEMENT[id4] occurrences ∈ {0..1} ∈ {..}     -- Age at initial onset
    ELEMENT[id5] occurrences ∈ {0..1} ∈ {..}     -- Severity
    ELEMENT[id9] occurrences ∈ {0..1} ∈ {..}     -- Clinical description
    ELEMENT[id10] occurrences ∈ {0..1} ∈ {..}    -- Date clinically received
    CLUSTER[id11] occurrences ∈ {0..*} ∈ {..}    -- Location
    CLUSTER[id14] occurrences ∈ {0..1} ∈ {..}    -- Aetiology
    CLUSTER[id18] occurrences ∈ {0..1} ∈ {..}    -- Occurrences or exacerb'ns
    CLUSTER[id26] occurrences ∈ {0..1} ∈ {..}    -- Related problems
    ELEMENT[id30] occurrences ∈ {0..1} ∈ {..}    -- Date of resolution
    ELEMENT[id31] occurrences ∈ {0..1} ∈ {..}    -- Age at resolution
}
```

To indicate significant ordering in the specialised problem-diagnosis archetype, the keywords `before` and `after` can be used, as follows:

```
/data[id3]/items ∈ {
    before [id3]
    ELEMENT[id2.1] ∈ {..}                          -- Diagnosis
    ELEMENT[at0.32] occurrences ∈ {0..1} ∈ {..}  -- Status
    after [id26]
    CLUSTER[at0.35] occurrences ∈ {0..1} ∈ {..} -- Diagnostic criteria
    CLUSTER[at0.37] occurrences ∈ {0..1} ∈ {..} -- Clinical Staging
}
```

These keywords are followed by a node identifier reference, and act to modify the node definition immediately following. Technically the following visual rendition would be more faithful, but it is less readable, and makes no difference to a parser:

```
    after [id26] CLUSTER[at0.35] occurrences ∈ {0..1} ∈ {..} -- etc
```

The rules for specifying ordering are as follows.

- Ordering is only applicable to object nodes defined within a multiply-valued attribute whose cardinality includes the `ordered` constraint;
- Any `before` or `after` statement can refer to the node identifier of any sibling node known in the flat form of the archetype, i.e.:
  - the identifier of any redefined node;
  - the identifier of any new node;
  - the identifier of any inherited node that is not redefined amongst the sibling nodes.
- If no ordering indications are given, redefined nodes should appear in the same position as the nodes they replace, while extension nodes should appear at the end.

If ordering indicators are used in an archetype that is itself further specialised, the following rules apply:

- If the referenced identifier becomes unavailable due to being redefined in the new archetype, it must be redefined to refer to an available sibling identifier as per the rules above.
- If this does not occur, a `before` reference will default to the *first* sibling node identifier currently available conforming to the original identifier, while an after reference will default to the *last* such identifier available in the current flat archetype.

TBD_6: should we also introduce 'first' and 'last' keywords to allow new nodes to always be forced to top or bottom of the list, regardless of what else is there?

If, due to multiple levels of redefinition, there is more than one candidate to go before (or after) a given node, the compiler should output a warning. The problem would be resolved by the choice of one of the candidates being changed to indicate that it is to be ordered before (after) another of the candidates rather than the originally stated node.

9.5 Object Redefinition

Object redefinition can occur for any object constraint in the parent archeype, and can include redefinition of node identifier, occurrences, reference model type. For certain kinds of object constraints, specific kinds of redefinition are possible.

9.5.1 Node Identifiers

In an archetype, node identifiers ('at-codes') are mandatory on all object constraints defined as children of a multiply-valued attribute and multiple same-typed children of single-valued attributes (see Node Identifiers on page 29). They are optional on other single child constraints of single-valued attributes. This rule applies in specialised as well as top-level archetypes.

Redefinition semantics for node identifiers are somewhat special. They can be redefined for purely semantic purposes (e.g. to redefine 'heart rate' to 'fetal heart rate') - this is what happens on the root node of every archetype. However, if any other aspect of the *immediate* object node (i.e. the one to which the node identifier is attached), i.e. occurrences, reference model type, or the constraint type is changed, the node identifier *must be redefined*.

The obvious alternative would have been to require that every identified node from the point of any redefinition (or extension or exclusion) to the top of the structure be deemed as redefined, and therefore carry a specialised node identifier, where appropriate. This approach is not used for two reasons. Firstly, the presence of any specialised node in the hierarchy from a leaf node to the root creates a specialised path particular to the leaf node, which can be used to identify data created by the specialised archetype. Secondly, the meaning of nodes above the changed node do not change as such, and creating specialised node identifiers would require the definition of superfluous codes in the archetype ontology.

Redefinition of an object node identifier for purely semantic purposes, unaccompanied by any other kind of constraint change is done as shown in the following example.

Parent archetype:

```
EVALUATION[id1] ∈ {       -- Medical Certificate
    data ∈ {
        ITEM_TREE[id2] ∈ {
            items ∈ {
                ELEMENT[id5] occurrences ∈ {0..1} ∈ {-- Description
                    value ∈ {
                        DV_TEXT[id7] ∈ {*}
                    }
                }
            }
        }
    }
}
```

Child archetype:

```
EVALUATION[id1.1] ∈ {           -- Singapore Medical Certificate
    /data[id2]/items ∈ {
        ELEMENT[id5.1]          -- Summary
    }
}
```

Here the `id5` ('Description') node is refined in meaning to `id5.1` ('Summary'). Since there is no other constraint to be stated, no further 'matches' block is required.

9.5.1.1 Node Redefinition Semantics

Node identifiers can be redefined in specialised archetypes. This is done.

TBD_7: if occurrences rule relaxed, then the same node id is ok

As a consequence, this means that the identifier of the root node of a specialised archetype will always be redefined, indicating the more specific 'type' of instances it defines. The first example above defines a 'thyroid function test' `[id1]` subset of 'laboratory result' `[id1.1]` `OBSERVATION` objects.

Specific redefinitions and extensions occur at lower points in the structure. In the first example above, the node `[id13]` is redefined into a number of more specialised nodes `[id13.2]` - `[id13.9]`, while in the second, the identifier `[id2]` is redefined to a single node `[id2.1]`.

The syntactic form of the identifier of a redefined node is a copy of the original followed by a dot ('.'), optionally intervening instances of the pattern '0.' and then a further non-zero number, i.e.:

- `atNNNN {.0}* .N`

This permits node identifiers from a given level to be redefined not just at the next level, but at multiple levels below.

Examples of redefined node identifiers:

- `id2.1` -- redefinition of `id1` at level 1 specialisation
- `id2.0.1` -- redefinition of `id1` node in level 2 specialisation archetype
- `id2.1.1` -- redefinition of `id2.1` in level 2 specialisation archetype.

Redefined versions of nodes with no node id in the parent archetype do not require a node identifier in the child archetype.

In both cases, there is a question of whether the original node being redefined (id13 and id2 respectively in the examples) remains available for further redefinition in subsequent child archetypes, or do the redefinition children *exhaustively* define the instance space for the given parent node?

> *Discussion:* This question can be considered in terms of ontological design. A full discussion is outside the scope of this document, but a couple of points can be made here. Firstly, the question of whether an exhaustive set of children can be defined will always be on a case-by-case basis, because the choice of children can be made to be exhaustive by a) definitional or b) perfect knowledge. To understand the first case, consider an ontology with a category 'hepatitis' with children 'hepatitis A', 'hepatitis B' and 'hepatitis non-A non-B'. The first two of these designate distinct phenomona in the real world (specific viruses which can be tested for) while the third is a classification that could be attached to anyone with another kind of hepatitis. The last category is a typical catch-all category designed to cover all cases not covered by the specifically known subtypes of the parent category. Should these children be considered exhaustive? One point of view says so, since all subsequently discovered varieties of hepratitis (C, D, E, etc) would now become children of 'hepatitis non-A non-B'. However this is likely to be sub-optimal, since now the category 'hepatitis non-A non-B' probably exists solely because of the order in which the various hepatitis virus tests were perfected. Therefore an alternative argument would say that the categories 'hepatitis C', 'hepatitis D' etc should be defined directly below 'hepatitis', as if 'hepatitis non-A non-B' had never existed. Under this argument, the children would not be declared, even when they are theoretically exhaustive.
>
> This kind of argument comes up time and again, and the need for catch-all categories (archetype nodes) and the possibility of future discoveries cannot be predicted. Even in situations such as a lab result (e.g. cholesterol), where the list of analytes seem to be known and fixed, experience of clinical modellers has shown that there is nevertheless no guarantee of not needing another data point, perhaps for something other than an analyte.

The default situation is that they do not, unless explicitly reversed, by marking the parent node 'closed' for further specialisation in the child. The keyword `closed` is used to do this. The first example would then become:

Parent archetype:

```
items cardinality ∈ {0..*; unordered} ∈ {
    CLUSTER[id4] occurrences ∈ {1} ∈ {..}             -- Specimen
    ELEMENT[id8] occurrences ∈ {0..1} ∈ {..}          -- Diagnostic services
    CLUSTER[id11] occurrences ∈ {0..*} ∈ {..}         -- level 1
    ELEMENT[id13] occurrences ∈ {0..*} ∈ {            -- panel item
        value ∈ {*}
    }
    ELEMENT[id17] occurrences ∈ {0..1} ∈ {..}         -- Overall Comment
    CLUSTER[id18] occurrences ∈ {0..1} ∈ {..}         -- Quality
    ELEMENT[id37] occurrences ∈ {0..1} ∈ {..}         -- Multimedia rep.
    }
}
```

Child archetype:

```
/data/events[id2]/data/items ∈ {
    ELEMENT[id13.2] occurrences ∈ {0..1} ∈ {..}       -- TSH
    ELEMENT[id13.7] occurrences ∈ {0..1} ∈ {..}       -- Free Triiodothyronine
    ELEMENT[id13.8] occurrences ∈ {0..1} ∈ {..}       -- Total Triiodothyronine
    ELEMENT[id13.3] occurrences ∈ {0..1} ∈ {..}       -- Free thyroxine (Free T4)
    ELEMENT[id13.4] occurrences ∈ {0..1} ∈ {..}       -- Total Thyroxine (Total T4)
    ELEMENT[id13.5] occurrences ∈ {0..1} ∈ {..}       -- T4 loaded uptake
    ELEMENT[id13.9] occurrences ∈ {0..1} ∈ {..}       -- Free Triiodothyronine index
    ELEMENT[id13.6] occurrences ∈ {0..1} ∈ {..}       -- Free thyroxine index (FTI)
    ELEMENT[id13] closed
}
```

Without the above specification, a further child archetype could then redefine both the original id13 node (e.g. into id13.0.1, id13.0.2), and any of the id13 nodes (e.g. id13.1.1, id13.1.2); with it, only the latter is possible.

9.5.1.2 Extension Nodes

Extension nodes carry node identifiers according to the rule mentioned above. The second example includes the new node identifiers `[at0.32]`, `[at0.35]` and `[at0.37]`, whose codes start with a '0'. indicating that they have no equivalent code in the parent archetype.

The node identifier syntax of an extension node commences with at least one instance of the pattern '0.'. The structure of node identifiers for both kinds of node thus always indicates at what level the identifier was introduced, given by the number of dots.

Examples of redefined node identifiers:

- `at0.1` -- identifier of extension node introduced at level 1
- `at0.0.1` -- identifier of extension node introduced at level 2

When a flat form is created, the level at which any given node was introduced or redefined is clear due to the identifier coding system.

TBD_8: to be properly formal, all containers should be declared open or closed in archetypes; only open ones should be extensible. Currently we are assuming that all containers are 'open'.

9.5.2 Occurrences Redefinition and Exclusion

The `occurrences` constraint on an object node indicates how many instances within the data may conform to that constraint (see Container Attributes on page 28). If occurrences is redefined on an identified node, the node identifier must be specialised. Within container attributes, `occurrences` is usually redefined in order to make a given object mandatory rather than optional; it can also be used to exclude an object constraint. In the following example, the occurrences of the `[id3]` node is redefined from {0..1} i.e. optional, to {1}, i.e. mandatory.

Parent (`openEHR-EHR-EVALUATION.problem.v1`):

```
EVALUATION[id1] ∈ {-- Problem
     data ∈ {
         ITEM_TREE[id2] ∈ {
             items cardinality ∈ {0..*; ordered} ∈ {
                 ELEMENT[id3] occurrences ∈ {1} ∈ {..}-- Problem
                 ELEMENT[id4] occurrences ∈ {0..1} ∈ {..}-- Date of initial onset
                 -- etc
             }
         }
     }
}
```

Child (`openEHR-EHR-EVALUATION.problem-diagnosis.v1`):

```
/data[id2]/items ∈ {
     ELEMENT[id4] occurrences ∈ {1} -- Date of initial onset
}
```

In the above we can see that if the only change in the redefinition is to occurrences, the remainder of the block from the parent is not repeated in the child. Occurrences is normally only constrained on child objects of container attributes, but can be set on objects of any attribute to effect exclusion of part of the instance space. This can be useful in archetypes where a number of alternatives for a single-valued attribute have been stated, and the need is to remove some alternatives in a specialised child archetype. For example, an archetype might have the following constraint:

```
ELEMENT[id3] ∈ {
     value ∈ {
         DV_QUANTITY[id4] ∈ {*}
         DV_INTERVAL<DV_QUANTITY>[id5] ∈ {*}
         DV_COUNT[id6] ∈ {*}
         DV_INTERVAL<DV_COUNT>[id7] ∈ {*}
     }
}
```

and the intention is to remove the `DV_INTERVAL<*>` alternatives. This is achieved by redefining the enclosing object to removed the relevant types:

```
ELEMENT[id3] ∈ {
     value ∈ {
         DV_INTERVAL<DV_QUANTITY>[id4] occurrences ∈ {0}
         DV_INTERVAL<DV_COUNT>[id7] occurrences ∈ {0}
     }
}
```

Exclusion by setting occurrences to {0} is also common in templates, and is used to remove specific child objects of container attributes, as in the following example:

```
/data[id2]/items ∈ {
     CLUSTER[id26] occurrences ∈ {0} -- remove 'Related problems'
     ELEMENT[id31] occurrences ∈ {0} -- remove 'Age at resolution'
}
```

If the whole attribute is to be removed, this can be done by redefining existence to {0}, as described in Existence Redefinition and Exclusion on page 88.

9.5.3 Reference Model Type Refinement

The type of an object may be redefined to one of its subtypes as defined by the reference model. A typical example of where this occurs in archetypes based on the *open*EHR reference model is when `ELEMENT`.*value* is constrained to '*' in a parent archetype, meaning 'no further constraint on its RM type of `DATA_VALUE`', but is then constrained in a specialised archetype to subtypes of `DATA_VALUE`, e.g. `DV_QUANTITY` or `DV_PROPORTION`[1]. The following figure containts a simplified extract of the data values part of the *open*EHR reference model, and is the basis for the examples below.



FIGURE 8 Example reference model type structure

The most basic form of type refinement is shown in the following example:

Parent archetype:

```
value ∈ {*} -- any subtype of DATA_VALUE, from the ref model
```

Specialised archetype:

```
.../value ∈ {
    DV_QUANTITY[id8] ∈ {*} -- now limit to the DV_QUANTITY subtype
}
```

The meaning of the above is that instance data constrained by the specialised archetype at the value node must match the `DV_QUANTITY` constraint only - no other subtype of `DATA_VALUE` is allowed.

When a type in an archetype is redefined into one of its subtypes, any existing constraints on the original type in the parent archetype are respected. In the following example, a `DV_AMOUNT` constraint that required *accuracy* to be present and in the range +/-5% is refined into a `DV_QUANTITY` in which two attributes of the subtype are constrained. The original *accuracy* attribute is inherited without change.

Parent archetype:

```
value ∈ {
    DV_AMOUNT[id4] ∈ {
        accuracy ∈ {|-0.05..0.05|}
    }
}
```

Specialised archetype:

```
.../value ∈ {
    DV_QUANTITY[id4] ∈ {
        magnitude ∈ {|2.0..10.0|}
```

1. See the *open*EHR data types specification at http://www.openehr.org/releases/1.0.1/architecture/rm/data_types_im.pdf for details.

```
            units ∈ {"mmol/ml"}
        }
    }
```

In the same manner, an object node can be specialised into more than one subtype, where each such constraint selects a mutually exclusive subset of the instance space. The following example shows a specialisation of the DV_AMOUNT constraint above into two subtyped constraints.

```
    .../value ∈ {
        DV_QUANTITY[id4] ∈ {
            magnitude ∈ {|2.0..10.0|}
            units ∈ {"mmol/ml"}
        }
        DV_PROPORTION[id5] ∈ {
            numerator ∈ {|2.0..10.0|}
            type ∈ {pk_unitary}
        }
    }
```

Here, instance data may only be of type DV_QUANTITY or DV_PROPORTION, and must satisfy the respective constraints for those types.

A final variant of subtyping is when the intention is to constraint the data to a supertype with exceptions for particular subtypes. In this case, constraints based on subtypes are matched first, with the constraint based on the parent type being used to constrain all other subtypes. The following example constrains data at the *value* node to be:

- an instance of DV_QUANTITY with *magnitude* within the given range etc;
- an instance of DV_PROPORTION with *numerator* in the given range etc;
- an instance of any other subtype of DV_AMOUNT, with *accuracy* in the given range.

```
    .../value ∈ {
        DV_QUANTITY[id4] ∈ {
            magnitude ∈ {|2.0..10.0|}
            units ∈ {"mmol/ml"}
        }
        DV_PROPORTION[id5] ∈ {
            numerator ∈ {|2.0..10.0|}
            type ∈ {pk_unitary}
        }
        DV_AMOUNT[id6] ∈ {
            accuracy ∈ {|-0.05..0.05|}
        }
    }
```

A typical use of this kind of refinement in *open*EHR would be to add an alternative for a DV_CODED_TEXT constraint for a specific terminology to an existing DV_TEXT constraint in a *name* attribute, as follows:

```
    name ∈ {
        DV_CODED_TEXT[id13] ∈ {
            defining_code ∈ {[Snomed_ct::]}
        }
        DV_TEXT[id14] ∈ {
            value ∈ {/.+/} -- non-empty string
        }
    }
```

All of the above specialisation based on reference model subtypes can be applied in the same way to identified object constraints.

9.5.4 Terminology External Subset Redefinition

A terminology external subset constraint is used to set the value set of a coded term to be one defined externally in a terminology, specified in the `constraint_definitions` sub-section of the `terminology` section, as shown in the following example.

```
definition
    ELEMENT [id13] ∈ { -- cuff size
        value ∈ {
            DV_CODED_TEXT[id4] ∈ {
                defining_code ∈ {[local::ac1]}
            }
        }
    }
terminology
    term_bindings = <
        ["snomed_ct"] = <
            items = <
                ["ac1"] = <http://terminology.org?query_id=12345>
            >
        >
    >
```

In a specialisation of the archetype, the placeholder constraint can be redefined in two different ways. The first is by redefinition of the placeholder constraint to a narrower one. This is a achieved by redefining the constraint code, and adding a new definition in the ontology of the specialised archetype, as follows.

```
definition
    ELEMENT [id13] ∈ { -- cuff size
        value ∈ {
            DV_CODED_TEXT[id14] ∈ {
                defining_code ∈ {[local::ac1.1]}
            }
        }
    }
terminology
    term_bindings = <
        ["snomed_ct"] = <
            items = <
                ["ac1.1"] = <http://terminology.org?query_id=12345-1>
            >
        >
    >
```

The second kind of redefinition is by an inline constraint of the same primitive ADL type `TERMINOLOGY_CODE`.

```
ELEMENT [id13] ∈ { -- cuff size
    value ∈ {
        DV_CODED_TEXT[id14] ∈ {
            defining_code ∈ {
                [local::
                at22, -- child cuff
                at23] -- infant cuff
            }
        }
    }
```

```
}
```

These redefinitions are assumed to be valid, although it is not directly validatable unless the terminology subset is available to the tooling.

A third variation on the same semantics is when a term constraint is used as a redefinition of a previously unconstrained term code, e.g. as illustrated by the following fragment.

```
ELEMENT[id13] ∈ { -- cuff size
   value ∈ {
      DV_CODED_TEXT[id14] ∈ {*}
   }
}
```

9.5.5 Internal Reference (Proxy Object) Redefinition

An archetype proxy object, or `use_node` constraint is used to refer to an object constraint from a point elsewhere in the archetype. These references can be redefined in two ways, as follows.

- *Target redefinition*: the target constraint of reference may be itself redefined. The meaning for this is that all internal references now assume the redefined form.
- *Reference redefinition*: specialised archetypes can redefine a use_node object into a normal inline concrete constraint that a) replaces the reference, and b) must be completely conformant to the structure which is the target of the original reference.

Note that if the intention is to redefine a structure referred to by `use_node` constraints, but to leave the constraints at the reference source points in form to which the reference points in the parent level, each `use_node` reference needs to be manually redefined as a copy of the target structure originally pointed to.

The second type of redefinition above is the most common, and is shown in the following example.

Parent archetype:

```
ENTRY[id1]∈ {
     data ∈ {
        CLUSTER[id2] ∈ {
           items ∈ {
              -- etc --
           }
        }
        use_node CLUSTER[id3] /data[id2]
     }
}
```

Child archetype:

```
ENTRY [id1.1]∈ {
     /data[id3]/items ∈ {
        ELEMENT [at0.1] ∈ {
           -- etc --
        }
     }
}
```

Remembering that the parent archetype is essentially just definition two sibling object structures with the identifiers id1 and id2 (defined by the use_node reference), the child is redefining the id2 node (it could also have redefined the id1 node as well). The result of this in the flattened output is as follows:

```
ENTRY [id1.1]∈ {
     data ∈ {
```

```
            CLUSTER[id2] ∈ {
                items ∈ {
                    -- etc --
                }
            }
            CLUSTER[id3] ∈ {
                items ∈ {
                    ELEMENT[at0.1] ∈ {
                        -- etc --
                    }
                }
            }
        }
    }
```

There is one subtlety to do with redefinition of occurrences of a use_node target: if it is redefined to have occurrences matches {0} (normally only in a template), then the effect of this is the same on any use_node reference definitions, unless they define occurrences locally at the reference point. The chance of this actually occurring appears vaninshingly small, since by the time 'exclusion' occurrence redefinition is being done in templates, use_node object definitions are most likely to have been locally overridden anyway.

Lastly, one further type of redefinition appears technically possible, but seems of no utility, and is therefore not part of ADL:

- *Reference re-targetting*: an internal reference could potentially be redefined into a reference to a different target whose structure conforms to the original target.

9.5.6 Slot Redefinition

A slot within an archetype can be specialised by any combination of the following:

- one or more slot-fillers;
- a redefinition of the slot itself, either to narrow the set of archetypes it matches, or to close it to filling in either further specialisations, or at runtime.

Both types of redefinition are generally used by templates rather than published archetypes, since the business of filling slots is mostly related to local use-case specific uses of archetypes rather than part of the initial design.

The following example shows a slot from a SECTION archetype for the concept 'history_medical_surgical' archetype.

```
SECTION[id1] ∈ {-- Past history
    items ∈ {
        allow_archetype EVALUATION[id2] ∈ {-- Past problems
            include
                archetype_id/value ∈ {
                    /openEHR-EHR-EVALUATION\.clinical_synopsis\.v1
                        |openEHR-EHR-EVALUATION\.excluded(-[a-z0-9_]+)*\.v1
                        |openEHR-EHR-EVALUATION\.injury\.v1
                        |openEHR-EHR-EVALUATION\.problem(-[a-z0-9_]+)*\.v1/}
        }
    }
}
```

This slot specification allows EVALUATION archetypes for the concepts 'clinical synopsis', various kinds of 'exclusions' and 'problems', and 'injury' to be used, and no others. The following fragment of ADL shows how the slot is filled in a template, using the keywords `use_archetype` and `use_template`. In this syntax, the node identification is a variation on the normal archetype at-codes. Within the template, the identifier of the used archetype is also the identifier of that node. However, the original at-code (if defined) must also be mentioned, so as to indicate *which* slot the used archetype is filling. Templates may also be used to fill slots in the same way. Thus, in the following example, two archetypes and a template are designated to fill the id2 slot defined in the above fragment of ADL. The slot definition is not mentioned, so it remains unchanged, i.e. 'open'.

```
SECTION[id1] ∈ {-- Past history
    /items ∈ {
       use_archetype EVALUATION[id2]
                       org.openehr::openEHR-EHR-EVALUATION.problem.v1
       use_template EVALUATION[id2]
                       uk.nhs.cfh::openEHR-EHR-EVALUATION.t_ed_diagnosis.v1
       use_archetype EVALUATION[id2]
                       org.openehr::openEHR-EHR-EVALUATION.clin_synopsis.v1
    }
}
```

Since node identifiers are only required to disambiguate multiple sibling nodes, they may not exist on all nodes in a typical archetype. It is therefore possible to have a slot that carries no node identifier (e.g. due to being under a single-valued attribute). A `use_archetype` specification within a template will accordingly only mention the archetype identifier, with no node id, as per the following example (archetype followed by a template).

```
ACTIVITY[id1] ∈ {-- Medication activity
    description ∈ {
       allow_archetype ITEM_TREE[id4] ∈ {
          include
             archetype_id/value ∈ {...}
       }
    }
}
```

A template containing a filler for this slot would be as follows:

```
use_archetype INSTRUCTION[openEHR-EHR-INSTRUCTION.medication.v1] ∈ {
    /activities[id1]/description ∈ {
       use_archetype ITEM_TREE[id4]
                       org.openehr::openEHR-EHR-ITEM_TREE.medication.v1
    }
}
```

Slots can be recursively filled in the above fashion, according to the possibilities offered by the chosen archetypes or templates. The following ADL fragment shows two levels of slot-filling:

```
use_archetype COMPOSITION[openEHR-EHR-COMPOSITION.xxx.v1] ∈ {
    /content ∈ {
       use_archetype SECTION[id1]
                          org.openehr::openEHR-EHR-SECTION.yyy.v1    ∈ {
          /items ∈ {
             use_template EVALUATION
                   [id2] uk.nhs.cfh::openEHR-EHR-EVALUATION.t_xx.v1
             use_archetype EVALUATION
                   [id2] org.openehr::openEHR-EHR-EVALUATION.xx.v1
             use_archetype EVALUATION
                   [id3] org.openehr::openEHR-EHR-EVALUATION.xx.v1
          }
```

```
            }
        }
    }
```

Note that in the above the archetype fillers are specified as published archetypes, but in reality, it is far more likely that template-specific specialisations of these archetypes would be used. The identification and organisation of such archetypes is described in the *open*EHR Templates document.

In addition to or instead of specifying slot fillers, it is possible in a slot specialisation to narrow the slot definition, or to close it. If fillers are specified, closing the slot as well is typical. The latter is done by including an overridden version of the archetype slot object itself, with the 'closed' constraint set, as in the following example:

```
use_archetype SECTION[org.openehr::openEHR-EHR-
                                        SECTION.history_medical_surgical.v1] ∈ {
    /items ∈ {
        use_archetype EVALUATION[id2] openEHR-EHR-EVALUATION.problem.v1
        allow_archetype EVALUATION[id2] closed
    }
}
```

Narrowing the slot is done with a replacement `allow_archetype` statement containing a narrowed set of match criteria.

9.5.7 'Filling' Unconstrained Attributes

The `use_archetype` keyword can be used to specify a filler for any attribute in the reference model that is otherwise unconstrained by an archetype or template. Technically this could occur in any kind of archetype but would normally be in a specialised archetype or template. This is no more than the standard use of an 'external reference' (see section 4.3.9 on page 37).

Any filler specified will have no slot, and is instead validity-checked against the appropriate part of the underlying reference model.

The following example from the *open*EHR reference model is typical.

```
COMPOSITION[id1] matches {                                -- Referral document
    category matches {...}
    context matches {
        EVENT_CONTEXT[id2] matches {
            participations matches {...}
            other_context matches {...}
        }
    }
}
```

The above cADL block partially specifies a `COMPOSITION` object, via constraints (often including slot definitions) on the *category* and *context* attributes defined on that class in the reference model. However, the attribute of most interest in a `COMPOSITION` object is usually the *content* attribute, which is not constrained at all here. The reference model defines it to be of type `List<CONTENT_ITEM>`. 'Filling' this attribute is done using a `use_archetype` declaration, i.e. an external reference. Using an external reference for 'filling' a notional 'slot' in an unarchetyped part of the RM structure is almost always done in specialised archetypes, but is technically valid in a top-level archetype.

The following example shows the use of `use_archetype` within a specialised archetype.

```
COMPOSITION[id1.1] matches {-- Referral document (specialisation)
    content matches {
```

```
            use_archetype SECTION[id2]
                            openEHR-EHR-SECTION.history_medical_surgical.v1
        }
    }
```

9.5.8 Primitive Object Redefinition

For terminal objects (i.e. elements of the type `C_PRIMITIVE_OBJECT`) redefinition consists of:

- redefined value ranges or sets using a narrower value range or set;
- exclusions on the previously defined value ranges or sets which have the effect of narrowing the original range or set.

The following example shows a redefined real value range.

Parent archetype:

```
    value ∈ {
        DV_QUANTITY[id3] ∈ {
            magnitude ∈ {|2.0..10.0|}
            units ∈ {"mmol/ml"}
        }
    }
```

Specialised archetype:

```
    .../value ∈ {
        DV_QUANTITY[id3] ∈ {
            magnitude ∈ {|4.0..6.5|}
        }
    }
```

The following example shows a redefined `CODE_PHRASE` value set.

Parent archetype:

```
    ELEMENT[id7] occurrences ∈ {0..*} ∈ {-- System
        name ∈ {
            DV_CODED_TEXT[id14] ∈ {
                defining_code ∈ {
                    [local::
                    at8, -- Cardiovascular system
                    at9, -- Respiratory system
                    at10, -- Gastro-intestinal system
                    at11, -- Reticulo-Endothelial system
                    at12, -- Genito-urinary system
                    at13, -- Endocrine System
                    at14, -- Central nervous system
                    at15] -- Musculoskeletal system
                }
            }
        }
    }
```

Specialised archetype:

```
    .../name[id14]/defining_code ∈ {
        [local::
        at10, -- Gastro-intestinal system
        at11, -- Reticulo-Endothelial system
        at12, -- Genito-urinary system
        at13, -- Endocrine System
```

```
        at15] -- Musculoskeletal system
    }
```

In the following example, the exclusion operator ∉ (text form: ‘not matches’) is used to remove particular values from a value set.

Parent archetype:

```
    ELEMENT[id7] occurrences ∈ {0..*} ∈ {-- System
        name ∈ {
            DV_CODED_TEXT[id14] ∈ {
                defining_code ∈ {
                    [local::
                    at8, -- Cardiovascular system
                    at9, -- Respiratory system
                    at10, -- Gastro-intestinal system
                    at11, -- Reticulo-Endothelial system
                    at12, -- Genito-urinary system
                    at13, -- Endocrine System
                    at14, -- Central nervous system
                    at15] -- Musculoskeletal system
                }
            }
        }
    }
```

Specialised archetype:

```
    .../name[id14]/defining_code ∉ {
        [local::
        at12, -- Genito-urinary system
        at13] -- Endocrine System
    }
```

The same kind of statement can be used to exclude values from integer ranges, lists, date patterns and all other primitive types, and also on all `C_DOMAIN_TYPE` instances (which appear as ODIN text sections in ADL).

To Be Continued:

9.6 Rules

Assertions in archetypes have the effect of further reducing the instance space that conforms to an archetype by specifying relationships between values that must hold. For example the main part of an archetype may specify that two nodes are of type `DV_QUANTITY` and have values greater than 0. An assertion may be added that requires the two values to be related by a fixed factor, for instance if they record weight in pounds and kilograms. This has the effect of reducing the set of instances that conform to only those where this relation holds.

In specialised archetypes, further invariants can be added, but existing ones cannot be changed. New invariants cannot logically contradict existing invariants.

To Be Continued:

9.7 Languages

9.7.1 Rules for Translations

TBD_9: at least one common language between specialised archetype and parent, or else no flattening possible.

rules for overlaying language domains e.g. en-UK on en, es-ES on es and so on.

9.7.2 Language Sections

9.8 Description Section

9.9 Bindings

TBD_10: bindings in a specialised archetype can include a binding to an at-code or ac-code defined in the current archetype or any parent archetype.

9.10 Conformance Rules

Appendix A Relationship of ADL to Other Formalisms

A.1 Overview

Whenever a new formalism is defined, it is reasonable to ask the question: are there not existing formalisms which would do the same job? Research to date has shown that in fact, no other formalism has been designed for the same use, and none easily express ADL's semantics. During ADL's initial development, it was felt that there was great value in analysing the problem space very carefully, and constructing an abstract syntax exactly matched to the solution, rather than attempting to use some other formalism - undoubtedly designed for a different purpose - to try and express the semantics of archetypes, or worse, to start with an XML-based exchange format, which often leads to the conflation of abstract and concrete representational semantics. Instead, the approach used has paid off, in that the resulting syntax is very simple and powerful, and in fact has allowed mappings to other formalisms to be more correctly defined and understood. The following sections compare ADL to other formalisms and show how it is different.

A.2 Constraint Syntaxes

A.2.1 OMG OCL (Object Constraint Language)

The OMG's Object Constraint Language (OCL) appears at first glance to be an obvious contender for writing archetypes. However, its designed use is to write constraints on *object models*, rather than on *data*, which is what archetypes are about. As a concrete example, OCL can be used to make statements about the *actors* attribute of a class `Company` - e.g. that *actors* must exist and contain the `Actor` who is the *lead* of `Company`. However, if used in the normal way to write constraints on a class model, it cannot describe the notion that for a particular kind of (acting) company, such as 'itinerant jugglers', there must be at least four actors, each of whom have among their *capabilities* 'advanced juggling', plus an Actor who has *skill* 'musician'. This is because doing so would constrain all instances of the class `Company` to conform to the specific configuration of instances corresponding to actors and jugglers, when what is intended is to allow a myriad of possibilities. ADL provides the ability to create numerous archetypes, each describing in detail a concrete configuration of instances of type `Company`.

OCL's constraint types include function pre- and post-conditions, and class invariants. There is no structural character to the syntax - all statements are essentially first-order predicate logic statements about elements in models expressed in UML, and are related to parts of a model by 'context' statements. This makes it impossible to use OCL to express an archetype in a structural way which is natural to domain experts. OCL also has some flaws, described by Beale [4].

However, OCL is in fact relevant to ADL. ADL archetypes include invariants (and one day, might include pre- and post-conditions). Currently these are expressed in a syntax very similar to OCL, with minor differences. The exact definition of the ADL invariant syntax in the future will depend somewhat on the progress of OCL through the OMG standards process.

A.3 Ontology Formalisms

A.3.1 OWL (Web Ontology Language)

The Web Ontology Language (OWL) [20] is a W3C initiative for defining Web-enabled ontologies which aim to allow the building of the "Semantic Web". OWL has an abstract syntax [13], developed at the University of Manchester, UK, and an exchange syntax, which is an extension of the XML-based syntax known as RDF (Resource Description Framework). We discuss OWL only in terms of its abstract syntax, since this is a semantic representation of the language unencumbered by XML or RDF details (there are tools which convert between abstract OWL and various exchange syntaxes).

OWL is a general purpose description logic (DL), and is primarily used to describe "classes" of things in such a way as to support *subsumptive* inferencing within the ontology, and by extension, on data which are instances of ontology classes. There is no general assumption that the data itself were built based on any particular class model - they might be audio-visual objects in an archive, technical documentation for an aircraft or the Web pages of a company. OWL's class definitions are therefore usually constraint statements on an *implied* model on which data *appears* to be based. However, the semantics of an information model can themselves be represented in OWL. Restrictions are the primary way of defining subclasses.

In intention, OWL is aimed at representing some 'reality' and then making inferences about it; for example in a medical ontology, it can infer that a particular patient is at risk of ischemic heart disease due to smoking and high cholesterol, if the knowledge that 'ischemic heart disease has-risk-factor smoking' and 'ischemic heart disease has-risk-factor high cholesterol' are in the ontology, along with a representation of the patient details themselves. OWL's inferencing works by subsumption, which is to say, asserting either that an 'individual' (OWL's equivalent of an object-oriented instance or a type) conforms to a 'class', or that a particular 'class' 'is-a' (subtype of another) 'class'; this approach can also be understood as category-based reasoning or set-containment.

ADL can also be thought of as being aimed at describing a 'reality', and allowing inferences to be made. However, the reality it describes is in terms of constraints on information structures (based on an underlying information model), and the inferencing is between data and the constraints. Some of the differences between ADL and OWL are as follows.

- ADL syntax is predicated on the existence of existing object-oriented reference models, expressed in UML or some similar formalism, and the constraints in an ADL archetype are in relation to types and attributes from such a model. In contrast, OWL is far more general, and requires the explicit expression of a reference model in OWL, before archetype-like constraints can be expressed.
- Because information structures are in general hierarchical compositions of nodes and elements, and may be quite deep, ADL enables constraints to be expressed in a structural, nested way, mimicking the tree-like nature of the data it constrains. OWL does not provide a native way to do this, and although it is possible to express approximately the same constraints in OWL, it is fairly inconvenient, and would probably only be made easy by machine conversion from a visual format more or less like ADL.
- As a natural consequence of dealing with heavily nested structures in a natural way, ADL also provides a path syntax, based on Xpath [21], enabling any node in an archetype to be referenced by a path or path pattern. OWL does not provide an inbuilt path mechanism; Xpath can presumably be used with the RDF representation, although it is not yet clear how meaningful the paths would be with respect to the named categories within an OWL ontology.
- ADL also natively takes care of disengaging natural language and terminology issues from constraint statements by having a separate ontology per archetype, which contains 'bindings' and language-specific translations. OWL has no inbuilt syntax for this, requiring such semantics to be represented from first principles.

- ADL provides a rich set of constraints on primitive types, including dates and times. OWL 1.0 (c 2005) did not provide any equivalents; OWL 1.1 (c 2007) look as though it provides some.

Research to date shows that the semantics of an archetype are likely to be representable inside OWL, assuming expected changes to improve its primitive constraint types occur. To do so would require the following steps:

- express the relevant reference models in OWL (this has been shown to be possible);
- express the relevant terminologies in OWL (research on this is ongoing);
- be able to represent concepts (i.e. constraints) independently of natural language (status unknown);
- convert the cADL part of an archetype to OWL; assuming the problem of primitive type constraints is solved, research to date shows that this should in principle be possible.

To *use* the archetype on data, the data themselves would have to be converted to OWL, i.e. be expressed as 'individuals'. In conclusion, we can say that mathematical equivalence between OWL and ADL is probably provable. However, it is clear that OWL is far from a convenient formalism to express archetypes, or to use them for modelling or reasoning against data. The ADL approach makes use of existing UML semantics and existing terminologies, and adds a convenient syntax for expressing the required constraints. It also appears fairly clear that even if all of the above conversions were achieved, using OWL-expressed archetypes to validate data (which would require massive amounts of data to be converted to OWL statements) is unlikely to be anywhere near as efficient as doing it with archetypes expressed in ADL or one of its concrete expressions.

Nevertheless, OWL provides a very powerful generic reasoning framework, and offers a great deal of inferencing power of far wider scope than the specific kind of 'reasoning' provided by archetypes. It appears that it could be useful for the following archetype-related purposes:

- providing access to ontological resources while authoring archetypes, including terminologies, pure domain-specific ontologies, etc;
- providing a semantic 'indexing' mechanism allowing archetype authors to find archetypes relating to specific subjects (which might not be mentioned literally within the archetypes);
- providing inferencing on archetypes in order to determine if a given archetype is subsumed within another archetype which it does not specialise (in the ADL sense);
- providing access to archetypes from within a semantic Web environment, such as an ebXML server or similar.

Research on these areas is active in the US, UK, Australia, Spain, Denmark and Turkey(mid 2004).

A.3.2 KIF (Knowledge Interchange Format)

The Knowledge Interchange Format (KIF) is a knowledge representation language whose goal is to be able to describe formal semantics which would be sharable among software entities, such as information systems in an airline and a travel agency. An example of KIF (taken from [10]) used to describe the simple concept of "units" in a `QUANTITY` class is as follows:

```
(defrelation BASIC-UNIT
    (=> (BASIC-UNIT ?u) ; basic units are distinguished
       (unit-of-measure ?u))) ; units of measure

(deffunction UNIT*
    ; Unit* maps all pairs of units to units
```

```
        (=> (and (unit-of-measure ?u1)
                 (unit-of-measure ?u2))
            (and (defined (UNIT* ?u1 ?u2))
                 (unit-of-measure (UNIT* ?u1 ?u2))))
    ; It is commutative
    (= (UNIT* ?u1 ?u2) (UNIT* ?u2 ?u1))
    ; It is associative
    (= (UNIT* ?u1 (UNIT* ?u2 ?u3))
        (UNIT* (UNIT* ?u1 ?u2) ?u3)))

    (deffunction UNIT^
        ; Unit^ maps all units and reals to units
        (=> (and (unit-of-measure ?u)
              (real-number ?r))
            (and (defined (UNIT^ ?u ?r))
                 (unit-of-measure (UNIT^ ?u ?r))))
    ; It has the algebraic properties of exponentiation
    (= (UNIT^ ?u 1) ?u)
    (= (unit* (UNIT^ ?u ?r1) (UNIT^ ?u ?r2))
        (UNIT^ ?u (+ ?r1 ?r2)))
    (= (UNIT^ (unit* ?u1 ?u2) ?r)
        (unit* (UNIT^ ?u1 ?r) (UNIT^ ?u2 ?r)))
```

It should be clear from the above that KIF is a definitional language - it defines all the concepts it mentions. However, the most common situation in which we find ourselves is that information models already exist, and may even have been deployed as software. Thus, to use KIF for expressing archetypes, the existing information model and relevant terminologies would have to be converted to KIF statements, before archetypes themselves could be expressed. This is essentially the same process as for expressing archetypes in OWL.

It should also be realised that KIF is intended as a knowledge exchange format, rather than a knowledge representation format, which is to say that it can (in theory) represent the semantics of any other knowledge representation language, such as OWL. This distinction today seems fine, since Web-enabled languages like OWL probably don't need an exchange format other than their XML equivalents to be shared. The relationship and relative strengths and deficiencies is explored by e.g. Martin [11].

A.4 XML-based Formalisms

A.4.1 XML-schema

Previously, archetypes have been expressed as XML instance documents conforming to W3C XML schemas, for example in the Good Electronic Health Record (GeHR; see http://www.gehr.org) and *open*EHR projects. The schemas used in those projects correspond technically to the XML expressions of information model-dependent object models shown in FIGURE 2. XML archetypes are accordingly equivalent to serialised instances of the parse tree, i.e. particular ADL archetypes serialised from objects into XML instance.

Appendix B Syntax Specifications

B.1 cADL Syntax

The grammar for the standard cADL syntax is shown below. The form used in *open*EHR is the same as this, but with custom additions, described in the *open*EHR Archetype Profile. The resulting grammar and lexical analysis specification used in the *open*EHR reference ADL parser is implemented using lex (.l file) and yacc (.y file) specifications for the Eiffel programming environment. The current release of these files is available at http://www.openehr.org/svn/ref_impl_eiffel/TRUNK/components/adl_compiler/src/syntax/cadl/parser. The .l and .y files can be converted for use in other yacc/lex-based programming environments. The production rules of the .y file are available as an HTML document.

B.1.1 Grammar

The following is an extract of the cADL parser production rules (yacc specification) for both flat and differential forms of archetype, as of revision 1834 of the Eiffel reference implementation repository (http://www.openehr.org/svn/ref_impl_eiffel). Note that because of interdependencies with path and assertion production rules, practical implementations may have to include all production rules in one parser.

The flat form syntax is shown first below as it is a subset of the differential form, for which only the changes are shown after. Tool implementers may handles this syntax variation in different ways, but the easiest route may be that used in the reference parser, which is a single syntax within a parser to which a flag is passed indicating the differential/flat status of the archetype text to be parsed.

```
----------------------- flat form of syntax ----------------------
input:
  c_complex_object
| error

c_complex_object:
  c_complex_object_head SYM_MATCHES SYM_START_CBLOCK c_complex_object_body
SYM_END_CBLOCK
| c_complex_object_head -- ok if occurrences redefined in differential arch.

c_complex_object_head:
  c_complex_object_id c_occurrences

c_complex_object_id:
  type_identifier V_ROOT_ID_CODE
| type_identifier V_ID_CODE
| sibling_order type_identifier V_ID_CODE

sibling_order:
  SYM_AFTER V_ID_CODE
| SYM_BEFORE V_ID_CODE

c_complex_object_body:
  c_any
| c_attributes

c_object:
```

```
  c_complex_object
| c_complex_object_proxy
| archetype_slot
| c_primitive_object
| error

c_archetype_root:
  SYM_USE_ARCHETYPE type_identifier V_ID_CODE c_occurrences V_ARCHETYPE_ID
| SYM_USE_NODE type_identifier error

c_complex_object_proxy:
  SYM_USE_NODE type_identifier V_ID_CODE c_occurrences absolute_path
| SYM_USE_NODE type_identifier error


archetype_slot:
  c_archetype_slot_head SYM_MATCHES SYM_START_CBLOCK c_includes c_excludes
SYM_END_CBLOCK
| c_archetype_slot_head -- ok if occurrences is {0}

c_archetype_slot_head:
  c_archetype_slot_id c_occurrences

c_archetype_slot_id:
  SYM_ALLOW_ARCHETYPE type_identifier V_ID_CODE
| sibling_order SYM_ALLOW_ARCHETYPE type_identifier V_ID_CODE
| SYM_ALLOW_ARCHETYPE type_identifier V_ID_CODE SYM_CLOSED
| SYM_ALLOW_ARCHETYPE error

c_primitive_object:
  c_primitive

c_primitive:
  c_integer
| c_real
| c_date
| c_time
| c_date_time
| c_duration
| c_string
| c_boolean
| c_terminology_code

c_any:
  *

c_attribute_defs:
  c_attribute_def
| c_attributes_defs c_attribute_def

c_attribute_def:
  c_attribute
| c_attributes_tuple

c_attribute:
  c_attr_head SYM_MATCHES SYM_START_CBLOCK c_attr_values SYM_END_CBLOCK
```

```
c_attr_head:
  V_ATTRIBUTE_IDENTIFIER c_existence c_cardinality
| V_ABS_PATH c_existence c_cardinality

c_attr_values:
  c_object
| c_attr_values c_object
| c_any

c_attribute_tuple:
  [ c_tuple_attr_ids ] SYM_MATCHES SYM_START_CBLOCK c_attr_tuple_values
SYM_END_CBLOCK

c_tuple_attr_ids:
  V_ATTRIBUTE_IDENTIFIER
| c_tuple_attr_ids , V_ATTRIBUTE_IDENTIFIER

c_attr_tuple_values:
  c_attr_tuple_value
| c_attr_tuple_values , c_attr_tuple_value

c_attr_tuple_value:
  [ c_tuple_values ]

c_tuple_values:
  SYM_START_CBLOCK c_primitive_object SYM_END_CBLOCK
| c_tuple_values , SYM_START_CBLOCK c_primitive_object SYM_END_CBLOCK

c_includes:
  -/-
| SYM_INCLUDE assertions

c_excludes:
  -/-
| SYM_EXCLUDE assertions

c_existence:
  -/-
| SYM_EXISTENCE SYM_MATCHES SYM_START_CBLOCK existence_spec SYM_END_CBLOCK

existence_spec:
  V_INTEGER
| V_INTEGER SYM_ELLIPSIS V_INTEGER

c_cardinality:
  -/-
| SYM_CARDINALITY SYM_MATCHES SYM_START_CBLOCK cardinality_range
SYM_END_CBLOCK

cardinality_range:
  occurrence_spec
| occurrence_spec ; SYM_ORDERED
| occurrence_spec ; SYM_UNORDERED
| occurrence_spec ; SYM_UNIQUE
| occurrence_spec ; SYM_ORDERED ; SYM_UNIQUE
| occurrence_spec ; SYM_UNORDERED ; SYM_UNIQUE
| occurrence_spec ; SYM_UNIQUE ; SYM_ORDERED
| occurrence_spec ; SYM_UNIQUE ; SYM_UNORDERED
```

```
c_occurrences:
  -/-
| SYM_OCCURRENCES SYM_MATCHES SYM_START_CBLOCK occurrence_spec
SYM_END_CBLOCK
| SYM_OCCURRENCES error

occurrence_spec:
  integer_value
| *
| V_INTEGER SYM_ELLIPSIS integer_value
| V_INTEGER SYM_ELLIPSIS *


c_integer:
  integer_value
| integer_list_value
| integer_interval_value
| c_integer ; integer_value
| c_integer ; error

c_real:
  real_value
| real_list_value
| real_interval_value
| c_real ; real_value
| c_real ; error

c_date:
  V_ISO8601_DATE_CONSTRAINT_PATTERN
| date_value
| date_interval_value
| c_date ; date_value
| c_date ; error

c_time:
  V_ISO8601_TIME_CONSTRAINT_PATTERN
| time_value
| time_interval_value
| c_time ; time_value
| c_time ; error

c_date_time:
  V_ISO8601_DATE_TIME_CONSTRAINT_PATTERN
| date_time_value
| date_time_interval_value
| c_date_time ; date_time_value
| c_date_time ; error

c_duration:
  V_ISO8601_DURATION_CONSTRAINT_PATTERN
| V_ISO8601_DURATION_CONSTRAINT_PATTERN / duration_interval_value
| duration_value
| duration_interval_value
| c_duration ; duration_value
| c_duration ; error
```

```
c_string:
    V_STRING
  | string_list_value
  | string_list_value_continue
  | V_REGEXP
  | c_string ; string_value
  | c_string ; error

c_boolean:
    SYM_TRUE
  | SYM_FALSE
  | SYM_TRUE , SYM_FALSE
  | SYM_FALSE , SYM_TRUE
  | c_boolean ; boolean_value
  | c_boolean ; error


any_identifier:
    type_identifier
  | V_ATTRIBUTE_IDENTIFIER

-- for string_value etc, see ODIN spec

-- for assertions, assertion, see Assertion spec
```

B.1.2 Symbols

The following shows the lexical specification for the cADL grammar. The keywords `SYM_BEFORE` and `SYM_AFTER` occur only in the differential form of the syntax.

```
----------/* definitions */ -----------------------------------------------
ALPHANUM [a-zA-Z0-9]
ALPHANUM_STR [a-zA-Z0-9_]+
IDCHAR [a-zA-Z0-9_]
NAMECHAR [a-zA-Z0-9._\-]
NAMECHAR_SPACE [a-zA-Z0-9._\- ]
NAMECHAR_PAREN [a-zA-Z0-9._\-()]
NAMESTR [a-zA-Z][a-zA-Z0-9_]+

ID_CODE_LEADER id
CODE_STR (0|[1-9][0-9]*)(\.(0|[1-9][0-9]*))*
PATH_SEG [a-z][a-zA-Z0-9_]*(\[id(0|[1-9][0-9]*)(\.(0|[1-9][0-9]*))*\])?

UTF8CHAR (([\xC2-\xDF][\x80-\xBF])|(\xE0[\xA0-\xBF][\x80-\xBF])|([\xE1-
\xEF][\x80-\xBF][\x80-\xBF])|(\xF0[\x90-\xBF][\x80-\xBF][\x80-\xBF])|([\xF1-
\xF7][\x80-\xBF][\x80-\xBF][\x80-\xBF]))

----------/* comments */ -------------------------------------------------
"--".*                                      -- Ignore comments
"--".*\n[ \t\r]*

----------/* symbols */ -------------------------------------------------
"-"         -- -> Minus_code
"+"         -- -> Plus_code
"*"         -- -> Star_code
"/"         -- -> Slash_code
```

```
"^"         -- -> Caret_code
"="         -- -> Equal_code
"."         -- -> Dot_code
";"         -- -> Semicolon_code
","         -- -> Comma_code
":"         -- -> Colon_code
"!"         -- -> Exclamation_code
"("         -- -> Left_parenthesis_code
")"         -- -> Right_parenthesis_code
"$"         -- -> Dollar_code

"??"        -- -> SYM_DT_UNKNOWN
"?"         -- -> Question_mark_code

"|"         -- -> SYM_INTERVAL_DELIM

"["         -- -> Left_bracket_code
"]"         -- -> Right_bracket_code

"{"         -- -> SYM_START_CBLOCK
"}"         -- -> SYM_END_CBLOCK

".."        -- -> SYM_ELLIPSIS
"..."       -- -> SYM_LIST_CONTINUE

----------/* common keywords */ --------------------------------------

[Mm][Aa][Tt][Cc][Hh][Ee][Ss] -- -> SYM_MATCHES

[Ii][Ss]_[Ii][Nn]            -- -> SYM_MATCHES

----------/* assertion keywords */ -----------------------------------

[Tt][Hh][Ee][Nn]             -- -> SYM_THEN

[Ee][Ll][Ss][Ee]             -- -> SYM_ELSE

[Aa][Nn][Dd]                 -- -> SYM_AND

[Oo][Rr]                     -- -> SYM_OR

[Xx][Oo][Rr]                 -- -> SYM_XOR

[Nn][Oo][Tt]                 -- -> SYM_NOT

[Ii][Mm][Pp][Ll][Ii][Ee][Ss]    -- -> SYM_IMPLIES

[Tt][Rr][Uu][Ee]             -- -> SYM_TRUE

[Ff][Aa][Ll][Ss][Ee]         -- -> SYM_FALSE

[Ff][Oo][Rr][_][Aa][Ll][Ll]     -- -> SYM_FORALL

[Ee][Xx][Ii][Ss][Tt][Ss]        -- -> SYM_EXISTS

---------/* cADL keywords */ ---------------------------------------
```

```
[Ee][Xx][Ii][Ss][Tt][Ee][Nn][Cc][Ee]                    -- -> SYM_EXISTENCE

[Oo][Cc][Cc][Uu][Rr][Rr][Ee][Nn][Cc][Ee][Ss]            -- -> SYM_OCCURRENCES

[Cc][Aa][Rr][Dd][Ii][Nn][Aa][Ll][Ii][Tt][Yy]            -- -> SYM_CARDINALITY

[Oo][Rr][Dd][Ee][Rr][Ee][Dd]                            -- -> SYM_ORDERED

[Uu][Nn][Oo][Rr][Dd][Ee][Rr][Ee][Dd]                    -- -> SYM_UNORDERED

[Uu][Nn][Ii][Qq][Uu][Ee]                                -- -> SYM_UNIQUE

[Ii][Nn][Ff][Ii][Nn][Ii][Tt][Yy]                        -- -> SYM_INFINITY

[Uu][Ss][Ee][_][Nn][Oo][Dd][Ee]                         -- -> SYM_USE_NODE

[Uu][Ss][Ee][_][Aa][Rr][Cc][Hh][Ee][Tt][Yy][Pp][Ee]  -- ->
SYM_ALLOW_ARCHETYPE

[Aa][Ll][Ll][Oo][Ww][_][Aa][Rr][Cc][Hh][Ee][Tt][Yy][Pp][Ee]
SYM_ALLOW_ARCHETYPE

[Ii][Nn][Cc][Ll][Uu][Dd][Ee]                            -- -> SYM_INCLUDE

[Ee][Xx][Cc][Ll][Uu][Dd][Ee]                            -- -> SYM_EXCLUDE

[Aa][Ff][Tt][Ee][Rr]                                    -- -> SYM_AFTER

[Bb][Ee][Ff][Oo][Rr][Ee]                                -- -> SYM_BEFORE

[Cc][Ll][Oo][Ss][Ee][Dd]                                -- -> SYM_CLOSED

----------/* V_URI */ -------------------------------------------------
[a-z]+:\/\/[^<>|\\{}^~"\[\] ]*

-------/* V_ROOT_ID_CODE: id code of form [at0], [at0.1] etc */ -----

\[{ID_CODE_LEADER}1(\.1)*\]

--------/* V_ID_CODE: id code of form [id1], [id1.4] */ -------------
\[{ID_CODE_LEADER}{CODE_STR}\]

----------/* V_ID_CODE_STR */ ---------------------------------------
{ID_CODE_LEADER}{CODE_STR}

------/* V_VALUE_SET_REF: term code of form [ac2], [ac0.0.2] */ -----
\[{NAMECHAR_PAREN}+::ac{CODE_STR}\]

---------/* V_VALUE_SET_DEF of form */ ------------
-- [terminology_id::code, -- comment
--              code, -- comment
--              code] -- comment
--
-- Form with assumed value
-- [terminology_id::code, -- comment
--            code; -- comment
--            code] -- an optional assumed value
--
```

```
\[[a-zA-Z0-9()._\-]+::[ \t\n]*        -- start IN_TERM_CONSTRAINT

<IN_VALUE_SET_DEF> {
    [ \t]*{NAMECHAR}+[ \t]*;[ \t]*
       -- match second last line with ';' termination (assumed value)

    [ \t]*{NAMECHAR}+[ \t]*,[ \t]*
          -- match any line, with ',' termination

    \n+         -- count line endings

    \-\-[^\n]*\n -- ignore comments

    [ \t]*{NAMECHAR}*[ \t]*\] -- match final line, terminating in ']'

----------/* archetype id */ ----------------------------------------
({NAMESTR}(\.{ALPHANUM_STR})*::)?{NAMESTR}-{ALPHANUM_STR}-
{NAMESTR}\.{NAMESTR}(-{ALPHANUM_STR})*\.v[0-9]+((\.[0-9]+){0,2}((-
rc|\+u|\+)[0-9]+)?)?

------/* V_ISO8601_EXTENDED_DATE_TIME */ ---
-- YYYY-MM-DDThh:mm:ss[,sss][Z|+/-nnnn]
--
[0-9]{4}-[0-1][0-9]-[0-3][0-9]T[0-2][0-9]:[0-6][0-9]:[0-6][0-9](,[0-
9]+)?(Z|[+-][0-9]{4})? |
[0-9]{4}-[0-1][0-9]-[0-3][0-9]T[0-2][0-9]:[0-6][0-9](Z|[+-][0-9]{4})? |
[0-9]{4}-[0-1][0-9]-[0-3][0-9]T[0-2][0-9](Z|[+-][0-9]{4})?
----------/* V_ISO8601_EXTENDED_TIME */ --------
-- hh:mm:ss[,sss][Z|+/-nnnn]
--
[0-2][0-9]:[0-6][0-9]:[0-6][0-9](,[0-9]+)?(Z|[+-][0-9]{4})? |
[0-2][0-9]:[0-6][0-9](Z|[+-][0-9]{4})?

----------/* V_ISO8601_DATE YYYY-MM-DD */ --------------------
[0-9]{4}-[0-1][0-9]-[0-3][0-9] |
[0-9]{4}-[0-1][0-9]

----------/* V_ISO8601_DURATION */ -------------------------
P([0-9]+[yY])?([0-9]+[mM])?([0-9]+[wW])?([0-9]+[dD])?T([0-9]+[hH])?([0-9]+[mM])?([0-9]+[sS])? |
P([0-9]+[yY])?([0-9]+[mM])?([0-9]+[wW])?([0-9]+[dD])?

----------/* V_ISO8601_DATE_CONSTRAINT_PATTERN */ -----------------
[yY][yY][yY][yY]-[mM?X][mM?X]-[dD?X][dD?X]

----------/* V_ISO8601_TIME_CONSTRAINT_PATTERN */ ------------------
[hH][hH]:[mM?X][mM?X]:[sS?X][sS?X]

----------/* V_ISO8601_DATE_TIME_CONSTRAINT_PATTERN */ -------------
[yY][yY][yY][yY]-[mM?][mM?]-[dD?X][dD?X][ T][hH?X][hH?X]:[mM?X][mM?X]:[sS?X][sS?X]

----------/* V_ISO8601_DURATION_CONSTRAINT_PATTERN */ --------------
P[yY]?[mM]?[wW]?[dD]?T[hH]?[mM]?[sS]? |
P[yY]?[mM]?[wW]?[dD]?

----------/* V_TYPE_IDENTIFIER */ ------------------------------------
[A-Z]{IDCHAR}*
```

```
----------/* V_GENERIC_TYPE_IDENTIFIER */ ----------------------------
[A-Z]{IDCHAR}*<[a-zA-Z0-9,_<>]+>

----------/* V_FEATURE_CALL_IDENTIFIER */ ----------------------------
[a-z]{IDCHAR}*[ ]*\(\)

----------/* V_ATTRIBUTE_IDENTIFIER */ ----------------------------
[a-z]{IDCHAR}*

----------/* V_GENERIC_TYPE_IDENTIFIER */ -------------------------------
[A-Z]{IDCHAR}*<[a-zA-Z0-9,_<>]+>

----------/* V_ATTRIBUTE_IDENTIFIER */ -------------------------------
[a-z]{IDCHAR}*

----------/* V_ABS_PATH */ -------------------------------------------
(\/{PATH_SEG})+

----------/* V_REL_PATH */ -------------------------------------------
{PATH_SEG}(\/{PATH_SEG})+

----------/* V_REGEXP */ -------------------------------------
-- {mini-parser specification}
"{/"                         -- start of regexp
<IN_REGEXP1>[^/]*\\\/        -- match any segments with quoted slashes
<IN_REGEXP1>[^/}]*\/         -- match final segment

\^[^^\n]*\^{                 -- regexp formed using '^' delimiters

----------/* V_INTEGER */ -----------------------------------------------
[0-9]+

----------/* V_REAL */ -----------------------------------------------
[0-9]+\.[0-9]+
[0-9]+\.[0-9]+[eE][+-]?[0-9]+

----------/* V_STRING */ -----------------------------------------------
\"[^\\\n"]*\"

\"[^\\\n"]*{              -- beginning of a multi-line string
<IN_STR> {
    \\\\                  -- match escaped backslash, i.e. \\ -> \
    \\\"                  -- match escaped double quote, i.e. \” -> “
    {UTF8CHAR}+           -- match UTF8 chars
    [^\\\n"]+             -- match any other characters
    \\\n[ \t\r]*          -- match LF in line
    [^\\\n"]*\"           -- match final end of string

    .|\n      |
    <<EOF>>               -- unclosed String -> ERR_STRING
}
```

B.2 Assertion Syntax

The assertion grammar is part of the cADL grammar, which is available as an HTML document. This grammar is implemented and tested using lex (.l file) and yacc (.y file) specifications for in the Eiffel programming environment. The current release of these files is available at http://www.openehr.org/svn/ref_impl_eiffel/TRUNK/components/adl_parser/src/syntax/cadl/parser. The .l and .y files can easily be converted for use in another yacc/lex-based programming environment.

B.2.1 Grammar

The following provides the cADL parser production rules (yacc specification) as of revision 1834 of the Eiffel reference implementation repository (http://www.openehr.org/svn/ref_impl_eiffel). Note that because of interdependcies with path and assertion production rules, practical implementations may have to include all production rules in one parser.

```
assertions:
  assertion
| assertions assertion

assertion:
  any_identifier : boolean_node
| boolean_node
| arch_outer_constraint_expr
| any_identifier : error

boolean_node:
  boolean_leaf
| boolean_unop_expr
| boolean_binop_expr
| arithmetic_relop_expr
| boolean_constraint_expr
| ( boolean_node )

arch_outer_constraint_expr:
  V_REL_PATH SYM_MATCHES SYM_START_CBLOCK c_primitive SYM_END_CBLOCK

boolean_constraint_expr:
  V_ABS_PATH SYM_MATCHES SYM_START_CBLOCK c_primitive SYM_END_CBLOCK
| V_ABS_PATH SYM_MATCHES SYM_START_CBLOCK c_code_phrase SYM_END_CBLOCK

boolean_unop_expr:
  SYM_EXISTS V_ABS_PATH
| SYM_NOT V_ABS_PATH
| SYM_NOT ( boolean_node )
| SYM_EXISTS error

boolean_binop_expr:
  boolean_node boolean_binop_symbol boolean_node

boolean_binop_symbol:
  SYM_OR
| SYM_AND
| SYM_XOR
| SYM_IMPLIES

boolean_leaf:
  SYM_TRUE
| SYM_FALSE
```

```
arithmetic_relop_expr:
  arithmetic_node relational_binop_symbol arithmetic_node

arithmetic_node:
  arithmetic_leaf
| arithmetic_arith_binop_expr
| ( arithmetic_node )

arithmetic_arith_binop_expr:
  arithmetic_node arithmetic_binop_symbol arithmetic_node

arithmetic_leaf:
  integer_value
| real_value
| V_ABS_PATH

relational_binop_symbol:
  =
| SYM_NE
| SYM_LE
| SYM_LT
| SYM_GE
| SYM_GT

arithmetic_binop_symbol:
  /
| *
| +
| -
| ^
```

B.3 Path Syntax

The Path syntax used in ADL is defined by the grammar below. This grammar is implemented using lex (.l file) and yacc (.y file) specifications for in the Eiffel programming environment. The current release of these files is available at http://www.openehr.org/svn/ref_impl_eiffel/libraries/common_libs/src/structures/object_graph/path. The .l and .y files can easily be converted for use in another yacc/lex-based programming environment. The ADL path grammar is available as an HTML document.

B.3.1 Grammar

The following provides the ADL path parser production rules (yacc specification) as of revision 166 of the Eiffel reference implementation repository (http://www.openehr.org/svn/ref_impl_eiffel).

```
input:
  movable_path
| absolute_path
| relative_path
| error

movable_path:
  SYM_MOVABLE_LEADER relative_path
```

```
absolute_path:
  / relative_path
| absolute_path / relative_path

relative_path:
  path_segment
| relative_path / path_segment

path_segment:
  V_ATTRIBUTE_IDENTIFIER V_LOCAL_TERM_CODE_REF
| V_ATTRIBUTE_IDENTIFIER
```

B.3.2 Symbols

The following specifies the symbols and lexical patterns used in the path grammar.

```
----------/* symbols */ ------------------------------------------------
"."         Dot_code
"/"         Slash_code

"["         Left_bracket_code
"]"         Right_bracket_code

"//"        SYM_MOVABLE_LEADER

----------/* local term code reference */ --------------------------------
\[a[ct][0-9.]+\]                     V_LOCAL_TERM_CODE_REF

----------/* identifiers */ ----------------------------------------------
[a-z][a-zA-Z0-9_]*                   V_ATTRIBUTE_IDENTIFIER
```

B.4 ADL Syntax

The following syntax and lexical specification are used to process an entire ADL file. Their main job is reading the header items, and then cutting it up into ODIN, cADL and assertion sections. With the advent of ADL2, this will no longer be needed, since every ADL text will in fact just be an ODIN text containing embedded sections in cADL and the assertion syntax.

See the *open*EHR Archetype Object Model (AOM) for more details on the ADL parsing process.

B.4.1 Grammar

This section describes the ADL grammar, as implemented and tested in revision 166 of the Eiffel reference implementation repository (http://svn.openehr.org/ref_impl_eiffel).

```
input:
  archetype
| specialised_archetype
| template
| template_overlay
| operational_template
| error

archetype:
```

```
  archetype_marker arch_meta_data archetype_id arch_language
arch_description arch_definition arch_rules arch_terminology
arch_annotations
```

specialised_archetype:

```
  archetype_marker arch_meta_data archetype_id arch_specialisation
arch_language arch_description arch_definition arch_rules arch_terminology
arch_annotations
```

template:

```
  template_marker arch_meta_data archetype_id arch_specialisation
arch_language arch_description arch_definition arch_rules arch_terminology
arch_annotations
```

template_overlay:

```
  template_overlay_marker arch_meta_data archetype_id arch_specialisation
arch_language arch_definition arch_terminology
```

operational_template:

```
  operational_template_marker arch_meta_data archetype_id arch_language
arch_description arch_definition arch_rules arch_terminology
arch_annotations arch_component_terminologies
```

archetype_marker:

```
  SYM_ARCHETYPE
```

template_marker:

```
  SYM_TEMPLATE
```

template_overlay_marker:

```
  SYM_TEMPLATE_OVERLAY
```

operational_template_marker:

```
  SYM_OPERATIONAL_TEMPLATE
```

archetype_id:

```
  V_ARCHETYPE_ID
```

arch_meta_data:

```
  -/-
| ( arch_meta_data_items )
```

arch_meta_data_items:

```
  arch_meta_data_item
| arch_meta_data_items ; arch_meta_data_item
```

arch_meta_data_item:

```
  SYM_ADL_VERSION = V_DOTTED_NUMERIC
| SYM_UID = V_DOTTED_NUMERIC
| SYM_UID = V_VALUE
| SYM_IS_CONTROLLED
| SYM_IS_GENERATED
| V_IDENTIFIER = V_IDENTIFIER
| V_IDENTIFIER = V_VALUE
| V_IDENTIFIER
| V_VALUE
```

arch_specialisation:

```
  SYM_SPECIALIZE V_ARCHETYPE_ID
| SYM_SPECIALIZE error
```

```
arch_language:
  SYM_LANGUAGE V_ODIN_TEXT
| SYM_LANGUAGE error

arch_description:
  SYM_DESCRIPTION V_ODIN_TEXT
| SYM_DESCRIPTION error

arch_definition:
  SYM_DEFINITION V_CADL_TEXT
| SYM_DEFINITION error

arch_rules:
  -/-
| SYM_RULES V_RULES_TEXT
| SYM_RULES error

arch_terminology:
  SYM_TERMINOLOGY V_ODIN_TEXT
| SYM_TERMINOLOGY error

arch_annotations:
  -/-
| SYM_ANNOTATIONS V_ODIN_TEXT
| SYM_ANNOTATIONS error

arch_component_terminologies:
  SYM_COMPONENT_TERMINOLOGIES V_ODIN_TEXT
| SYM_COMPONENT_TERMINOLOGIES error
```

B.4.2 Symbols

The following shows the ADL lexical specification.

```
----------/* symbols */ -------------------------------------------------
“-”       Minus_code
“+”       Plus_code
“*”       Star_code
“/”       Slash_code
“^”       Caret_code
“=”       Equal_code
“.”       Dot_code
“;”       Semicolon_code
“,”       Comma_code
“:”       Colon_code
“!”       Exclamation_code
“(“       Left_parenthesis_code
“)”       Right_parenthesis_code
“$”       Dollar_code
“?”       Question_mark_code

“[“       Left_bracket_code
“]”       Right_bracket_code

----------/* keywords */ -------------------------------------------------
^[Aa][Rr][Cc][Hh][Ee][Tt][Yy][Pp][Ee][ \t\r]*     SYM_ARCHETYPE
^[Tt][Ee][Mm][Pp][Ll][Aa][Tt][Ee][ \t\r]*         SYM_TEMPLATE
```

```
^[Tt][Ee][Mm][Pp][Ll][Aa][Tt][Ee]_[Oo][Vv][Ee][Rr][Ll][Aa][Yy][ \t\r\n]+
                                                SYM_TEMPLATE_OVERLAY
^[Oo][Pp][Ee][Rr][Aa][Tt][Ii][Oo][Nn][Aa][Ll]_[Tt][Ee][Mm][Pp][Ll][Aa][Tt][E
e][ \t\r]*                                      SYM_OPERAIONAL_TEMPLATE

[Aa][Dd][Ll]_[Vv][Ee][Rr][Ss][Ii][Oo][Nn]       SYM_ADL_VERSION
[Cc][Oo][Nn][Nn][Tt][Rr][Oo][Ll][Ll][Ee][Dd]    SYM_IS_CONTROLLED
[Gg][Ee][Nn][Ee][Rr][Aa][Tt][Ee][Dd]            SYM_IS_GENERATED

^[Ss][Pp][Ee][Cc][Ii][Aa][Ll][Ii][SsZz][Ee][ \t\r]*\n        SYM_SPECIALIZE
^[Cc][Oo][Nn][Cc][Ee][Pp][Tt][ \t\r]*\n                      SYM_CONCEPT
^[Dd][Ee][Ff][Ii][Nn][Ii][Tt][Ii][Oo][Nn][ \t\r]*\n          SYM_DEFINITION
    -- mini-parser to match V_DADL_TEXT

^[Ll][Aa][Nn][Gg][Uu][Aa][Gg][Ee][ \t\r]*\n                  SYM_LANGUAGE
    -- mini-parser to match V_DADL_TEXT

^[Dd][Ee][Ss][Cc][Rr][Ii][Pp][Tt][Ii][Oo][Nn][ \t\r]*\n      SYM_DESCRIPTION
    -- mini-parser to match V_CADL_TEXT

^[Ii][Nn][Vv][Aa][Rr][Ii][Aa][Nn][Tt][ \t\r]*\n              SYM_INVARIANT
    -- mini-parser to match V_ASSERTION_TEXT

^[Tt][Ee][Rr][Mm][Ii][Nn][Oo][Ll][Oo][Gg][Yy][ \t\r]*\n      SYM_TERMINOLOGY
    -- mini-parser to match V_DADL_TEXT

^[Aa][Nn][Nn][Oo][Tt][Aa][Tt][Ii][Oo][Nn][Ss][ \t\r]*\n      SYM_ANNOTATIONS
    -- mini-parser to match V_DADL_TEXT

^[Cc][Oo][Mm][Pp][Oo][Nn][Ee][Nn][Tt]_[Tt][Ee][Rr][Mm][Ii][Nn][Oo][Ll][Oo][G
g][Ii][Ee][Ss][ \t\r]*\n                        SYM_COMPONENT_TERMINOLOGIES
    -- mini-parser to match V_DADL_TEXT

----------/* V_DADL_TEXT */ -------------------------------------
<IN_DADL_SECTION>{
    -- the following 2 patterns are a hack, until ADL2 comes into being;
    -- until then, ODIN blocks in an archetype finish when they
    -- hit EOF, or else the 'description' or 'definition' keywords.
    -- It's not nice, but it's simple ;-)
    -- For both these patterns, the lexer has to unread what it
    -- has just matched, store the ODIN text so far, then get out
    -- of the IN_DADL_SECTION state
    ^[Dd][Ee][Ss][Cc][Rr][Ii][Pp][Tt][Ii][Oo][Nn][ \t\r]*\n|
    ^[Dd][Ee][Ff][Ii][Nn][Ii][Tt][Ii][Oo][Nn][ \t\r]*\n |
    ^[Aa][Nn][Nn][Oo][Tt][Aa][Tt][Ii][Oo][Nn][Ss][ \t\r]*\n|
^[Cc][Oo][Mm][Pp][Oo][Nn][Ee][Nn][Tt]_[Tt][Ee][Rr][Mm][Ii][Nn][Oo][Ll][Oo][G
g][Ii][Ee][Ss][ \t\r]*\n

    [^\n]+\n             -- any text on line with a LF
    [^\n]+               -- any text on line with no LF
    \n                   -- LF
    <<EOF>>              -- (escape condition)
    (.|\n)               -- ignore unmatched chars
}

----------/* V_CADL_TEXT */ -------------------------------------
<IN_CADL_SECTION>{
    ^[ \t]+[^\n]*\n      -- non-blank lines
```

```
    \n+                       -- blank lines
    ^[^ \t]                   -- non-white space at start (escape condition)
}

----------/* V_ASSERTION_TEXT */ -------------------------------------
<IN_ASSERTION_SECTION>{
    ^[ \t]+[^\n]*\n           -- non-blank lines
    ^[^ \t]                   -- non-white space at start (escape condition)
}

----------/* V_VERSION_STRING */ -------------------------------------
[0-9]+\.[0-9]+(\.[0-9]+)*                                   V_VERSION_STRING

----------/* archetype id */ ---------------------------------------------
({NAMESTR}(\.{ALPHANUM_STR})*::)?{NAMESTR}-{ALPHANUM_STR}-
{NAMESTR}\.{NAMESTR}(-{ALPHANUM_STR})*\.v[0-9]+((\.[0-9]+){0,2}((-
rc|\+u|\+)[0-9]+)?)?
                                                           V_ARCHETYPE_ID

----------/* identifiers */ ----------------------------------------------
{NAMESTR}                                              V_IDENTIFIER
```

References

Publications

1 Beale T. *Archetypes: Constraint-based Domain Models for Future-proof Information Systems*. OOPSLA 2002 workshop on behavioural semantics.
Available at http://www.deepthought.com.au/it/archetypes.html.

2 Beale T. *Archetypes: Constraint-based Domain Models for Future-proof Information Systems*. 2000. Available at http://www.deepthought.com.au/it/archetypes.html.

3 Beale T, Heard S. The openEHR Archetype Object Model. See http://www.openehr.org/repositories/spec-dev/latest/publishing/architecture/archetypes/archetype_model/REV_HIST.html.

4 Beale T. *A Short Review of OCL*.
See http://www.deepthought.com.au/it/ocl_review.html.

5 Dolin R, Elkin P, Mead C *et al*. *HL7 Templates Proposal*. 2002.
Available at http://www.hl7.org.

6 Heard S, Beale T. Archetype Definitions and Principles. See http://www.openehr.org/repositories/spec-dev/latest/publishing/architecture/archetypes/principles/REV_HIST.html.

7 Heard S, Beale T. *The openEHR Archetype System*. See http://www.openehr.org/repositories/spec-dev/latest/publishing/architecture/archetypes/system/REV_HIST.html.

8 Hein J L. *Discrete Structures, Logic and Computability (2nd Ed)*. Jones and Bartlett 2002.

9 Kilov H, Ross J. *Information Modelling: an Object-Oriented Approach*. Prentice Hall 1994.

10 Gruber T R. *Toward Principles for the Design of Ontologies Used for Knowledge Sharing*. in Formal Ontology in Conceptual Analysis and Knowledge Representation. Eds Guarino N, Poli R. Kluwer Academic Publishers. 1993 (Aug revision).

11 Martin P. Translations between UML, OWL, KIF and the WebKB-2 languages (For-Taxonomy, Frame-CG, Formalized English). May/June 2003. Available at http://meganesia.int.gu.edu.au/~phmartin/WebKB/doc/model/comparisons.html as at Aug 2004.

12 Meyer B. *Eiffel the Language (2nd Ed)*. Prentice Hall, 1992.

13 Patel-Schneider P, Horrocks I, Hayes P. *OWL Web Ontology Language Semantics and Abstract Syntax*. See http://w3c.org/TR/owl-semantics/.

14 Smith G. *The Object Z Specification Language*. Kluwer Academic Publishers 2000. See http://www.itee.uq.edu.au/~smith/objectz.html.

15 Sowa J F. *Knowledge Representation: Logical, philosophical and Computational Foundations*. 2000, Brooks/Cole, California.

Resources

16 HL7 v3 RIM. See http://www.hl7.org.

17 HL7 Templates Proposal. See http://www.hl7.org.

18 *open*EHR. EHR Reference Model. See http://www.openehr.org/repositories/spec-dev/latest/publishing/architecture/top.html.

19 Perl Regular Expressions. See http://www.perldoc.com/perl5.6/pod/perlre.html.

20 SynEx project, UCL. http://www.chime.ucl.ac.uk/HealthI/SynEx/.

21 W3C. *OWL - the Web Ontology Language*.
See http://www.w3.org/TR/2003/CR-owl-ref-20030818/.

22 W3C. XML Path Language. See http://w3c.org/TR/xpath.

END OF DOCUMENT